

# *A Gentle Introduction To Mercury*

*The Functional-Logic Programming Language*

*Zetian Lin*



# *A Gentle Introduction to Mercury*

*The Functional-Logic Programming Language*

Zetian Lin

(Very early draft; October 2025)

© 2025 Zetian Lin. Some rights reserved.  
<https://sebastian.graphics>

Licensed under Creative Commons Attribution-NonCommercial-NoDerivatives 4.0 International with additional terms. To view a copy of the Creative Commons license, please visit <https://creativecommons.org/licenses/by-nc-nd/4.0/>.

Additional terms requested by the author:

All content is shared according to the policies above with all parties EXCEPT any Wikimedia Foundation projects. Under NO CIRCUMSTANCES may the content of this work be reproduced, distributed or transmitted in any form or by any means, including photocopying, recording, or other electronic or mechanical methods for the use of ANY existing and future Wikimedia Foundation project. Permission for any individual or organization to use the content of this work in any form for Wikimedia Foundation projects is hereby explicitly revoked. Any individual or organization that uses the content for any Wikimedia Foundation project will be subject to legal action.



# Contents

## Contents

<b>1</b>	<b>Introduction</b>	<b>1</b>
1.1	What is Mercury? . . . . .	1
1.2	The state of things as of now . . . . .	2
1.3	Who's this book for . . . . .	4
1.4	The tool we will be using . . . . .	4
1.5	More importantly... <i>why?</i> . . . . .	5
<b>2</b>	<b>A First Taste of Mercury</b>	<b>7</b>
2.1	Prerequisite: logic programming . . . . .	7
2.2	Writing the program . . . . .	14
<b>3</b>	<b>Programming in Mercury</b>	<b>29</b>
3.1	Trace Goals . . . . .	29
3.2	Using the standard library . . . . .	31
3.3	Basic Datatypes . . . . .	32
3.4	If-Then-Else . . . . .	39
3.5	Fundamental concepts . . . . .	39
3.6	Defining functions . . . . .	41
3.7	User-defined types . . . . .	43
3.8	Example 1: Interpreter of a small language . . . . .	48
3.9	Higher-order programming . . . . .	55
3.10	Exception handling . . . . .	58
3.11	Advanced goals . . . . .	62
3.12	Using multiple modules . . . . .	64
3.13	Installing and using libraries . . . . .	68
<b>4</b>	<b>Advanced Concepts &amp; Usage</b>	<b>71</b>
4.1	Instantiatedness . . . . .	71
4.2	Modes . . . . .	73

## CONTENTS

4.3	Multiple solutions . . . . .	73
4.4	Definite Clause Grammar . . . . .	73
4.5	Typeclasses . . . . .	73
4.6	Existential Types . . . . .	73
4.7	Purity System . . . . .	73
<b>5</b>	<b>Parsing With Definite Clause Grammar</b>	<b>75</b>
<b>6</b>	<b>Foreign Language Interface</b>	<b>77</b>
6.1	Calling back and forth . . . . .	77
6.2	Working with the Java grade . . . . .	77
6.3	Working with the CSharp grade . . . . .	77
<b>7</b>	<b>Debugging Mercury Programs</b>	<b>79</b>
<b>A</b>	<b>.gitignore for Mercury projects</b>	<b>81</b>
A.1	Building artifacts . . . . .	81
<b>B</b>	<b>Full source code for mmgemtext</b>	<b>83</b>
B.1	Source code listing . . . . .	83

# Introduction

## 1.1 What is Mercury?

Mercury<sup>1</sup> is a *functional logic* programming language, which is a paradigm that combines features common in certain functional languages (e.g. higher-order constructs, strong advanced typing) and logic programming semantics (e.g. non-determinism, built-in search). It attempts to combine the clarity and expressiveness of declarative programming with advanced static analysis and error detection features. According to its creators, it's a “general purpose language intended to support the creation of large, reliable and efficient applications.” Although a short-lived renewed interest in this paradigm has been generated by Epic Games's programming language Verse<sup>2</sup> a few years earlier, it has now seemingly died down unfortunately; the paradigm remains interesting regardless, for the obvious reason that it provides extra edge against both purely functional and purely logic languages.

As are most of the languages that are of similar categories, Mercury has been mostly used for academic/research purposes, but a few commercial products have chosen (or once chosen) Mercury as their main language as well, like PrinceXML<sup>3</sup>, which is an HTML rendering tool used in publishing.

---

<sup>1</sup><https://mercurylang.org>

<sup>2</sup><https://dev.epicgames.com/documentation/en-us/fortnite/verse-language-reference>

<sup>3</sup><https://www.princexml.com>

## Other similar languages

A few languages have been said to be similar to Mercury on its different aspects:

- Curry<sup>4</sup>: When discussing functional logic programming, people naturally lean towards a conceptual Haskell-Prolog continuum; if one were to say Mercury is closer to the Prolog side, Curry is clearly closer to the Haskell side of things.
- Clean<sup>5</sup>: Clean is the lesser-known sibling of Haskell famed for its uniqueness typing. It was said that the ability to explicitly specify uniqueness is not unlike that of Mercury.

## 1.2 The state of things as of now

Mercury has been considered a difficult language to learn, even for people who has previous functional language experiences. While Mercury do have unfamiliar aspects from the viewpoint of programmers who have only written in “conventional” languages like Scala or Haskell, I personally find those aspects aren’t as difficult as one may think; I believe, that to a certain point this “difficult” feel comes from other factors, like the lack of a visible software ecosystem:

- Lack of example, tutorials and other introductory text: To be honest, it’s not that we don’t *have* those, it’s simply that most of them tend to be slightly too confusing for common programmers who are used to conventional imperative/functional languages. As of the time of writing (October 2025), we have the following attempts:
  - The tutorial written by Ralph Becket which is listed on Mercury’s official website<sup>6</sup>. Unfortunately this tutorial was left unfinished years ago and its author was said to have since stopped doing Mercury-related things.
  - The tutorials on Mercury’s GitHub repository wiki<sup>7</sup>, which seemingly has ceased major activities since 2018.

---

<sup>4</sup><https://curry-lang.org>

<sup>5</sup><https://clean-lang.org>

<sup>6</sup><https://mercurylang.org/documentation/learning.html>

<sup>7</sup><https://github.com/Mercury-Language/mercury/wiki/Tutorial>



- “Learn Mercury By Y Minute”<sup>8</sup>, which didn’t do a very good job at explaining things to a depth a learner would need their materials to have.
  - “Mercury Crash Course” written by J Fondren, on the now-defunct `mercury-in.space` website<sup>9</sup>, which honestly was too confusing for newcomers to provide any help.
  - A book titled “Declarative Programming in Mercury” is also available by compiling from the TeX source yourself using the source code<sup>10</sup>, but this one really isn’t an introductory text...
- Lack of package manager: Same as above, it’s not like we *don’t* have one - in fact we have two, one called Merchant by Stewart Slocum<sup>11</sup>, and one called `mmc-get`<sup>12</sup> by J Fondren, which used to have a handful of packages<sup>13</sup>. They are not compatible with each other (of course), which I doubt have ever created any kind of problem since nearly no one used them.
  - Lack of libraries: This is the real pain point. If a language did not have enough libraries, people are much less likely to use the language to make things; and if not enough people are using the language, there wouldn’t be enough demand and thus there wouldn’t be enough libraries. It’s a vicious cycle that requires a *lot* of effort to break.<sup>14</sup>

But all of these situations won’t change if we simply do nothing and hope for that one dumbass to put their love and effort for little to none rewards, which is why I have decided to become that dumbass myself and write this book to bridge the gap somewhat.

---

<sup>8</sup><https://learnxinyminutes.com/mercury/>

<sup>9</sup>You can still read it on Wayback Machine: <https://web.archive.org/web/20220815044519/https://mercury-in.space/crash.html>

<sup>10</sup>Available on GitHub: [https://github.com/Mercury-Language/books/tree/master/decl\\_prog](https://github.com/Mercury-Language/books/tree/master/decl_prog)

<sup>11</sup><https://github.com/stewy33/merchant>

<sup>12</sup><https://github.com/jrfondren/mmc-get>

<sup>13</sup><https://web.archive.org/web/20221130042103/https://mercury-in.space/packages/packages.list>

<sup>14</sup>However, a list of available resources (managed by Volker Wisk) can be found at <https://volker-wysk.de/mercury/resources.html>; although they might not necessary be the ones you want.

### 1.3 Who’s this book for

This book is for programmers who have previous experiences with imperative/functional languages who knows how programming works in principle and are used to filling in the details which could’ve been thought as incomplete or even missing. While learning Mercury as your first programming language might be a fun experiment to run on yourself, as the state of things being they are right now, it’s just a little too much to be truly beginner-friendly, so I can’t say I recommend it. Maybe years later when we have as much things as Haskell, this option would be viable again.

### 1.4 The tool we will be using

As of now (October 2025) there is only one implementation of Mercury: the Melbourne Mercury Compiler. All code in this book has been tested against version 22.01.8. Installing the compiler is probably one of the most painful part: since there is no official prebuilt binaries, you will have to download the source code, compile it and install it yourself. If you’re using an operating system with a package manager (like most Linuxes), their package repositories might have packages that ease the complexity of managing a little bit, but the package might simply be a automated build instructions without any pre-built binaries and your machine still needs to build the compiler, which takes a very long time.

Strangely, for the Melbourne Mercury Compiler, I actually recommend you to build from the source code instead of relying on the package repository available: MMC contains different backends and configurations called “grades” and is actually capable of generating code for Java and C Sharp:

```
-s grade  
--grade grade
```

Select the compilation model. This model, which Mercury calls a grade, specifies what properties the resulting executable or library should have. Properties such as ‘**generates profiling data for mprof when executed**’ and ‘**can be debugged with the Mercury debugger**’. As such, it controls decisions that must be made the same way in all the modules of a program.

— *From the man page of mmc*

Certain things also requires a separate “grade” being available (e.g. weirdly, single-precision floating point numbers; see the section about basic

types in Chapter 3). The package in the package repository of your operating system might not be configured with the grades you're interested in enabled.

## 1.5 More importantly... *why?*

A standard answer to this question would be “why not?”<sup>15</sup>, but I believe that people who would ask would like a different answer. The biggest appeal of Mercury, I think, is probably its ability to provide the power of both functional and logic programming in a rigorous way that can support developments of large-scale projects, much like having a static type system has been proven to be helpful (at least to certain degree) in this regard – before this, most logic programming languages either don't have the static checks Mercury do or were only really suitable for academic purposes. You may argue that you don't need all those and can still do everything with conventional languages like C; I will give you the win since I have no arguments against that, but can you truly convince yourself that that's a good argument?

---

<sup>15</sup>But seriously, if you've got the time, *why not?*



# A First Taste of Mercury

Unfortunately, due to Mercury currently not having a read-eval-print-loop, it's nigh impossible to properly introduce the basic part of Mercury in details without introducing some kind of scaffolding first. For this reason, in this chapter we'll be forcing our way into writing and completing a tiny utility program; after this, we will have a basic idea of how to write a bare-bone command line program in Mercury, which we can then use as a starting point for exploring the language.

## 2.1 Prerequisite: logic programming

While it might not seem obvious from an outsider's view, logic programming did in fact come from first-order predicate logic; more specifically, it comes from a specific interpretation of a specific class of first-order predicate logic formula. This class of formula, called **Horn Clause** after the logician Alfred Horn, can be one of the following forms:

- “Rule”, which takes the form of  $Q \leftarrow P_0 \wedge P_1 \wedge \dots \wedge P_n$ . We will read this as “ $Q$  holds when all of  $P_0$  and  $P_1$  and the rest – all the way to  $P_n$  – holds”
- “Fact”, which takes the form of  $P \leftarrow \top$ . We will read this as “ $P$  holds (regardless)”. Sometimes it's also shortened further into  $P$ . in some literature.
- “Goal”, which takes the form of  $\perp \leftarrow P_0 \wedge P_1 \wedge \dots \wedge P_n$ . We will read this as “Show all of  $P_0$  and  $P_1$  and the rest holds”.

Within these formula, all the  $Q$  and  $P_n$  have the form of  $f(arg_0, arg_1, \dots, arg_n)$  where  $f$  is simply a name and all the  $arg_n$  are themselves first-order predicate logic terms. If how this has anything to do with writing programs isn't clear to you, it probably will after reading the definition of the factorial function, written in Horn clauses:

---


$$\begin{aligned} fac(0, 1) &\leftarrow \top \\ fac(1, 1) &\leftarrow \top \\ fac(N, M) &\leftarrow minus(N, 1, N_0) \wedge fac(N_0, M_0) \wedge mult(M_0, N, M) \end{aligned}$$


---

Compare this with the following:

---


$$\begin{aligned} fac(0) &= 1 \\ fac(1) &= 1 \\ fac(N) &= M \textbf{ where } N_0 = N - 1, M_0 = fac(N_0), M = N \times M_0 \end{aligned}$$


---

...and the following code in Python:

```

1 def fac(n):
2     if n == 0: return 0
3     elif n == 1: return 1
4     else:
5         n0 = n - 1
6         m0 = fac(n0)
7         return n * m0

```

The similarity should be obvious at this point; in a very inaccurate sense, you can think of it as:

- Rules as function definition;
- Predicates as function calls;
- Conjunction as sequencing;
- Facts as base cases;
- Goals as function calls & entry points;

This, of course, is not all there is to it. We've covered the representation, we shall now cover the reduction.

## Unification

One thing to keep in mind is that theoretically there is no “computation” (as in calculation) involved; what we have (mostly) is a thing we call *unification*.<sup>1</sup> Unification roughly means to check if two terms “match”.

All terms are of one of the following forms:

- Variables, which we will denote with words that starts with an uppercase letter;
- Atoms, which we will denote with words that starts with a lowercase letter;
  - (This name, despite being the commonly used name for languages like LISP or Erlang, might come as obscure for other people; a more descriptive name would probably be “scalar values”.)
- Functors<sup>2</sup>, which is of the form  $f(a_0, \dots, a_n)$ , in which  $f$  is an atom and all the  $a$  are terms (which can themselves be variables or atoms or functors, etc.)

The unification of two terms thus can be said as follows:

- A variable matches with anything (unless it forms recursive reference; see below);
- Atoms only match with themselves, e.g. the atom *apple* would only match with *apple* and will not match with *pear*. (But obviously it would still match with variables, since variables match with anything)
- Functors match if and only if all the following conditions are met:
  - The “head” (or “the functor’s name”) matches, e.g.  $f(A, B)$  matches with  $f(a, b)$  but not  $g(a, b)$  because the latter is  $g$  instead of  $f$ .

---

<sup>1</sup>To some people this (and resolution, which we will discuss later) looks like a rewriting system, and some people do claim that Prolog is just a rewriting system with backtracking. As someone who’ve worked with rewriting systems before, I’m not going to comment on their opinions, but they do be looking similar.

<sup>2</sup>The name “functor” has been used to refer to a handful of different concepts; in the context of logic programming, especially languages that’s derived from Prolog, you can simply think of it as “some data that’s shaped like a function call”.

- The arity (i.e. the number of arguments) matches, e.g.  $f(A, B)$  matches  $f(a, b)$  but not  $f(c, d, e)$  despite both being  $f$ ; the former has 2 arguments while the latter has 3. The convention is to denote the arity after the functor’s name, e.g.  $f/2$  and  $f/3$ .
- All of the arguments match with each other, e.g.  $f(a, B)$  matches with  $f(a, b)$  and  $f(a, c)$  but not with  $f(b, a)$  despite both being  $f/2$ ; the first arguments, namely  $a$  and  $b$ , does not match.

The action of unification results in a mapping from variables to terms; (one could say that this is how “value assigning” happens, although that really isn’t the accurate wording.) A common name for this mapping would be “subst”, short for substitution. The observant might want to ask if one can match a variable with a term that contains the same variable either directly or indirectly (e.g.  $A$  with  $f(A)$ ). Theoretically this is not valid because one can never have a fully grounded solution for such a subst, but whether an occurrence of such case will trigger some kind of runtime panic depends on the language and its implementation. Some languages (like Prolog) permits it (even if this kind of situations almost certainly end in a failure state), some languages (like miniKanren) performs strict occur check and does not allow any such occurrences.

Some of you may ask if pattern matching can also be seen as a form of unification. I’ve met people online who were very pedantic about the name and fervently insisted that pattern matching is not unification; almost all of those kind of people are doing this because they want to be perceived as more sophisticated than they truly are so they can look down upon other people more safely. I would say one can think of it as somewhat similar; one must understand that pattern matching (and the subsequent binding of variables) is a “one way thing”, e.g. unifying  $f(a, J)$  with  $f(K, b)$  would result in a subst of  $J = b, K = a$ , but with pattern matching one side is always a fully grounded term, if we can even consider values as terms in this sense.

## Resolution

In this section I would only introduce one way of resolution. It is – I believe – conceptually the simplest kind of resolution. There probably are other kind of resolution technique out there, but they are out of scope for this tutorial.

Remember that a goal is of the form  $\perp \leftarrow P_0 \wedge \dots \wedge P_n$  (we’ll omit the  $\perp$  part for brevity from now on) and act as entry points in our interpretation;



naturally, if there's no more  $P$ , the goal is empty, and the computation halts.

The simplest method of resolution is thus as follows. For the goal  $\leftarrow P_0 \wedge \dots \wedge P_n$ , we take the left-most subterm, in this case  $P_0$ ; we attempt to find among all the previously defined rules and facts the one that can successfully unify with  $P_0$  (instantiated according to the current subst). If we couldn't find one, then the current attempt is considered a failure (what happens after the failure depends; see later section on cut); If we were successful in finding one, then two things happen:

- Because we performed a unification, we now have a subst;
- And because we picked a rule/fact, we now have a “right-hand side”, which could be empty or containing other goals.

This subst resulted from the unification is combined with all the subst we've had so far, and the “right-hand side” of the rule/fact we found would be appended to the left-hand side of the current goal.

Let's look at an example. Assume we have the following program in Horn clauses:

---

$parent(tom, john) \leftarrow \top$   
 $parent(tom, evans) \leftarrow \top$   
 $parent(tom, sarah) \leftarrow \top$   
 $sibling(A, B) \leftarrow parent(C, A) \wedge parent(C, B) \wedge not(equal(A, B))$

---

...and the goal we're asking is  $\leftarrow sibling(john, A)$ ; the following things happen:

- $sibling(john, A)$  matches with  $sibling(A, B)$ , (but we cannot directly say that  $john = A, A = B$ , because the two  $A$  should clearly mean different things! with a bit of renaming, we shall say the goal is  $sibling(john, A_0)$ ).
- The unification of the step above results in the subst  $john = A, A_0 = B$ .
- It contains a right-hand side of positive length; we shall add them to the goal with respect of the subst. The goal is now  $\leftarrow parent(C, john) \wedge parent(C, A_0) \wedge not(equal(john, A_0))$ .

- We now try to find a rule/fact that can unify  $parent(C, john)$ , and we found  $parent(tom, john)$ . This gives us  $C = tom$ . Since it's a fact, there is no right-hand side to be added. The goal is now  $\leftarrow parent(tom, A_0) \wedge not(equal(john, A_0))$ .
- We now try to find a rule/fact that can unify  $parent(tom, A_0)$ , and we again found  $parent(tom, john)$ . This gives us  $A_0 = john$ .
- The goal, which is now  $\leftarrow not(equal(john, john))$  with respect to the current subst, fails. If we have committed to this choice (by cut or by other means), the entire thing fails; but we didn't, thus we backtrack. The goal is now once again  $\leftarrow parent(tom, A_0) \wedge not(equal(john, A_0))$ , and we attempt to find the next thing that can unify with  $parent(tom, A_0)$ .
- We found  $parent(tom, evans)$ , which results in the subst  $A_0 = evans$ .
- The goal is now  $\leftarrow not(equal(john, evans))$ , which would succeed without a subst (or with an empty subst, depending on how you see it), leaving the goal empty. Since the goal is now empty, we consider whatever we have now is a valid solution. The subst for this solution would be  $A = john, C = tom, A_0 = evans$ , and the solution itself would be  $sibling(john, evans)$ .
- If more result is requested, the latest unification and its subst would be discarded and a new instance of  $parent(tom, A_0)$  would be looked for, which would result in the subst  $A_0 = sarah$  and the solution  $sibling(john, sarah)$ .
- It's easy to see that we can no longer find any new solutions after this point, so the resolution for the goal ends here.

## Cut

Cut is mainly a Prolog thing; when triggered, it forces the resolution to commit to all the choices it has made up till that point, artificially cutting the backtracking tree. This is one of the major reasons why Prolog is being considered as “not so pure”. Similar constructs exists in other logic programming languages (e.g. Mercury's `cc_multi` and `cc_nondet`), albeit they may work in a fundamentally different manner.

Using our program above as an example; if we write the first fact as  $parent(tom, john) \leftarrow !$  instead of  $parent(tom, john) \leftarrow \top$  (the exclamation mark is the “cut operator” in Prolog), we wouldn't have any solutions, since

by the time we reached the goal  $\leftarrow \text{not}(\text{equal}(\text{john}, \text{john}))$ , we have already committed to the choices of *john* and discarded other possibilities.

## Disjunction

Other than conjunction which can be interpreted as sequencing, one can also use disjunction in Horn clauses, which can be interpreted as branching. The resolution of a disjunct goal naturally depends on the resolution of its subgoals, and any solution is a solution of the overall disjunct goal if it is also a solution of any of its subgoals. The resolution of a disjunct goal starts from one of its subgoals; if that subgoal fails, it moves onto the next subgoal, until any one of these subgoals succeed (which results in an overall success) or all of them failed (which results in an overall failure). The order in which each branch gets resolved theoretically shouldn't matter, but in reality different order would lead to different backtracking tree, which in languages with cut (e.g. Prolog) and similar constructs could lead to order-dependent results.

## Unification, part 2

In short, due to how unification works, you can do certain thing “in reverse”. For example, assume that we have a predicate **append/3** which, at surface, is something you append two lists together:

---


$$\begin{aligned} \text{append}(\text{nil}, \text{nil}, \text{nil}) &\leftarrow \top \\ \text{append}(\text{nil}, A, A) &\leftarrow \top \\ \text{append}(\text{cons}(H, T), A, \text{cons}(H, R)) &\leftarrow \text{append}(T, A, R) \end{aligned}$$


---

If you query goals where the first two arguments are “solid” terms, you will get the result bound to the variable at the third argument; but at the same time, you can:

- Call the predicate with three “solid” lists, with which you can check if appending the first two lists does end up with the same value as the third one. If that *is* the case, then you’ll eventually hit one of the first two facts and have an overall success; if that *isn’t* the case, you’ll have a failure.
- Call the predicate with two variables and one “solid” list, with which you have a goal that asks “which two lists can be appended and end up with the same value as this third list”.

For the second one, we'll discuss the case *append*(*A*, *B*, *cons*(*c*, *nil*)) here; you can try to resolve a few examples on your own as an exercise. In the case of *append*(*A*, *B*, *cons*(*c*, *nil*)), the resolution would first match with the second fact (which will give *A* = *nil*, *B* = *cons*(*c*, *nil*)) and then it would match with the third rule (which, if you work it out, will give *A* = *cons*(*c*, *nil*), *B* = *nil*.)

## 2.2 Writing the program

In this part of the chapter, we will try to write a program which I called **rall**, which stands for “adding carriage return to **all** lines”. From this “full name” you can tell it is a very small utility that converts Unix-style newlines (LF) to Windows-style newlines (CRLF). It covers some of the most basic things when it comes to software development, and I reckon it would serve as a nice beginner (in terms of language, not programming in general) project.

### File name / module name

Each Mercury source file would be a module, whose name should be the same as the file name (without the extension name part).<sup>3</sup> Each module would need to have at least one **interface** section and one **implementation** section.

For our program, we would save the file under the name **rall.m**, thus the module name would be **rall**.

```

1  :- module rall.
2  :- interface.
3  :- implementation.
```

### Entry point

With Mercury, the entry point of the program (e.g. **main** in C) must be defined as a **main/2**, i.e. **main** with 2 parameters. If you attempt to define **main** as anything else, you would get a compile error:<sup>4</sup>

<sup>3</sup>It's possible to have the module name *not* to be the same as the file name, but it's too early to talk about it here. See Chapter 3.

<sup>4</sup>As you may have noticed, this is technically a *linking error* instead of a *compile error*, which means that the compiling *itself* did not generate any error. **main** with other arities does not get special treatment like **main/2** do, and some people do adopt the pattern of having both a **main/2** and a different **main** with a different arity.

```

1 % mmc ./rall.m
2 /usr/bin/ld: rall_init.o: in function `mercury_init':
3 rall_init.c:(.text+0x4a0): undefined reference to `<predicate '
  main'/2 mode 0>'
4 collect2: error: ld returned 1 exit status

```

The two parameters of this `main/2` must be of type `io.state`. If you don't define it as `io.state`, you would get a compile error:

```

1 ./rall.m:005: Error: `main'/2 must have mode `(di,uo)'.
2 ./rall.m:005: Error: both arguments of `main/2' must have type `io
  .state'.

```

To be able to use `io.state`, we should first import the `io` module, which is done with the `import_module` declaration.

So we now have the following:

```

1 :- interface.
2 % NOTE: to use io.state we must import the io module.
3 :- import_module io.
4 :- pred main(io.state, io.state).

```

You might wonder what exactly is `io.state` and why `main` needs two of them; we will explain this later. At the very least, these are **not** the same as `argc` and `argv` which you would see in C. There are dedicated predicates for retrieving command line arguments; we will get to that later.

The `(di,uo)` part needs explanation. The message says it's a **mode**. What is a mode? Modes are things that describe the change of instantiatedness before and after the execution of a goal. Instantiatedness – that's a long word. What is instantiatedness? Roughly speaking, it's a concept for describing whether a slot is free or bound. (In Mercury, instantiatedness is actually a tree, because obviously you'd have – or at the very least you could manually construct – terms that are only bounded at certain parts.) Two kinds of basic instantiatedness exists in Mercury: **free**, which refers to variables that are not bound by any values; **bound**, which means that the term is not a variable but rather some concrete term at least at that level. Consider the term `blah(A, b)`; it's **bound** at the top (with `blah`) and at one of the children node (at `b`) but **free** at another children node (at `A`, provided that `A` does not have a value already). A mode, **ground**, also exists, and refers to the terms that are *completely* bound (e.g. `blah(a, b)` is **ground** but `blah(A, b)` isn't, provided that `A` is still **free**.)

With this knowledge, we should first understand what `in` and `out` modes actually are. If you have worked with some other languages you might have seen thing similar to this before: C libraries often ask you to pass in a reference for retrieving the actual result because the return value is used for returning error status; you can think of it as “in” (inputting) parameters

and “out” (returning) parameters; some languages (e.g. Ada) even explicitly label them as such. I do not wish to introduce you to wrong analogies that will become detrimental for your future learning, but I have to say they do be somewhat similar.

That said, `in` and `out` is properly defined in Mercury, as follows:

```
1 :- mode in == ground >> ground.
2 :- mode out == free >> ground.
```

This largely fits with our intuition about `in` and `out` parameters: we expect `in` arguments to be fully grounded and they stay grounded, and we expect `out` arguments are variables which we will provide a solid term to (thus making it grounded).

There are also “polymorphic” version of `in` and `out`, defined as follows:

```
1 :- mode in(Inst) == Inst >> Inst.
2 :- mode out(Inst) == free >> Inst.
```

You can indeed define your own instantiatedness, with which then you can use these definitions to get the `in` and `out` version of it; but we will not need this for our example this time.

Now we can finally come back to `di` and `uo`. `di` stands for “destructive input”, and `uo` stands for “unique output”. If you have used Rust (or, in the case where you are really adventurous, Clean) before, you might have a vague idea of what this is. In Mercury, there are two special instantiatedness named `unique` and `dead`, the former conceptually refers to values that can only have one reference at all time, and the latter refers to reference that are once “unique” but is now “dead”.

```
1 % unique output - used to create a "unique" value
2 :- mode uo == free >> unique.
3
4 % unique input - used to inspect a unique value without causing
5 % reference to become dead
6 :- mode ui == unique >> unique.
7
8 % destructive input - used to deallocate or reuse the memory
9 % occupied by a value that will not be used.
10 :- mode di == unique >> dead.
```

Up to now, our code would be something that’s similar to this:

```
1 :- module mercury_rall.
2
3 :- interface.
4
5 :- import_module io.
6 :- pred main(io.state, io.state).
7
8 :- implementation.
9
```

```

10  main(_, _) :-
11      % some dummy body for our main predicate.
12      1 = 1.

```

If you compile this, the Mercury compiler would complain about not having a mode declaration. For this reason, we will add the following line and compile:

```

1  :- mode main(di, uo).

```

But this time we would see the compiler complain about different things:

```

1  ./rall.m:007: Error: no determinism declaration for exported
    predicate
2  ./rall.m:007:      `main'/2.
3  ./rall.m:012: In clause for `main(di, uo)':
4  ./rall.m:012:      mode error: argument 2 did not get sufficiently
5  ./rall.m:012:      instantiated.
6  ./rall.m:012:      Final instantiatedness of `HeadVar__2' was `free',
7  ./rall.m:012:      expected final instantiatedness was `unique'.

```

The compiler expects the second argument would be supplied with a `free` argument and that argument should become a `unique` value at the end of `main`! How do we do such a thing? Fortunately the `io` module has what we need to stuff the body of `main`, and we will use one of them:

```

1  main(In, Out) :-
2      io.write_string("blah", In, Out).

```

But this time the compiler started complaining about other things:

```

1  ./rall.m:007: Error: no determinism declaration for exported
    predicate
2  ./rall.m:007:      `main'/2.

```

Determinism in this case, informally speaking, refers to “how a certain thing would succeed/fail”. It’s not something we’d care about in other languages, at least not in an explicit, supported-by-the-language-itself manner; (we normally only talk about when programs terminate at a certain state or not.) In Mercury, we have the following determinism categories:

- Deterministic (**det**): guaranteed to have one and exactly one solution.
- Semideterministic (**semidet**): have exactly one solution, but does not guarantee to produce it.
- Multisolution (**multi**): guaranteed to have a solution among possibly many solutions.
- Nondeterministic (**nondet**): have possibly many solutions, does not guarantee to produce one.

- Failure (**failure**): cases where there's zero solutions. They are not actual errors but a part of the logic (e.g. arity mismatch during unification, which will never produce a solution because the arity is different).
- Errorneous (**errorneous**): also have zero solutions, but they **do** represent actual errors which in other languages would be represented in the form of runtime exception throw or panic.

Some people might not understand why **det** and **semidet** are separate things. Imagine a function that takes the "head" of a linked list; this function is obviously only defined on non-empty list and not defined on empty lists, so for any given list there's either only one solution or no solution, thus **semidet** instead of **det**.

The difference between **errorneous** and **failure** might be clearer if I explain their behaviours when it comes to negation. Basically in Mercury (and other logic programming languages), you can take a predicate, put a negation on it, and then ask for a case where it **does not succeed**. Naturally, the negation of determinisms that are guaranteed to produce a result like **det** and **multi** would be **failure** (i.e. negating a definite success would be a failure), the negation of **failure** would be **det** (i.e. negating a definite failure would be a definite success), and the negation of **semidet** and **nondet** would be **semidet** (i.e. negating these cases turns it into a case that asks if the match would be successful or not); but negating an **errorneous** would only produce an **errorneous**.

(If you're going to ask this question – yes, solving the problem of perfect determinism inference does mean solving the halting problem, and is thus undecidable.)

Other than these six categories there are also this thing called the "committed choice nondeterminism", which adds two more determinism class: **cc\_multi** and **cc\_nondet**. The difference between committed choice nondeterminism and "normal" nondeterminism is that normal nondeterminism supports backtracking while CC nondeterminism, despite potentially having more than one solutions, commits to only one of them and thus does not backtrack. The **main** entry point can be defined to be **det** or **cc\_multi**, since both of them guarantee one and only one solution (**det**, of course, is more strict than **cc\_multi**, and if the Mercury compiler can determine something that should be able to be a **det** got labelled as a **cc\_multi**, it would spit out a warning saying you could've gone with the stricter option.)

In this case, we should add the string **is det** at the end of our mode declaration, so that the whole line would be **:- mode main(di, uo) is**



det.. After this modification, the compiler should have finally stopped complaining and gives you an executable; when you run it, it would display a string `blah`, which should be obvious. At this point, the code shall look like something like this:

```

1  :- module rall.
2
3  :- interface.
4
5  :- import_module io.
6  :- pred main(io.state, io.state).
7  :- mode main(di, uo) is det.
8
9  % also: you can combine the `pred` and the `mode` line into one
10 % like this:
11 %      :- pred main(io.state::di, io.state::uo) is det.
12
13 :- implementation.
14
15 main(In, Out) :-
16     io.write_string("blah", In, Out).
```

Now, if you attempt to write multiple strings like this:

```

1  main(In, Out) :-
2      io.write_string("blah", In, Out),
3      io.write_string("blah", In, Out).
```

The compiler would produce this error:

```

1  ./rall.m:013: In clause for `main(di, uo)':
2  ./rall.m:013:   in argument 2 of call to predicate `io.
3  write_string'/3:
4  ./rall.m:013:   unique-mode error: the called procedure would
5  clobber its
6  ./rall.m:013:   argument, but variable `In' is still live.
7  For more information, recompile with `-E'.
```

If we look up the declaration of `io.write_string`:

```

1  :- pred write_string(string::in, io::di, io::uo) is det.
```

You should already know that `di` specifies a turn of a `unique` value into a `dead` value, so `In` after the first `write_string` would be considered `dead` and not `unique`, which does not fit the requirement of the `di` of the second `write_string`. If you try to experiment and do this:

```

1  main(In, Out) :-
2      io.write_string("blah", In, Out),
3      io.write_string("blah", Out, Out2).
```

This would have the following violation: since we declared `Out` to be `uo` which is `free >> unique`, `Out` must be `unique` at the end of `main`, but the second `write_string` turns `Out` into a `dead` because that slot is `di` which is `unique >> dead`. A solution to this would be to do this:

```

1  main(In, Out) :-
2      io.write_string("blah", In, Out1),
3      io.write_string("blah", Out1, Out).

```

This would compile, and the generated executable would write "blah" twice, as expected.

To keep coming up with new variable name is tedious if this gets long. To solve this problem, Mercury have something called the "state variable". With state variables, the example above can be written like this:

```

1  main(!IO, !:IO) :-
2      io.write_string("blah", !IO, !:IO),
3      io.write_string("blah", !IO, !:IO).

```

`!IO` refers to the value bound to `IO` at the current moment, and `!:IO` makes the value at that slot bound to `IO`; so in the first `write_string`, the *current* value `IO` was destroyed, and the value bound to `Out1` in our non-state-variable version was bound to `IO` as its next value; and in the second `write_string`, the value of `Out1`, which itself has become the current value of `IO` and being referred to with `!IO`, was destroyed, and the value bound to `Out` in our non-state-variable version was bound to `IO` as its next value, and subsequently became the output of `main`.

You can write `!IO, !:IO` as `!IO`, because the former is also quite tedious as well:

```

1  main(!IO) :-
2      io.write_string("blah", !IO),
3      io.write_string("blah", !IO).

```

It would still feel a bit tedious for people who are used to other programming languages, but I suppose this is what you give up for being as explicit as possible for the sake of improved correctness and things...

## Command line arguments

The predicate we need for retrieving the command line arguments is also in the `io` module:

```

1  main(!IO) :-
2      io.command_line_arguments(Argv, !IO),
3      % ...

```

This would bound the actual value to `Argv`. From the definition of `command_line_arguments` we can know that the type of `Argv` would be a `list(string)` – a list of strings. It would be necessary to import the `list` module from now on. This value does not contain the name of the program

(for which `io` module has other predicates to retrieve). It would also be useful to check if the length of `Argv` is exactly 2:

```

1 :- import_module list.
2 % ...
3 main(!IO) :-
4     io.command_line_arguments(Argv, !IO),
5     ( if length(Argv) \= 2
6       then (
7           % you can specify which stream you are writing to btw
8           io.write_string(
9               io.stderr_stream,
10              "usage: rall [inputfilename] [outputfilename]\n",
11              !IO)
12         )
13     else (
14         % ...
15     )
16 ).

```

(Note that if you have previous experience with Prolog, the condition for the `if` construct might be surprising, since in Prolog, `=` unify things *structurally*; for example, the goal `3 + 4 = 7` would fail in Prolog despite 3 plus 4 do equal to 7, since `3 + 4` is a functor (interpreted as ``+`(3,4)`) and 7 is an atom; this is different in Mercury, where things *would* be evaluated. Also, `length` in Mercury is defined both as a predicate and as a **function**, and in this case the function got evaluated as it's defined in the standard library resulting an actual number.

The `then` clause is simple – we write the usage string and call it done:

```

1 :- import_module list.
2 % ...
3 main(!IO) :-
4     io.command_line_arguments(Argv, !IO),
5     ( if length(Argv) \= 2
6       then (
7           % you can specify which stream you are writing to btw
8           io.write_string(
9               io.stderr_stream,
10              "usage: rall [inputfilename] [outputfilename]\n",
11              !IO)
12         )
13     else (
14         % ...
15     )
16 ).

```

Now we extract the command line arguments. `list` module contains `index0` and `index1`, which are predicates/functions that uses 0-based index and 1-based index respectively; but these aren't the ones we would choose to use here because they are `semidet`. (Exercise: can you figure out the reason why they're `semidet` instead of `det` on your own?) We use their

det counterparts: `det_index0` and `det_index1`, which will throw runtime exceptions when failing:

```

1 :- import_module list.
2 % ...
3 main(!IO) :-
4     io.command_line_arguments(Argv, !IO),
5     ( if length(Argv) \= 2
6         then (
7             io.write_string(
8                 io.stderr_stream,
9                 "usage: rall [inputfilename] [outputfilename]\n",
10                !IO)
11         )
12     else (
13         InputFilePath = det_index0(Argv, 0),
14         OutputFilePath = det_index0(Argv, 1),
15         rall(InputFilePath, OutputFilePath, !IO)
16         % ...
17     )
18 ).

```

By the way, if we import module simply by doing `:- import_module name`, We don't need to fully specify the module name, as long as there's no ambiguity. Naturally, the good practice for a lot of common cases is to fully specify; but some editors (Emacs, for example) might implement some kind of "smart indent" algorithm which could put the cursor at undesirable positions and be a nuisance. In cases like this, you might find writing code this way to be more comfortable.

It would be also be a good time to write our first predicate that is not the entry point...

## Opening & closing files for input/output

In the last code snippet we've write `rall(InputFilePath, OutputFilePath, !IO)`. Despite this looks like `rall/3`, but it's secretly `rall/4`, and we should write our declaration as such:

```

1 :- pred rall(string, string, io.state, io.state).
2 :- mode rall(in, in, di, uo) is det.

```

`io` module has `open_input/4` and `open_output/4`, both of which are `det`:

```

1 rall(InputFilePath, OutputFilePath, !IO) :-
2     io.open_input(InputFilePath, RIn, !IO),
3     % ...

```

The declaration of `open_input` is as follows:

```

1 :- pred open_input(
2     string::in, io.res(io.text_input_stream)::out,

```

```
3   io::di, io::uo) is det.
```

We know that whatever variable we put at the second slot would be the input stream we want (or in the case of error, whatever error we could catch). Its value is of type `io.res`, whose definition is as follows:

```
1   :- type res(T)
2   --->   ok(T)
3   ;      error(io.error).
```

This is how Mercury defines algebraic datatypes. If you have previous experience with Haskell, OCaml, Rust or similar languages, this should be natural for you.

Pattern matching is quite simple – you just do `Var = ok(OkValue)` and `Var = error(ErrorValue)` like this:

```
1   % ...
2   ( Var = ok(OkValue),
3   % `ok` clause
4   );
5   ( Var = error(ErrorValue),
6   % `error` clause
7   )
```

Since we combined them with a disjunction, when the unification in the first branch (in this case, `Var = ok(OkValue)`) fails, the program flow would go to the second branch, and if the unification there fails (and there's a third branch), it would go to the third clause, etc.; and if all branches fail, the whole thing fails (and of course we'd expect the Mercury compiler gives us a determinism verdict of something like `nondet`, but I digress). We'll handle all the cases a value of type `res(T)` could have, and if we do that, the compiler should recognize that we'll not fail as long as both of our `ok` clause and `error` clause don't have a determinism verdict that implies a possibility of failing.

The definition of `io.error` is as follows:

```
1   :- type io.error.    % Use error_message to decode it.
```

This type is opaque; we need `error_message` to handle it. Its definition is as follows:

```
1   % Look up the error message corresponding to a particular error
   % code.
2   :- func error_message(io.error) = string.
3   :- pred error_message(io.error::in, string::out) is det.
```

Now we know we can obtain a `string` from an `io.error`; we can decide either to `throw` (which would cause the program to exit prematurely) or to write it to standard error (and try to exit afterwards). In this example we use `throw` because it's simpler:

```

1  % throw/1 requires the exception module.
2  :- import_module exception.
3
4  % ...
5  ( RIn = error(ErrorValue), throw(error_message(ErrorValue)) );

```

Before we move on with it, I must also explain something else. This isn't the case in Prolog, but in Mercury calls are "curried". Using the declaration of `error_message` above as an example, the term `error_message(Var)` matches both `func error_message/1` and `pred error_message/2`, and would thus be judged as having the type `string` from `func error_message/1` and the type `pred(string)` from `pred error_message/2` (which is a `pred` that takes a `string` as its argument) *at the same time*, i.e. a type ambiguity. In Prolog it will only match `error_message/2`, but in Mercury this is how it is...

To resolve this you need to explicitly state the type you want by writing `{type};` in this case, we'll choose the `func` one and specify it as `string`, just like this:

```

1  %                                     here, the ":string" part. ---v
2  ( RIn = error(ErrorValue), throw(error_message(ErrorValue):string)
   );

```

If you don't do this, you'll get a compile error like this:

```

1  ./rall.m:025: In clause for predicate main'/2:
2  ./rall.m:025:   warning: variable OutputFilePath' occurs only once
   in
3  ./rall.m:025:   this scope.
4  ./rall.m:031: In clause for predicate rall'/3:
5  ./rall.m:031:   error: ambiguous overloading causes type ambiguity
   .
6  ./rall.m:031:   Possible type assignments include:
7  ./rall.m:031:   V_20:
8  ./rall.m:031:     pred(
9  ./rall.m:031:       string
10 ./rall.m:031:     )'
11 ./rall.m:031:   or
12 ./rall.m:031:     string'
13 ./rall.m:031:   You will need to add an explicit type
   qualification to
14 ./rall.m:031:   resolve the type ambiguity. The way to add an
   explicit
15 ./rall.m:031:   type qualification is to use "with_type". For
   details see
16 ./rall.m:031:   the "Explicit type qualification" sub-section of
   the
17 ./rall.m:031:   "Data-terms" section of the "Syntax" chapter of
   the
18 ./rall.m:031:   Mercury language reference manual.

```

Now we can have the following code. Opening an output stream is a near-identical process:

```

1 % throw/1 requires module `exception`.
2 :- import_module exception.
3 % ...
4
5 :- pred rall(string, string, io.state, io.state).
6 :- mode rall(in, in, di, uo) is det.
7 rall(InputFilePath, OutputFilePath, !IO) :-
8     open_input(InputFilePath, RIn, !IO),
9     ( ( RIn = error(ErrorValue), throw(error_message(ErrorValue):
10         string) );
11         ( RIn = ok(InputStream),
12             open_output(OutputFilePath, ROut, !IO),
13             ( ( ROut = error(ErrorValue), throw(error_message(
14                 ErrorValue):string) );
15                 ( ROut = ok(OutputStream),
16                     process_stream(InputStream, OutputStream, !IO)
17                 )
18             ),
19             close_output(OutputStream, !IO),
20             close_input(InputStream, !IO)
21         ).

```

## Actually processing the data

We'll do it as follows: each time we read a single character from the input stream, we check if it's a line feed; if it is, we write a carriage return *and* a line feed to the output stream; or else, we simply write that character to the output stream; we repeat this until we've hit the EOF of the input stream. `read_char` produces a different kind of result named `result`, which has a separate constructor `eof`. You should know how to handle this by now...

Character literals are enclosed with single quotes, by the way:

```

1 :- pred process_stream(
2     io.text_input_stream,
3     io.text_output_stream,
4     io.state, io.state).
5 :- mode process_stream(in, in, di, uo) is det.
6 process_stream(InStream, OutStream, !IO) :-
7     read_char(InStream, InRes, !IO),
8     ( ( InRes = eof );
9         ( InRes = error(Error), throw(error_message(Error):string) );
10         ( InRes = ok(Char),
11             ( if Char = '\n'
12                 then (
13                     write_char(OutStream, '\r', !IO),
14                     write_char(OutStream, '\n', !IO)
15                 )
16                 else (
17                     write_char(OutStream, Char, !IO)
18                 )
19             ),

```

```

20     process_stream(InStream, OutStream, !IO)
21   )
22   ).
23

```

## Full program listing

The following code is the listing of the completed `rall` program:

```

1  :- module rall.
2  :- interface.
3  :- import_module io.
4
5  :- pred main(io.state, io.state).
6  :- mode main(di, uo) is det.
7
8  :- implementation.
9
10 :- import_module list, exception.
11
12 main(!IO) :-
13     io.command_line_arguments(Argv, !IO),
14     ( if length(Argv) \= 2
15       then (
16           io.write_string(
17               io.stderr_stream,
18               "usage: rall [inputfilename] [outputfilename]\n",
19               !IO)
20         )
21       else (
22           InputFilePath = det_index0(Argv, 0),
23           OutputFilePath = det_index0(Argv, 1),
24           rall(InputFilePath, OutputFilePath, !IO)
25         )
26     ).
27
28 :- pred rall(string, string, io.state, io.state).
29 :- mode rall(in, in, di, uo) is det.
30 rall(InputFilePath, OutputFilePath, !IO) :-
31     open_input(InputFilePath, RIn, !IO),
32     ( ( RIn = error(ErrorValue), throw(error_message(ErrorValue):
33         string) );
34       ( RIn = ok(InputStream),
35         open_output(OutputFilePath, ROut, !IO),
36         ( ( ROut = error(ErrorValue), throw(error_message(
37             ErrorValue):string) );
38           ( ROut = ok(OutputStream),
39             process_stream(InputStream, OutputStream, !IO)
40           )
41         ),
42         close_output(OutputStream, !IO),
43         close_input(InputStream, !IO)
44     )
45     ).
46 :- pred process_stream(

```



```

46     io.text_input_stream,
47     io.text_output_stream,
48     io.state, io.state).
49 :- mode process_stream(in, in, di, uo) is det.
50 process_stream(InStream, OutStream, !IO) :-
51     read_char(InStream, InRes, !IO),
52     ( ( InRes = eof );
53       ( InRes = error(Error), throw(error_message(Error):string) );
54       ( InRes = ok(Char),
55         ( if Char = '\n'
56           then (
57             write_char(OutStream, '\r', !IO),
58             write_char(OutStream, '\n', !IO)
59           )
60         else (
61           write_char(OutStream, Char, !IO)
62         )
63       ),
64     process_stream(InStream, OutStream, !IO)
65   )
66 ).

```

Listing 2.1: Full code for `rall.m`

It might not be idiomatic Mercury code, but at least it compiles and runs.

We can compile it and produce an executable by calling the command `mmc rall.m`.



# Programming in Mercury

## 3.1 Trace Goals

Before we continue we must talk about how to explore things on one's own. A common strategy of debugging in conventional languages would be to insert printing statements at places where one would expect the program to reach with its state containing certain values; this strategy doesn't work well with Mercury, since to perform IO in Mercury you would need to do it through `io.state`, which naturally has the implication of spreading out and forcing everything that uses the predicate to adopt two extra `io.state` parameters.<sup>1</sup>

For example, assume we have the following program that prints out the factorial of 5:

```

1 :- module ex01.
2 :- interface.
3 :- import_module io.
4 :- pred main(io.state::di, io.state::uo) is det.
5 :- implementation.
6
7 :- import_module int.
8
9 :- pred fac(int, int).
10 :- mode fac(in, out) is det.
11 fac(N, M) :-
12   ( if ( N = 0 ) then ( M = 1 )
13     else if ( N = 1 ) then ( M = 1 )
14     else if ( N < 0 ) then ( M = 1 )
15     else ( fac(N-1, M1), M = N * M1 )
16   )

```

<sup>1</sup>Which is not unlike the famous “color of function” problem in languages with `async` (e.g. JavaScript)

```

17 .
18
19 main(!IO) :-
20     io.write_string("Hello, world!\n", !IO),
21     fac(5, M),
22     io.write_int(M, !IO).

```

Let's say we want to add a printing goal in `fac`. Printing, naturally, requires two `io.state` just like we've been mentioning. If we do it by adding `io.state`, we need to perform the following changes:

- Change the `pred` line into `:- pred fac(int, int, io.state, io.state);`
- Change the `mode` line in a similar fashion;
- Change every occurrence of `fac` accordingly;
- Since for every call site of `fac` we now need to pass two extra `io.state` arguments, the environment surrounding the call site should have those extra arguments ready;
  - and for some predicates this could mean adding extra parameters just like what we've done to `fac`;
  - and now the call sites of *those* predicates also need to have their signature changed, and this change would also affect call sites that calls *them*, and the need of change would thus propagate upwards...
- ...and after the debugging is done, we need to change them *back*.

This kind of coding process not only goes against the spirit of Mercury but is also just extremely tedious. Luckily, for this Mercury prepared a feature called **trace goals**.

Trace goals in Mercury are of the format `trace [param1, ...] Goal`. In such a goal, all variables bound in the parameter list part is available within the *Goal* part. There are many different parameters that one can use, but the most common is probably going to be `io`; by having `io(!IO)` which bound a pair of `io.state` to the state variable of your choice, you can use it to display debug messages in *Goal* freely without having to do the changes listed above. For example, assume we want our program to display the value of each call to `fac` before they return, we can add a trace goal that introduces a pair of `io.state` arguments like this:

```

1 :- module ex01.
2 :- interface.
3 :- import_module io.

```

```

4 :- pred main(io.state::di, io.state::uo) is det.
5 :- implementation.
6
7 :- import_module int.
8
9 :- pred fac(int, int).
10 :- mode fac(in, out) is det.
11 fac(N, M) :-
12     ( if ( N = 0 ) then ( M = 1 )
13       else if ( N = 1 ) then ( M = 1 )
14       else if ( N < 0 ) then ( M = 1 )
15       else ( fac(N-1, M1), M = N * M1 )
16     ),
17     trace [ io(!IO) ] (
18         io.write_string("fac(", !IO),
19         io.write_int(N, !IO),
20         io.write_string(") = ", !IO),
21         io.write_int(M, !IO),
22         io.write_string("\n", !IO)
23     )
24 .
25
26 main(!IO) :-
27     io.write_string("Hello, world!\n", !IO),
28     fac(5, M),
29     io.write_int(M, !IO).

```

Notice that without changing the signatures of `fac` we can still obtain `io.state` and display things.

## 3.2 Using the standard library

As you read through the documentation of Mercury’s standard libraries<sup>2</sup>, you’ll see the notice saying that “we recommend that if possible you use only those modules whose stability is described as either ‘medium to high’ or ‘high’”, you’ll see a `stability` declaration in each of the documentation page, and you’ll see that certain modules that *should* be stable, like `int` and `uint`, has a stability of `low`; should you rely on these modules that are of low stability? Sometimes the answer is yes – these modules have reached `high` stability in the latest nightly, which means that the interface of these modules, as it’s described in the documentation of latest nightly, shouldn’t change so much that it breaks basic programs in the future, and thus should be safe to use. In general, my recommendation is to read the

---

<sup>2</sup>Available at [https://mercurylang.org/information/doc-release/mercury\\_library/index.html](https://mercurylang.org/information/doc-release/mercury_library/index.html) and [https://mercurylang.org/information/doc-latest/mercury\\_library\\_manual/index.html](https://mercurylang.org/information/doc-latest/mercury_library_manual/index.html), for latest stable and latest nightly, respectively.

documentation for the latest nightly when you're not sure if you can still use the same library predicate when the next major release come out.

### 3.3 Basic Datatypes

Finally, after all that build-up, we can look at the basic datatypes.

#### Signed and unsigned integers

In Mercury, the types of signed integers are: `int`, `int8`, `int16`, `int32` and `int64`, and the types of unsigned integers are: `uint`, `uint8`, `uint16`, `uint32` and `uint64`. The types with a number suffix are exactly what you think they are, e.g. `int8` means an 8-bit signed integer, etc.. The length of `int` is said to be “implementation-defined” but at least 32 bits. If you want to know how long an `int` actually is, you can have the following program:

```

1  :- module ex01.
2
3  :- interface.
4
5  :- use_module io.
6  :- pred main(io.state::di, io.state::uo) is det.
7
8  :- implementation.
9  :- import_module int, string.
10
11
12 main(!IO) :-
13     int.bits_per_int(BPI),
14     string.int_to_string(BPI, BPIStr),
15     io.write_string(BPIStr, !IO),
16     io.write_string("\n", !IO).
```

As of now (October 2025), to enable the arithmetic and comparing operations of these types you must import the corresponding module, e.g. to be able to compare `uint` you must import the `uint` module. Actually, as a rule of thumb, if you need to do stuff with any of these types (and the basic types described in sections below), it's good measure to import the corresponding module even if you ended up not needing them.

#### Syntax of integers

A decimal integer is any sequence of decimal digits. Unlike in some languages, having the prefix `0` doesn't make it octal. In Mercury, the corresponding prefixes for non-decimal integers are: `0b` for binary, `0o` for octal, and `0x` for hexadecimal. Mercury also supports a prefix of `0'` followed by

any single character; this would result in the character code that character has. (Note that the character immediately after the single quote is considered part of the integer literal regardless of its common usage in other parts of Mercury code; e.g. to get the character code of the character `,`, you write `0'`, with no extra measure of escaping. This works for Unicode characters as well; see the example below.)

A suffix may also be present, with different suffixes and their meanings listed as follows:

Suffix	Type
i or no suffix	Signed ( <code>int</code> )
i8	Signed 8-bit ( <code>int8</code> )
i16	Signed 16-bit ( <code>int16</code> )
i32	Signed 32-bit ( <code>int32</code> )
i64	Signed 64-bit ( <code>int64</code> )
u	Unsigned ( <code>uint</code> )
u8	Unsigned 8-bit ( <code>uint8</code> )
u16	Unsigned 16-bit ( <code>uint16</code> )
u32	Unsigned 32-bit ( <code>uint32</code> )
u64	Unsigned 64-bit ( <code>uint64</code> )

An arbitrary number of underscore `_` can be inserted between the radix prefix and the first digit (e.g. `0x_123`), between the digits (e.g. `0b_0001_0011`), and between the last digit and the type suffix (e.g. `0b10010011_u8`. Inserting underscores do not affect the actual value.

An example program showcasing the syntax for integers is listed as follows.<sup>3</sup>

```

1 :- module ex01.
2
3 :- interface.
4
5 :- use_module io.
6 :- pred main(io.state::di, io.state::uo) is det.
7
8 :- implementation.
9 :- import_module string.
10 :- import_module int, int8, int16, int32, int64.
11 :- import_module uint, uint8, uint16, uint32, uint64.
12
13 main(!IO) :-
14     io.write_string(string.int_to_string(32i) ++ "\n", !IO),
15     io.write_string(string.int8_to_string(32i8) ++ "\n", !IO),
16     io.write_string(string.int16_to_string(32i16) ++ "\n", !IO),
17     io.write_string(string.int32_to_string(32i32) ++ "\n", !IO),
18     io.write_string(string.int64_to_string(32i64) ++ "\n", !IO),

```

<sup>3</sup>As you can see, sometimes some lexers cannot properly handle valid Mercury code...

```

19 io.write_string(string.uint_to_string(32u) ++ "\n", !IO),
20 io.write_string(string.uint8_to_string(32u8) ++ "\n", !IO),
21 io.write_string(string.uint16_to_string(32u16) ++ "\n", !IO),
22 io.write_string(string.uint32_to_string(32u32) ++ "\n", !IO),
23 io.write_string(string.uint64_to_string(32u64) ++ "\n", !IO),
24 io.write_string("\n", !IO),
25 io.write_string(string.uint_to_string(0x_123u) ++ "\n", !IO),
26 io.write_string(string.uint_to_string(0b_0001_0011u) ++ "\n", !IO)
27 ,
28 io.write_string(string.uint8_to_string(0b0001_0011_u8) ++ "\n", !
29 IO),
30 io.write_string(string.int_to_string(0'K) ++ "\n", !IO),
31 io.write_string(string.int_to_string(0') ++ "\n", !IO),
io.write_string(string.int_to_string(0',) ++ "\n", !IO),
io.write_string(string.int_to_string(0':) ++ "\n", !IO).

```

## Floating-point numbers

In Mercury, there is only one type of floating-point number: `float`. As of now (October 2025; MMC 22.01.8), the default size of this type is double-precision (i.e. 64 bits). Single-precision can be enabled by using the `.spf` grade suffix, which means to use the command line argument `--grade asm_fast.gc.spf` or `--grade hlc.gc.spf` or others while compiling, depending on which backend you want to use. `.spf` isn't always available; if your call to `mmc` results in an error like this one:

```

1 mercury_compile: error: the Mercury standard library cannot be found
  in grade
2   `asm_fast.gc.spf'.

```

It means that this particular configuration of backend isn't installed and is thus unavailable. This is likely to be the case, since the default configuration does not include `.spf` grades.

## Character and string

In Mercury, the character type is `char` and the string type is `string`.

### Syntax of string

Strings in Mercury are enclosed in double-quotes `"`. A few escape sequences are supported:



Escape sequence	Meaning	Example
\\	Backslash \	\\
\'	Single quote	\'
\"	Double quote	\"
\a	Beep	\a
\b	Backspace	\b
\r	Carriage return	\r
\f	Form feed	\f
\t	Tab	\t
\n	Newline	\n
\v	Vertical tab	\v
\xXX\	Hexadecimal escape	\x37\
\XXX\	Octal escape	\037\
\uXXXX	Unicode	\u3707
\UXXXXXXXX	Unicode	\U00003707

The length of a hexadecimal or an octal escape sequence is not pre-determined (viz. there can be 2, 3, 4 or potentially more hexadecimal/octal digits that comes after a `\x/\` prefix); but if the escaped value turned out to be invalid (e.g. a very big number that goes over the Unicode range), a compile error would occur.

The ending backslash for a hexadecimal/octal escape sequence is necessary. For example, this small program:

```

1  :- module ex01.
2
3  :- interface.
4
5  :- use_module io.
6  :- pred main(io.state::di, io.state::uo) is det.
7
8  :- implementation.
9
10 main(!IO) :-
11     io__write_string("\x4b\\\n", !IO),
12     io__write_string("\x4b\\\n", !IO).
```

...prints two lines of text: `K\` and `K\n`, with the second line *not* ending in a newline character. This is because the first string is being read as the combination of `\x4b\`, `\\` and `\n`, but the second string is being read as the combination of `\x4b\`, `\\` and `n`.

Other than `\"`, you can also use `"` to represent one single double-quote character within a string.

Multi-line string literals are also possible by having a single backslash at the end of the line; the backslash character and the newline character after it would be ignored. Unlike some languages, you need to insert newline

manually; for example, the first call of the following would only produce *one* single line of output:

```

1      % ...
2      io__write_string("a\
3      multi\
4          line\
5              string\
6                  literal\
7      ", !IO),
8      io__write_string("a\n\
9      multi\n\
10         line\n\
11             string\n\
12                 literal\n\
13      ", !IO),
14      % ...

```

### Syntax of character

The syntax of character is slightly trickier to describe. In many languages, characters are quoted with single-quote `'`; but per Prolog tradition, single-quotes are also for atoms (or *names*, as defined in Mercury's language manual), which may contain more than one character. Currently, it seems like that Mercury treats quoted names containing one single character (or escape sequence) as character literals under the right context.

One needs to be careful when using quoted atoms when they intend to write character literals: if a character is an operator, simply quoting it isn't enough, since Mercury considers single-quoted operators are still operators themselves. To prevent such cases from being interpreted as an operation of values, one should surround the quoted name with parentheses. For example, `Char = '+'` is syntatically invalid, since it's being interpreted as `Char = +`; one should write `Char = ('+')` in this case.

### Tuples

Mercury supports using N-tuple for any N larger than or equal to zero. In Mercury, tuples are enclosed with braces `{}` both for types and for values. See the example below:

```

1 :- module ex01.
2
3 :- interface.
4
5 :- use_module io.
6 :- pred main(io.state::di, io.state::uo) is det.
7
8 :- implementation.
9 :- import_module int, string.

```

```

10
11 :- pred f1({int, string}::in, {string, int}::out) is det.
12 f1({X, Y}, {Y, X}).
13
14 main(!IO) :-
15     % bind X to a tuple value.
16     X = {3, "abc"},
17
18     % get a reversed tuple w/ f1 (defined above).
19     f1(X, Reversed_X),
20
21     % break the reversed tuple down into different values.
22     {MyString, MyInteger} = Reversed_X,
23     io.write_string(string.int_to_string(MyInteger) ++ ", " ++
MyString ++ "\n", !IO).

```

## Maybe

**maybe** is a very common construct and occurs in many ML<sup>4</sup>-inspired languages under names like “Maybe”, “Option” or “Optional”. In Mercury, it’s defined as follows:

```

1 :- type maybe(T)
2     --->    no
3     ;      yes(T).

```

Essentially, it’s a type whose values can be either a wrapper that contains something (i.e. `yes(SomeValue)`) or an empty value that doesn’t contain anything (i.e. `no`). It’s often seen as a stricter (in terms of type-checking) version of having the null pointer, and some OOP languages have adopted this for this reason.

Many variants of Maybe exists for slightly different purposes. The most common variant is probably where one combines one “succeed” type and one “failure” type of value (which is often a string type representing the error message), like the Haskell `Either` and the Rust `Result`. Mercury provide three different variants of this usage:

```

1 :- type maybe_error
2     --->    ok
3     ;      error(string).
4
5 :- type maybe_error(T) == maybe_error(T, string).
6
7     % Either a T, or an error E.
8 :- type maybe_error(T, E)
9     --->    ok(T)
10    ;      error(E).

```

---

<sup>4</sup>“Metalanguage”, i.e. the “ML” in Standard ML and OCaml, not “machine learning”.

## Universal type

In Mercury, this type (named `univ`) is simply a type that can be used to contain any value, similar to `Object` in Java for objects. It is sometimes used to produce values that need to be able to have different types; for example, the `list` type in Mercury is homogeneous (i.e. its elements must be of the same type), so one way to achieve heterogeneous lists is to convert all the values of different types into `univ` type and have a value of `list(univ)`.

Two main predicates have been provided in the `univ` module: `type_to_univ/2` and `univ_to_type/2`, which converts value into `univ` and converts `univ` into other types of values respectively. A function `univ/1` is also defined to have the same effect of predicate `type_to_univ/2`, which also probably get used more often.

Converting from `univ` to other types needs extra caution; since you don't know what the underlying type of an `univ` actually is during compile time, any such conversion could have a chance of failure. The predicate `univ_to_type/2` is thus naturally of determinism `semidet`. You can also use `det_univ_to_type/2`, which would throw a runtime error when the type mismatches.

Here is an example program showing how one can implement a form of heterogeneous list with `univ`:

```

1 :- module ex01.
2
3 :- interface.
4
5 :- use_module io.
6 :- pred main(io.state::di, io.state::uo) is det.
7
8 :- implementation.
9
10 :- import_module string, list, exception, univ.
11
12 :- func elem_to_string(univ::in) = (string::out) is det.
13 elem_to_string(X) = Y :-
14     % the `:int`, `:string` and `:float` in the end is necessary.
15     % this is used to specify which type you wish to convert the univ
16     % value to.
17     (univ_to_type(X, XVal:int) -> string.int_to_string(XVal, Y)
18     ; univ_to_type(X, XVal:string) -> Y = XVal
19     ; univ_to_type(X, XVal:float) -> string.float_to_string(XVal, Y)
20     ; throw("Invalid type")
21     ).
22
23 :- pred write_elem_list(list(univ)::in, io.state::di, io.state::uo) is
    det.
24 write_elem_list([], !IO).
25 write_elem_list([X|Xs], !IO) :-
26     io.write_string(elem_to_string(X), !IO),
27     io.write_string("\n", !IO),
28     write_elem_list(Xs, !IO).

```

```

29
30 main(!IO) :-
31     X = [univ("x"), univ(123), univ("y"), univ(456.78)],
32     write_elem_list(X, !IO).

```

## 3.4 If-Then-Else

There are two syntaxes for if-then-else:

```

1 % syntax 1:
2 if [condition]
3 then [then-clause]
4 else [else-clause]
5
6 % syntax 2:
7 [condition]
8 -> [then-clause]
9 ; [else-clause]

```

The second one is a carry-over from Prolog.

These are not just used for goals; you can use them as expressions as well. See the example below:

```

1 main(!IO) :-
2     io__write_int((3 > 5 -> 999; 1000), !IO).
3
4 % prints "1000"

```

## 3.5 Fundamental concepts

We must go through some of the concepts that occurs frequently in Mercury before we move on; since understanding them is vital for understanding some of the more advanced things about this language...

### Instantiatedness and mode

In Chapter 2 we discussed about some of the most fundamental instantiatedness<sup>5</sup> (**free**, **bound**, **ground**) and modes (**in**, **out**). Insts in Mercury are not necessarily about being bound or not – it’s something that describes a term’s properties beyond its type (and mode describes the *change* of a term’s insts instead of the insts themselves; so what we’ve been calling as the “mode signature” should really be called something like “inst declaration”, but I digress). We will discuss this more in Chapter 4.

<sup>5</sup>The word “instantiatedness” is too long. From now on, we will be calling it “inst”s instead.

## Determinism

As we have seen in Chapter 2, in Mercury, the following determinism categories exist:

- Deterministic (**det**): guaranteed to have one and exactly one solution.
- Semideterministic (**semidet**): have exactly one solution, but does not guarantee to produce it.
- Multisolution (**multi**): guaranteed to have a solution among possibly many solutions.
- Nondeterministic (**nondet**): have possibly many solutions, does not guarantee to produce one.
- Failure (**failure**): cases where there's zero solutions. They are not actual errors but a part of the logic (e.g. arity mismatch during unification, which will never produce a solution because the arity is different).
- Errorneous (**errorneous**): also have zero solutions, but they **do** represent actual errors which in other languages would be represented in the form of runtime exception throw or panic.

## Committed-Choice Determinism

Other than these six categories there are also two “committed-choice” classes: **cc\_nondet** and **cc\_multi**, which are committed-choice variants of **nondet** and **multi** respectively. **cc\_nondet** and **cc\_multi** goals backtrack, but when declared to be of **cc\_nondet** or **cc\_multi**, they *commit* to the first choice they would make during the resolution process and will not backtrack. This kind of determinism is thus used in cases where there could be multiple solutions but you don't care which one you end up with.

In Mercury, the **main** entry point can be defined to be either **det** or **cc\_multi**, since both of them guarantee one and would only produce one solution (**det**, of course, is more strict than **cc\_multi**, and if the Mercury compiler can determine something that should be able to be a **det** got labelled as a **cc\_multi**, it would spit out a warning saying you could've gone with the stricter option.)

## 3.6 Defining functions

By now we have all seen how to define *predicates*: you use the `:- pred` declaration like this:

```
1 :- pred pred_name(arg1_type, arg2_type, ..., argn_type).
2 :- mode pred_name(arg1_mode, arg2_mode, ..., argn_mode) is pred_det.
```

For example, we have defined a few predicates in Chapter 2 like this:

```
1 % "string" and "io.state" are types.
2 % "in", "di" and "uo" are modes.
3 % "det" is determinism.
4 :- pred rall(string, string, io.state, io.state).
5 :- mode rall(in, in, di, uo) is det.
```

Sometimes people declare the type signature and the mode signature separately, because in logic programming predicates can work “in reverse” and thus would require multiple mode signature. Take the signatures of the predicate `list.append` for example:

```
1 :- pred append(list(T), list(T), list(T)).
2 :- mode append(di, di, uo) is det.
3 :- mode append(in, in, out) is det.
4 :- mode append(in, in, in) is semidet.      % implied
5 :- mode append(in, out, in) is semidet.
6 :- mode append(out, out, in) is multi.
```

To explain this line by line:

- `list(T)` is a type, where `T` is a type variable, declaring that it works for any `list`.
- The first mode is simply a version of the second one that works on unique inputs.
- The second mode is for appending two lists together into a third list.
- The third mode is to check if combining the first two arguments result in the same value as the third argument. The check could succeed or fail, thus having the overall determinism of `semidet`.
- The fourth mode is to remove the “prefix” (described by the first argument) from the third argument. The third argument might not contain the first one as a prefix, thus having the overall determinism of `semidet`.
- The fifth mode asks the question of “which two lists can be combined and create the third one”. There obviously must be at least two solutions (exercise: can you figure out why this is the case?) and

there could be even more for some input, thus having the overall determinism of `multi`.

```
1 :- func f_name(arg1type, arg2type, ...) = result_type.
2 :- mode f_name(arg1mode, arg2mode, ...) = result_mode is f_det.
```

For example, a factorial function might have the following declaration:

```
1 :- func factorial(int) = int.
2 :- mode factorial(in) = out is det.
```

You also need to write `f_name(args) = result` instead of `f_name(args, result)` when providing the definition; other than that, the rest is mostly the same. Compare the following two definitions of the factorial function, one defined as a predicate with only one mode, one defined as a function:

```
1 :- pred factorial1(int, int).
2 :- mode factorial1(in, out) is det.
3 factorial1(X, Y) :-
4   ( X =< 1 -> Y = 1
5     ; ( factorial1(X - 1, Y0),
6         Y = X * Y0
7       )
8     ).
9
10 :- func factorial2(int) = int.
11 :- mode factorial2(in) = out is det.
12 factorial2(X) = Y :-
13   ( X =< 1 -> Y = 1
14     ; Y = X * factorial2(X - 1)
15   ).
```

Just like predicate declaration, you can combine the type signature and the mode signature in one; and if you do, you need to wrap the return type in parentheses, like this:

```
1 :- func factorial2(int::in) = (int::out) is det.
```

You can technically also declare functions (and predicates) like this, but this is technically not the “proper” way. The semantics of this, is to declare the function as a whole is of some higher-order function/predicate type.

```
1 :- func factorial2 : (func(int) = int).
```

A function, just like a predicate, can have multiple mode declaration. While most (if not all) functions we have come across would have their arguments having the mode `in` and their return value the mode `out`, but in Mercury, the reverse (i.e. arguments having `out` and return value having `in`) is technically possible. See the following example, in which the definition of function `f1` can be seen as asking the question “which tuple value can be seen as a result of reversing which tuple’s components”:



```

1  :- module ex01.
2
3  :- interface.
4
5  :- use_module io.
6  :- pred main(io.state::di, io.state::uo) is det.
7
8  :- implementation.
9  :- import_module int, string.
10
11 :- func f1({int, string}) = {string, int}.
12 :- mode f1(out) = in is det.
13 f1({X, Y}) = {Y, X}.
14
15 :- func first({T1, T2}) = T1.
16 :- mode first(in) = out is det.
17 first({X, _}) = X.
18
19 main(!IO) :-
20     {"abc", 3} = f1(P),
21     io.write_string(string.int_to_string(first(P)) ++ "\n", !IO).

```

## 3.7 User-defined types

### Equivalence types

The phrase “equivalence types” in Mercury refers to what’s more commonly known as *type aliases*, i.e. they are types that are defined to be fully equivalent in every way to some other types. This is often done for the sake of code readability, for example:

```

1  :- type money == int.
2  :- type assoc_list(Key, Value)
3      == list(pair(Key, Value)).

```

While `money` and `assoc_list(Key, Value)` are equivalent to `int` and `list(pair(Key, Value))`, one reading the code would have a basic idea that certain `ints` are intended for values representing an amount of money and certain `lists` are intended to be used as an association list.

(Honestly, the name “equivalence types” might bring confusion for people who have a Standard ML background, in which the idea of *equality type* refers to types that one can compare values with `=` in instead of type aliases, similar to types that have an `Eq` typeclass constraint in Haskell.)

### Discriminated unions

Discriminated union, or “tagged union” is a very useful (and thus commonly supported among functional languages) construct. First of all, it can be

used to implement what’s often called the “enumerated types” in other languages. For example:

```

1 :- type days_of_week
2   ---> monday
3       ;    tuesday
4       ;    wednesday
5       ;    thursday
6       ;    friday
7       ;    saturday
8       ;    sunday.

```

This is not unlike the following in C:

```

1 typedef enum DaysOfWeek {
2     MONDAY,
3     TUESDAY,
4     WEDNESDAY,
5     THURSDAY,
6     FRIDAY,
7     SATURDAY,
8     SUNDAY,
9 } DaysOfWeek;

```

The general syntax is as follows. (Note that the arrow here consists of three dashes. Mercury uses 2-dash arrow and 1-dash arrow for other things, so one should be mindful not to mix them up.)

```

1 :- type {type_name} --->
2       {variant_1} ; {variant_2} ; ... ; {variant_n} .

```

The core idea of a discriminated union type is exactly how it sounds: it’s a union type with a special value (often called a “tag”; this is also why discriminated unions are also called “tagged union”) that one uses to discern the different possible forms of the union. For example, you might often see patterns like this in languages like C:

```

1 typedef enum ExprType {
2     LIT_INTEGER,
3     BIN_OP,
4     UNARY_OP,
5     FUNC_CALL,
6 } ExprType;
7
8 typedef struct Expr {
9     ExprType typ;
10    // the ~typ~ field is checked before using the ~value~ field.
11    union {
12        int lit_int;
13        struct { struct Expr *l, *r; char* bop; } bin_op;
14        struct { struct Expr *b; char* uop; } un_op;
15        struct { char* fun_name; struct Expr* args } func_call;
16    } value;
17 }

```

It can be simply defined in Mercury as follows:

```

1 :- type expr
2     ---> lit_integer(int)
3     ;    bin_op(expr, expr, string)
4     ;    unary_op(expr, string)
5     ;    func_call(string, list(expr)).

```

We don't need to have a separate `:- type exprtype`; the tags are “built into” the type `expr`. When a type is defined this way, these tags are also defined as constructors, i.e. we can obtain a value of `expr` by simply using the terms `lit_integer/1`, `bin_op/3`, `unary_op/2` or `func_call/2`; for example, the term `lit_integer(3)` and the term `bin_op( lit_integer(3), lit_integer(4), "+")` would've automatically determined to have the type of `expr`.

Certain commonly-used types are actually defined this way. See the definition of lists and key-value lists:

```

1 % in module `list`
2 :- type list(T)
3     ---> []
4     ;    [T | list(T)].
5
6 % in module `kv_list`
7 :- type kv_list(K, V)
8     ---> kv_nil
9     ;    kv_cons(K, V, kv_list(K, V)).

```

Booleans are also defined this way:

```

1 % in module `bool`
2 :- type bool
3     ---> no
4     ;    yes.

```

To retrieve the different values stored within, we can use unification:

```

1 % ...
2 SomeVar = some_funcall(some_args),
3 ( ( SomeVar = lit_integer(IntVal), % ...do something w/ IntVal
4   );
5   ( SomeVar = bin_op(Left, Right, Op), % ...do something
6   );
7   ( SomeVar = unary_op(Body, Op), % ...do something
8   );
9   ( SomeVar = func_call(Head, Args), % ...do something
10  )
11 )

```

This can be tedious at times, especially in cases where you need to have a new value where only some of the fields have changed, in which you have no choice other than reconstructing the term from the ground up once you've defined your data type this way. Fortunately Mercury provides record syntax for easy field access and updates. If we choose to give fields of our `expr` type names like this (the syntax would be `name :: type`):

```

1 :- type expr
2   ---> lit_integer(lit_int :: int)
3       ; bin_op(l :: expr, r :: expr, bop :: string)
4       ; unary_op(b :: expr, uop :: string)
5       ; func_call(fun_name :: string, args :: list(expr)).

```

We can do the following extra things:

- *field\_name*(Term) returns the value of the specified *field\_name*. For example, `lit_int(lit_integer(3))` returns 3, and `uop(unary_op(lit_integer(3), "+"))` returns "+".
  - According to the language reference, unless you explicitly declare its mode, you cannot use it like a normal predicate in higher-order constructs since such functions generated from directly naming them in the `type` declaration are “implemented directly as unifications without the code of a function being generated for them”.
- `Term ^ field_name` also returns the value of the specified *field\_name*; this is equivalent to `field(Term)`. One can chain many of them together and retrieve values buried deep within.
- One can write `Term ^ field := Value` to create a new term with the specified field being replaced by *Value*. Same as accessing the fields, one can chain many *fieldtogetherandreplacevaluesburieddeepwithin*.

For example, assume we have the following definition:

```

1 :- type my_substruct
2   ---> my_substruct(field_2_1 :: string).
3 :- type my_struct
4   ---> my_struct(field_1 :: int, field_2 :: my_substruct).
5
6 main(!IO) :-
7   M = my_struct(3, my_substruct("blah")),
8   % ...

```

After binding the value to *M*, we can do the following:

```

1 % shows my_struct(3, my_substruct("blah"))
2 io.write(M, !IO)
3
4 % shows 3
5 io.write(M ^ field_1)
6
7 % shows my_substruct("blah")
8 io.write(M ^ field_2)
9
10 % shows "blah"

```

```

11 io.write(M ^ field_2 ^ field_2_1)
12
13 % bind MM with my_struct(4, my_substruct("blah"))
14 MM = (M ^ field_1 := 4)
15
16 % bind MM with my_struct(3, my_substruct("foo"))
17 MM = (M ^ field_2 ^ field_2_1 := "foo")

```

You don't need to give every single field a name when defining the data type, and you can create custom “getters” and “setters”, because the `Term ^ Field` and `Term ^ Field := Value` syntax are syntactic sugar and does not depend on the term being defined that way: `Term ^ Field` is equivalent to `Field(Term)`, and `Term ^ Field := Value` is equivalent to `Field := (Term, Value)`. With the example above, we can define the following getter and setter and use it like any other field access function:

```

1 :- func field_1_str(my_struct::in) = (string::out) is det.
2 field_1_str(my_struct(V, _)) = string.int_to_string(V).
3
4 :- func 'field_1_dbl :='(my_struct::in, int::in) = (my_struct::out) is
   det.
5 'field_1_dbl :='(my_struct(_, K), NewV) = my_struct(NewV*2, K).
6
7 main(!IO) :-
8     M = my_struct(3, my_substruct("blah")),
9     % shows "3"
10    io__write(M ^ field_1_str, !IO), io__nl(!IO),
11
12    MM = (M ^ field_1_dbl := 4),
13    % shows my_struct(8, my_substruct("blah"))
14    io__write(MM, !IO).

```

These field access functions can also have extra arguments; `Term ^ Field(Arg1,...)` is expanded into `Field(Arg1,...,Term)`, and `Term ^ Field(Arg1,...) := Val` is expanded into `'Field :='(Arg1,...,Term, Val)`. For example, we can have the following getter:

```

1 :- func field_1_sstr(string, string, my_struct) = string.
2 field_1_sstr(Left, Right, my_struct(V, _)) = Res :-
3     Res = Left ++ string.int_to_string(V) ++ Right.
4
5 main(!IO) :-
6     M = my_struct(3, my_substruct("blah")),
7     % shows "<3>"
8     io__write(M ^ field_1_sstr("<", ">"), !IO).

```

## Abstract types

Simply put, they are the declarations we've discussed above but without the definition part. These are mainly used in the `interface` section of a

module for the cases where you don't wish to expose the internals of a type to other module:

```

1 :- interface.
2 :- type t1.
3 :- type t2(T1, T2).
4 % ...
5
6 :- implementation.
7 :- type t1 --> % ...
8 :- type t2(T1, T2) == % ...
9 % ...

```

Of course, you must properly define them in the `implementation` section of the same module.

## 3.8 Example 1: Interpreter of a small language

With what we have learned up till now we can now try to write some small programs.

### The language

The language we'll be implementing would be a small, imperative one with very basic features like assignment, branching and loops. Its grammar can be described with the following Backus-Naur Form:

$$\begin{aligned}
 \langle stmt \rangle &::= \text{IDENT} \text{ ':=' } \langle expr \rangle \\
 &| \text{ 'IF' } \langle expr \rangle \text{ 'THEN' } \langle stmt \rangle \text{ 'ELSE' } \langle stmt \rangle \\
 &| \text{ 'BEGIN' } \langle stmt \rangle^* \text{ 'END' } \\
 &| \text{ 'WHILE' } \langle rel \rangle \text{ 'DO' } \langle stmt \rangle \\
 &| \text{ 'PRINT' } \langle expr \rangle \\
 \\
 \langle expr \rangle &::= \text{INTEGER} \\
 &| \text{ VAR} \\
 &| \langle expr \rangle ( \text{'+'} | \text{'-'} | \text{'*'} | \text{'/'} ) \langle expr \rangle \\
 &| ( \text{'-'} ) \langle expr \rangle \\
 \\
 \langle rel \rangle &::= \text{TRUE} \\
 &| \text{ FALSE} \\
 &| \langle expr \rangle ( \text{'>'} | \text{'<'} | \text{'>='} | \text{'<='} | \text{'!='} | \text{'=='} ) \langle expr \rangle \\
 &| \langle rel \rangle ( \text{'\&\&'} | \text{'||'} ) \langle rel \rangle \\
 &| \text{'!'} \langle rel \rangle
 \end{aligned}$$

From this BNF we can easily obtain the type definition:

```

1 :- import_module list.
2 :- type imp_expr
3   ---> e_val(int)
4   ; e_var(string)
5   ; e_binop(string, imp_expr, imp_expr)
6   ; e_unaryop(string, imp_expr).
7 :- type imp_rel
8   ---> r_rel(string, imp_expr, imp_expr)
9   ; r_binop(string, imp_rel, imp_rel)
10  ; r_unaryop(string, imp_rel)
11  ; r_true
12  ; r_false.
13 :- type imp_stmt
14   ---> s_assign(string, imp_expr)
15   ; s_begin(list(imp_stmt))
16   ; s_if(imp_rel, imp_stmt, imp_stmt)
17   ; s_while(imp_rel, imp_stmt)
18   ; s_print(imp_expr).

```

We will need a place to store the mapping between the variable names and their corresponding values:

```

1 :- import_module map.
2 :- type imp_env == map(string, int).

```

The evaluation of expression is easy. Since we don't need to update the environment while evaluating expressions for this language, we don't need an extra `imp_env::out` parameter:

```

1 % needed by `throw`
2 :- import_module exception.
3
4 :- pred imp_eval_expr(imp_expr::in, imp_env::in, int::out) is det.
5 imp_eval_expr(X, Env, Result) :-
6   ( X = e_val(XVal), Result = XVal );
7   ( X = e_var(XVar),
8     ( map__contains(Env, XVar) -> Result = lookup(Env, XVar)
9     ; throw("Cannot found var: " ++ XVar)
10    )
11   );
12   ( X = e_binop(Op, L, R),
13     imp_eval_expr(L, Env, LRes),
14     imp_eval_expr(R, Env, RRes),
15     ( Op = "+" -> Result = LRes + RRes
16     ; Op = "-" -> Result = LRes - RRes
17     ; Op = "*" -> Result = LRes * RRes
18     ; Op = "/" -> Result = LRes div RRes
19     ; throw("Unsupported bin op: " ++ Op)
20    )
21   );
22   ( X = e_unaryop(Op, Body),
23     imp_eval_expr(Body, Env, BRes),
24     ( Op = "-" -> Result = 0 - BRes
25     ; throw("Unsupported unary op: " ++ Op)
26    )
27   ).

```

Evaluating relations, however, requires some not-that-intuitive steps if you choose to first evaluate things into boolean values and dispatch on it. The comparison of numerical values are technically predicate calls, which only succeed or fail instead of producing a value like function calls; if we need a boolean value, we must convert this “succeed or fail” state into a boolean value. This is actually very simple, you just do this:

```

1 % the instantiatedness here restricts its argument to those that
2 % conforms to certain constraints. we will talk about this more
3 % in the section about instantiatedness.
4 :- func pred_to_bool( (pred)::in((pred) is semidet) )
5     = (bool::out) is det.
6 pred_to_bool(P) = (if P then yes else no).
```

This is available in the standard library by importing the `bool` module.

Equality check is even more different - when written directly, they’re considered as a goal of unification instead a predicate call, which means that we can’t even use `pred_to_bool`. One way to go around this is define a dedicated predicate:

```

1 :- func int_eq(int::in, int::in) = (bool::out) is det.
2 int_eq(X, Y) = (if X = Y then yes else no).
```

With this being available we can define the predicate for evaluating relations like this:

```

1 :- import_module bool.
2 :- pred imp_eval_rel(imp_rel::in, imp_env::in, bool::out) is det.
3 imp_eval_rel(X, Env, Result) :-
4     ( X = r_rel(Op, L, R),
5       imp_eval_expr(L, Env, LRes),
6       imp_eval_expr(R, Env, RRes),
7       ( Op = ">" -> Result = pred_to_bool((RRes < LRes))
8         ; Op = "<" -> Result = pred_to_bool((LRes < RRes))
9         ; Op = "==" -> Result = int_eq(LRes, RRes)
10        ; Op = ">=" -> Result = pred_to_bool((LRes >= RRes))
11        ; Op = "<=" -> Result = pred_to_bool((LRes <= RRes))
12        ; Op = "!=" -> Result = not(int_eq(LRes, RRes))
13        ; throw("Unsupported rel op: " ++ Op)
14      )
15     );
16 ( X = r_binop(Op, L, R),
17   imp_eval_rel(L, Env, LRes),
18   imp_eval_rel(R, Env, RRes),
19   ( Op = "&&" -> Result = and(LRes, RRes)
20     ; Op = "||" -> Result = or(LRes, RRes)
21     ; throw("Unsupported rel op: " ++ Op)
22   )
23 );
24 ( X = r_unaryop(Op, B),
25   imp_eval_rel(B, Env, BRes),
26   ( Op = "!" -> Result = not(BRes)
27     ; throw("Unsupported rel op: " ++ Op)
28   )
29 );
```



```

30 ( X = r_true, Result = yes );
31 ( X = r_false, Result = no ).

```

The predicate for executing statements can thus be defined as follows. Evaluating a statement could update the environment (by assignment statements), so we need an `imp_env::out` to receive the newest version of it:

```

1
2 :- pred imp_exec_stmt(imp_stmt::in, imp_env::in, imp_env::out) is det.
3 :- pred imp_exec_stmt_list(list(imp_stmt)::in, imp_env::in, imp_env::
  out) is det.
4 :- pred imp_exec_stmt_while(imp_rel::in, imp_stmt::in, imp_env::in,
  imp_env::out) is det.
5
6 imp_exec_stmt(X, Env, NewEnv) :-
7   ( X = s_assign(Var, Expr),
8     imp_eval_expr(Expr, Env, Val),
9     NewEnv = map__set(Env, Var, Val)
10  );
11   ( X = s_begin(Body),
12     imp_exec_stmt_list(Body, Env, NewEnv)
13  );
14   ( X = s_if(Cond, Then, Else),
15     imp_eval_rel(Cond, Env, CondRes),
16     ( CondRes = yes -> imp_exec_stmt(Then, Env, NewEnv)
17     ; imp_exec_stmt(Else, Env, NewEnv)
18     )
19  );
20   ( X = s_while(Cond, Body),
21     imp_exec_stmt_while(Cond, Body, Env, NewEnv)
22  );
23   ( X = s_print(Expr),
24     imp_eval_expr(Expr, Env, Val),
25     trace [ io(!IO) ] (
26       io__write_int(Val, !IO),
27       io__write_string("\n", !IO)
28     ),
29     NewEnv = Env
30  ).
31
32 imp_exec_stmt_list([], X, X).
33 imp_exec_stmt_list([X|Xs], Env, NewEnv) :-
34   imp_exec_stmt(X, Env, NewEnv1),
35   imp_exec_stmt_list(Xs, NewEnv1, NewEnv).
36
37 imp_exec_stmt_while(Cond, Body, Env, NewEnv) :-
38   imp_eval_rel(Cond, Env, CondRes),
39   ( CondRes = yes -> (
40     imp_exec_stmt(Body, Env, NewEnv1),
41     imp_exec_stmt_while(Cond, Body, NewEnv1, NewEnv)
42   )
43   ; NewEnv = Env
44  ).

```

You can tell the Mercury compiler didn't pick up the fact that `imp_exec_stmt_while` being potentially diverge since its determinism `det` somehow checks out.

A program in our language is conceptually just a list of statements. We will execute it by calling `imp_exec_stmt_list` with an empty environment:

```

1 :- pred imp_run(list(imp_stmt)::in) is det.
2 imp_run(X) :-
3     Env = map__init,
4     imp_exec_stmt_list(X, Env, _).

```

Then we can call it in our main:

```

1 main(!IO) :-
2     % X := 100
3     % SUM := 0
4     % WHILE X > 0 DO
5     %     BEGIN
6     %         SUM := SUM + X
7     %         X := X - 1
8     %     END
9     % PRINT SUM
10    imp_run(
11        [
12            s_assign("X", e_val(100)),
13            s_assign("SUM", e_val(0)),
14            s_while(
15                r_rel(">", e_var("X"), e_val(0)),
16                [
17                    s_assign("SUM", e_binop("+", e_var("SUM"), e_var("
18X"))),
19                    s_assign("X", e_binop("-", e_var("X"), e_val(1)))
20                ],
21            ),
22            s_print(e_var("SUM"))
23        ]
24    ).

```

Compile and run the program, and you'll see the number 5050 is printed.

## Full listing

```

1 :- module ex01.
2
3 :- interface.
4
5 :- use_module io.
6 :- pred main(io.state::di, io.state::uo) is det.
7
8 :- implementation.
9
10 :- import_module int, string, list, exception, bool, map.
11
12 :- type imp_expr
13     --> e_val(int)
14     ; e_var(string)
15     ; e_binop(string, imp_expr, imp_expr)
16     ; e_unaryop(string, imp_expr).
17 :- type imp_rel

```

```

18 ---> r_rel(string, imp_expr, imp_expr)
19 ; r_binop(string, imp_rel, imp_rel)
20 ; r_unaryop(string, imp_rel)
21 ; r_true
22 ; r_false.
23 :- type imp_stmt
24 ---> s_assign(string, imp_expr)
25 ; s_begin(list(imp_stmt))
26 ; s_if(imp_rel, imp_stmt, imp_stmt)
27 ; s_while(imp_rel, imp_stmt)
28 ; s_print(imp_expr).
29
30 :- type imp_env == map(string, int).
31
32 :- pred imp_eval_expr(imp_expr::in, imp_env::in, int::out) is det.
33 imp_eval_expr(X, Env, Result) :-
34   ( X = e_val(XVal), Result = XVal );
35   ( X = e_var(XVar),
36     ( map__contains(Env, XVar) -> Result = lookup(Env, XVar)
37       ; throw("Cannot found var: " ++ XVar)
38     )
39   );
40   ( X = e_binop(Op, L, R),
41     imp_eval_expr(L, Env, LRes),
42     imp_eval_expr(R, Env, RRes),
43     ( Op = "+" -> Result = LRes + RRes
44       ; Op = "-" -> Result = LRes - RRes
45       ; Op = "*" -> Result = LRes * RRes
46       ; Op = "/" -> Result = LRes div RRes
47       ; throw("Unsupported bin op: " ++ Op)
48     )
49   );
50   ( X = e_unaryop(Op, Body),
51     imp_eval_expr(Body, Env, BRes),
52     ( Op = "-" -> Result = 0 - BRes
53       ; throw("Unsupported unary op: " ++ Op)
54     )
55   ).
56
57 :- func int_eq(int::in, int::in) = (bool::out) is det.
58 int_eq(X, Y) = (if X = Y then yes else no).
59
60 :- pred imp_eval_rel(imp_rel::in, imp_env::in, bool::out) is det.
61 imp_eval_rel(X, Env, Result) :-
62   ( X = r_rel(Op, L, R),
63     imp_eval_expr(L, Env, LRes),
64     imp_eval_expr(R, Env, RRes),
65     ( Op = ">" -> Result = pred_to_bool((RRes < LRes))
66       ; Op = "<" -> Result = pred_to_bool((LRes < RRes))
67       % ; Op = "==" -> Result = (LRes, RRes)
68       ; Op = "==" -> Result = int_eq(LRes, RRes)
69       ; Op = ">=" -> Result = pred_to_bool((LRes >= RRes))
70       ; Op = "<=" -> Result = pred_to_bool((LRes <= RRes))
71       ; Op = "!=" -> Result = not(int_eq(LRes, RRes))
72       ; throw("Unsupported rel op: " ++ Op)
73     )
74   );
75   ( X = r_binop(Op, L, R),
76     imp_eval_rel(L, Env, LRes),

```

```

77     imp_eval_rel(R, Env, RRes),
78     ( Op = "&&" -> Result = and(LRes, RRes)
79     ; Op = "||" -> Result = or(LRes, RRes)
80     ; throw("Unsupported rel op: " ++ Op)
81     )
82 );
83 ( X = r_unaryop(Op, B),
84   imp_eval_rel(B, Env, BRes),
85   ( Op = "!" -> Result = not(BRes)
86   ; throw("Unsupported rel op: " ++ Op)
87   )
88 );
89 ( X = r_true, Result = yes );
90 ( X = r_false, Result = no ).
91
92 :- pred imp_exec_stmt(imp_stmt::in, imp_env::in, imp_env::out) is det.
93 :- pred imp_exec_stmt_list(list(imp_stmt)::in, imp_env::in, imp_env::
94   out) is det.
95 :- pred imp_exec_stmt_while(imp_rel::in, imp_stmt::in, imp_env::in,
96   imp_env::out) is det.
97
98 imp_exec_stmt(X, Env, NewEnv) :-
99   ( X = s_assign(Var, Expr),
100     imp_eval_expr(Expr, Env, Val),
101     NewEnv = map__set(Env, Var, Val)
102   );
103   ( X = s_begin(Body),
104     imp_exec_stmt_list(Body, Env, NewEnv)
105   );
106   ( X = s_if(Cond, Then, Else),
107     imp_eval_rel(Cond, Env, CondRes),
108     ( CondRes = yes -> imp_exec_stmt(Then, Env, NewEnv)
109     ; imp_exec_stmt(Else, Env, NewEnv)
110     )
111   );
112   ( X = s_while(Cond, Body),
113     imp_exec_stmt_while(Cond, Body, Env, NewEnv)
114   );
115   ( X = s_print(Expr),
116     imp_eval_expr(Expr, Env, Val),
117     trace [ io(!IO) ] (
118       io__write_int(Val, !IO),
119       io__write_string("\n", !IO)
120     ),
121     NewEnv = Env
122   ).
123
124 imp_exec_stmt_list([], X, X).
125 imp_exec_stmt_list([X|Xs], Env, NewEnv) :-
126   imp_exec_stmt(X, Env, NewEnv1),
127   imp_exec_stmt_list(Xs, NewEnv1, NewEnv).
128
129 imp_exec_stmt_while(Cond, Body, Env, NewEnv) :-
130   imp_eval_rel(Cond, Env, CondRes),
131   ( CondRes = yes -> (
132     imp_exec_stmt(Body, Env, NewEnv1),
133     imp_exec_stmt_while(Cond, Body, NewEnv1, NewEnv)
134   )
135   ; NewEnv = Env
136   ).

```

```

134     ).
135
136 :- pred imp_run(list(imp_stmt)::in) is det.
137 imp_run(X) :-
138     Env = map__init,
139     imp_exec_stmt_list(X, Env, _).
140
141 main(!IO) :-
142     imp_run(
143         [
144             s_assign("X", e_val(100)),
145             s_assign("SUM", e_val(0)),
146             s_while(
147                 r_rel(">", e_var("X"), e_val(0)),
148                 s_begin(
149                     [
150                         s_assign("SUM", e_binop("+", e_var("SUM"),
151 e_var("X"))),
152                         s_assign("X", e_binop("-", e_var("X"), e_val
153 (1)))
154                     ]
155                 ),
156             ),
157             s_print(e_var("SUM"))
158         ]
159     ).

```

## 3.9 Higher-order programming

In Mercury, predicates and functions are “first-class citizen” in commonly-used functional language lingo; this means that we can produce them as output values, take them as arguments, and create them without binding them to any name. The form of these predicates/functions without name is aptly called “anonymous predicate/function”; they are also (and more often) called “*lambda* (expression)”, which is a name from the lambda-calculus.

Currently you must explicitly state the mode and determinism in your anonymous predicate/function, but type is not required as long as Mercury could infer them. It’s still recommended to be explicit about types anyway. The way you state both the type and the mode of a parameter is to write `arg_name:type::mode`, like this:

```

1 main(!IO) :-
2     P = (pred(In:int::in, Out:int::out) is det :-
3         trace [ io(!IO) ] (
4             io__write_int(In, !IO),
5             io__write_string("\n", !IO)
6         ),
7         Out = In + 1
8     ),
9     P(3, _).

```

State variables cannot be arguments. For example, this is not allowed:

```

1 main(!IO) :-
2   P = (pred(In:int::in, !IO) is det :-
3       io__write_int(In, !IO),
4       io__write_string("\n", !IO)
5   ),
6   P(3, !IO).
```

You must state them separately with `!.`  and `!:`, after which you can use it like normal:

```

1 main(!IO) :-
2   P = (pred(In:int::in, !.IO:io.state::di, !:IO:io.state::uo) is det
3   :-
4       io__write_int(In, !IO),
5       io__write_string("\n", !IO)
6   ),
7   P(3, !IO).
```

Anonymous functions can be written in a similar fashion, but you can omit the mode declaration, since functions are considered to have the default `inst of func(in, in, ...) = out is det`.

```

1 main(!IO) :-
2   P = (func(In:int) = (Out:int) :-
3       trace [ io(!DebugIO) ] (
4           io__write_int(In, !DebugIO),
5           io__write_string("\n", !DebugIO)
6       ),
7       Out = In + 1
8   ),
9   io__write_int(P(3), !IO),
10  io__write_string("\n", !IO).
11
12 % outputs two line: "3" and "4".
```

Functions with no arguments need special care when calling. If you define a predicate with no arguments, you can still call it directly. For example:

```

1 main(!IO) :-
2   P = ((pred) is det :-
3       trace [ io(!DebugIO) ] ( io__write_string("P\n", !DebugIO
4   ) )
5   ),
6   % this actually calls the predicate bound to `P`.
7   P,
8   io__write_string("\n", !IO).
```

You can't do the same to functions with no arguments. For example, this will complain about a type mismatch between `func = int` and the required `int`:

```

1 main(!IO) :-
2   P = ((func) = 3),
3   io__write_int(P, !IO),
4   io__write_string("\n", !IO).

```

You can't solve this by writing `P:int` (which states that `P` should be of type `int`) either, and writing `P()` is syntax error. In order to call this, you need to use `apply`:

```

1 main(!IO) :-
2   P = ((func) = 3),
3   io__write_int(apply(P), !IO),
4   io__write_string("\n", !IO).
5
6 % compiles and prints "3".

```

`apply` is a built-in expression that applies its first argument on its second, third, etc. arguments. For example, assume we have `P = (func(A, B, C) = A + B + C)`, the expression `apply(P,3,4,5)` would have the same result as `P(3,4,5)`. A similar construct named `call` is also available for calling predicates.

Predicates and functions, when applied to a number of arguments that is less than required, is automatically curried. For example, you can write `=<(0)` to create a function that can be used to filter out all the values that are equal or larger than zero (since `=<(A, B)` actually means `A =< B`, in our case we're effectively writing `0 =<`):

```

1 :- pred write_int_list(list(int)::in, io.state::di, io.state::uo) is
2   det.
3 write_int_list([], !_).
4 write_int_list([X|Xs], !IO) :-
5   io__write_int(X, !IO), io__write_string(", ", !IO),
6   write_int_list(Xs, !IO).
7
8 main(!IO) :-
9   list__filter(=<(0), [0, 1, 3, -2, 3, -4], Res),
10  write_int_list(Res, !IO),
11  io__write_string("\n", !IO).
12 % writes "0, 1, 3, 3, "

```

Built-in language constructs like `=` and `apply`, however, cannot be used this way, and requires one use an explicit anonymous predicate/function. For example, this is illegal:

```

1 % ...
2 lst__filter(\=(2), [1, 2, 3], Res)

```

...instead, you must write this or define a named predicate separately:

```

1 % ...
2 lst__filter((pred(X:int::in) is semidet :- X \= 2), [1, 2, 3], Res)

```

One thing to remember: you cannot use the whole state variable within an anonymous predicate/function; you can only use the “current value”, whose semantics doesn’t involve modifying the state-of-the-world and is thus allowed. For example, the compiler would complain about this code, saying that you can only use `!.IO` within the lambda:

```

1 main(!IO) :-
2   P = (pred(Out:int::out) is det :-
3         io_write_string("hey!\n", !IO),
4         Out = 3
5       ),
6   P(_).
```

## 3.10 Exception handling

We already know that we can throw exceptions using `throw/1`. Now we learn how to handle them.

### Handling with predicates

A few predicates have been provided in the `exception` module:<sup>6</sup>

- `try(Goal, Result)`: This goal succeeds regardless, with `Result` containing either `failed` (showing that `Goal` has failed without throwing exceptions), `succeeded(Val)` (showing that `Goal` has succeeded and producing a result of `Val` without throwing exception) or `exception(E)` (showing that `Goal` has thrown an exception, which is captured in `E`).
- `try_io(Goal, Result, !IO)`: Same as `try/2`, but with extra parameters for IO. `try/2` requires its goal to be a `pred(out)` and we can’t directly use the IO state variable from the outside, thus dedicated predicate that would thread the state variable for us becomes necessary.
- `try_all(Goal, MaybeException, Solution)`: Collect all solutions of `Goal` which might throw exceptions into `Solution`. `MaybeException` would be bound with a value of type `maybe(univ)`; if an exception did get thrown, that `maybe` value would contain that exception; and if there has been no exception, it would simply be an empty value.

---

<sup>6</sup>There are other predicates, which have slightly more niche usage so we will not cover them here.



- `finally(Goal, Res, Cleanup, CleanupRes, !IO)`: The `Goal` and `Res` is the same as `try/2`, but `finally` would ensure to call `Cleanup` afterwards regardless of the result in `Res`, thus making this the “finally” construct in certain languages.

To unwrap thrown values from `throw/1` one needs to do some extra work when compared with other languages since the return value of `try/2` is a `exception_result(T)` which only contains the exception as an univ. For example, assume we have this goal:

```
1  % this goal would definitely throw an exception since we
2  % are indexing the 4th element of an empty list.
3  try((pred(Out:int::out) is det :- det_index0([], 3, Out)), R),
```

...we would first need to unwrap the `exception_result(T)` part. We don't need to handle the `failed` case since our call of `try/2` uses a `det` goal, which results in `R` having the inst of `cannot_fail` which describes a subset of `exception_result` that only contains `succeeded/1` and `exception`<sup>7</sup>

```
1  ( ( R = succeeded(K), io_write_string("succeeded!\n", !IO) );
2  ( R = exception(K),
3    % ... unwrap exception here.
4  )
5  )
6  .
```

`K` thus contains whatever values our goal have thrown. Mercury's standard libraries throw a dedicated type named `software_error`, which is a simple wrapper around a string:

```
1  det_univ_to_type(K, KE:software_error),
2  KE = software_error(KStr),
3  io_write_string("exception! string: ", !IO),
4  io_write_string(KStr, !IO) )
```

If what we have thrown are simpler things, e.g. a string, the handling would be simpler:

```
1  main(!IO) :-
2    try((pred(Out:int::out) is det :- throw("this is errmsg")),
3      R),
4    ( ( R = succeeded(K), io_write_string("succeeded!\n", !IO) );
5    ( R = exception(K),
6      det_univ_to_type(K, KE:string),
7      io_write_string("exception! string: " ++ KE ++ "\n", !IO)
8    )
9  )
10 .
```

<sup>7</sup>For how this works, see Chapter 4.

**finally/6**

As mentioned above, **finally** is like the “finally” construct in other languages (unlike **try** which has **try\_io**, **finally** only has one version, which supports threading in IO), but one might be surprised that **finally** rethrow the thrown exception *after* the cleanup goal has been successfully run without itself throwing an exception. To properly mimic the behaviour of “try ... catch ... finally”, one should use **try** *inside* a call to **finally**.

**Handling with try goals**

To create anonymous predicates every time we want to **try** something isn’t really that ergonomical. Good thing that Mercury provides a built-in **try** construct. Its grammar is as follows:

```

1 try Params Goal
2 then ThenClause
3 else ElseClause
4 catch Term1 -> CatchClause1
5 catch Term2 -> CatchClause2
6 ...
7 catch_any CatchAnyVar -> CatchAnyClause

```

Seems awfully complicated, at least more so than other conventional languages. Let’s break it down bit by bit:

- The **then** clause and the **else** clause are here to deal with cases where **Goal** does *not* throw an exception. If we only look at this part, it works similar to an if-then-else: if **Goal** succeeded, **ThenClause** is executed, or else **ElseClause** is executed. The **then** clause is mandatory.

Variables bound inside **Goal** would be available inside **ThenClause**.

- The **Params** part may look similar to the parameters in trace goals considering the language reference do mention (and only mentioned) an **io** parameter, but in this case this **io** parameter works exactly the opposite way. In a trace goal, **io** *gives* you an IO state variable that you can use inside the trace goal; but in a try goal, it’s *you* who have to pass in an IO state variable from the outside via **io** so that you can do IO inside **Goal**.

- If we choose to thread in a state variable, we can use that state variable to perform IO within **Goal**, but **Goal** itself cannot fail (and thus the try goal is not allowed to have an **else** clause); if we choose not to use any state variable, we can’t do IO within

Goal, but it is allowed to fail (and subsequently get handled by `else` clause).

- If `Goal` throws an exception, it will get unified with `Term1`, `Term2`, etc., until the unification succeeds, in which case the corresponding `catch` clause is executed. `Terms` can be a lot of things like actual literals (which checks if the exception is exactly those values) and variables with type declarations (which checks if the exception is of those types).
- Finally, the `catch_any` is optional. Just like its name suggests, it's intended to be a “catch-all” clause. It binds whatever is thrown to a variable (`CatchAnyVar` does indeed have to be a variable) which is available within `CatchAnyClause`. If all the `catch` clauses of a try goal fail to unify with the exception and the goal does not have a `catch_any` clause, that exception is rethrown to the outside.

Here is an example modified from the one in the language reference:

```

1 :- pred p(string::out) is det.
2 p(Output) :-
3     Output = "some string".
4
5 :- pred p_carefully(io.state::di, io.state::uo) is cc_multi.
6 p_carefully(!IO) :-
7     (try [io(!IO)] (
8         io.write_string("Calling p\n", !IO),
9         p(Output)
10    )
11    then
12        io.write_string("p returned: ", !IO),
13        io.write(Output, !IO),
14        io.nl(!IO)
15    catch S ->
16        io.write_string("p threw a string: ", !IO),
17        io.write_string(S, !IO),
18        io.nl(!IO)
19    catch 42 ->
20        io.write_string("p threw 42\n", !IO)
21    catch N:int ->
22        io.write_string("p threw other number\n", !IO)
23    catch_any Other ->
24        io.write_string("p threw something: ", !IO),
25        io.write(Other, !IO),
26        % Rethrow the value.
27        throw(Other)
28    ).

```

As the program above currently is, if we call `p_carefully`, it would print two lines of string: `Calling p` and `p returned: "some string"`. By throwing different things as exception in `p` we can change this behaviour. For example, changing `p`'s definition to this:

```

1 :- pred p(string::out) is det.
2 p(Output) :-
3     Output = "some string",
4     throw(42).

```

...would make the program print “p threw 42”.

One thing you might have noticed is the fact that the `catch S` clause seems to be capturing any kind of values; but the truth is the later use of `io.write_string` has restricted the type of `S` to `string`; this means that this particular `catch` clause would only catch strings. You can test it out by checking the difference of the output between the case where `p` throws a random string and the case where `p` throws a boolean value.

A try goal is `cc_multi` by default, regardless of what might get called inside `Goal`.

### 3.11 Advanced goals

Up till now we have seen the following forms of goals:

- Conjunction, written as `A, B`;
- Disjunction, written as `A; B`;
- Branching, written as `if Cond then A else B` or `(Cond -> A; B)`;
- Unification, written as `A = B`;
- Trace goals, written as `trace [Args] Goal`;

Fact is, there is much more than the one listed above. We will go through a few more of them in this section. Good thing is that Mercury provides a built-in try construct.

#### Other basic goals

- Success, written as `true`, is a goal that always succeed.
- Failure, written as `fail`, is a goal that always fail.
- Negation is written as `not Goal` or `\+ Goal`.
- Inequality, which is written as `A B`. This is equivalent to `not (A = B)`.

- Parallel conjunction, written as `A & B`, is a goal that executes both `A` and `B` in parallel. The actual behaviour depends on the backend (called “grade” in Mercury lingo), and in certain cases might act like normal conjunction. Both `A` and `B` must be either of determinism `det` or `cc_multi`.

## Existential quantification

Within the same clause of the definition of a predicate/function, all variables share the same scope; for example, in the snippet below, variables `N`, `M` and `M1` all share the same scope; new variables within this clause are thus required to not share the same name:

```

1 fac(N, M) :-
2   ( if ( N = 0 ) then ( M = 1 )
3     else if ( N = 1 ) then ( M = 1 )
4     else if ( N < 0 ) then ( M = 1 )
5     else ( fac(N-1, M1), M = N * M1 )
6   ).

```

Sometimes this might prove to be bothersome, especially in predicates that are more substantial. In this case, one might want to have separate scopes, where one can reuse certain variable names as if they have never been used. One can achieve this by using existential quantification, written as `some [Vars] Goal`; all variables declared in `[Vars]` can be used without fear within `Goal`. For example:

```

1 main(!IO) :-
2   X = 3,
3   some [X] (
4     X = "blah\n",
5     io.write_string(X, !IO)
6   ),
7   io.write_int(X, !IO).

```

This would compile without error (albeit the compiler would produce a warning complaining about `X` being reused) despite the variable `X` being bound to values of different types.

Technically all normal goals are implicitly existential quantified; making them explicit won’t introduce any change in semantics and/or actual behaviour.

Mercury also support universal quantification goals, which are of the form `all [Vars] Goal` and are equivalent to `not (some [Vars] (not Goal))`.

## Shorthands for certain compound goals

- Implication: `A => B`, which is equivalent to `not (A, not B)`.

- Reverse implication:  $A \leq B$ , which is equivalent to  $B \Rightarrow A$ , which is in turn equivalent to `not (B, not A)`.
- Logical equivalence:  $A \Leftrightarrow B$ , which is equivalent to  $(A \Rightarrow B), (B \Rightarrow A)$

## 3.12 Using multiple modules

There will be times where you would want to put things into different modules; it's better to talk about how to do this early than late.

### Basic module usage

There are two ways to import a module:

- `import_module`, which we have been using up till now. Names imported this way do not need to be fully qualified; the module name, when referring to the names defined within, can be omitted until there's ambiguity.
- `use_module`, whose syntax is the same as `import_module`, but the names imported this way needs to be fully qualified all the time.

Most of the time people would use a period `.` to separate the module name and the referred name, but Mercury supports using two underscores `__` as well; for example, instead of calling `io.write_string`, you can call `io__write_string`.

There is also this thing called “module-local mutable variable”, which does exactly what it sounds like. However, this feature involves Mercury's purity system, and for that reason we'll talk about it when we get there.

### Building with multiple modules

The Melbourne Mercury Compiler comes with two build tools named `mmake` and `mmc --make`. The reason why the latter is named `mmc --make` is because that to use this build tool you do need use the same executable but with a special command line argument. Both covers roughly the same set of features (with an amount of difference in details), and to be honest – both are under-documented and confusing. The official documentation recommends `mmc --make` because “this is the build tool that is likely to receive more development in the future”; this is also the one we will be using,

since the Mercury development team has designed it to make certain things easier.

Also, up till now we have directly called `mmc` with the file path; this has to change when we start using `mmc --make`, since `mmc --make` accepts *module name* as its argument instead of file path. If you pass in a path like `./ex01.m`, it would complain about the argument having unwanted stuff:

```
1 % mmc --make ./ex01.m
2 mercury_compile: a make target may not contain a directory component,
   but
3  `./ex01.m' does.
```

...and if you try to `mmc --make ex01.m`, you'll get this:

```
1 % mmc --make ex01.m
2 ** Nothing to be done for `ex01.m'.
```

This is *not* because there's truly nothing to be done for the *file* `ex01.m`; it's just that `mmc --make` has read that as a module name (see the section below about submodules) and since the file of the module (which is expected to be `ex01.m.m`) does not exist, it concluded that there's nothing to do. To properly make from the *file* `ex01.m`, you must refer to it by its module name, which is often `ex01`. For example, here we have a minimal “hello world” program with the module name `ex01`:

```
1 :- module ex01.
2
3 :- interface.
4
5 :- use_module io.
6 :- pred main(io.state::di, io.state::uo) is det.
7
8 :- implementation.
9
10 main(!IO) :-
11     io__write_string("hello world!", !IO).
```

If we run the command `mmc --make ex01`, it would produce outputs similar to this:

```
1 Making Mercury/int3s/ex01.int3
2 Making Mercury/ints/ex01.int
3 Making Mercury/cs/ex01.c
4 Making Mercury/os/ex01.o
5 Making ex01
```

## Submodules

There are times when you'd like to group certain things within a module into its own subgroup; in Mercury this can be achieved by defining a submodule.

There are two kinds of submodule in Mercury: ones that are defined and embedded within the “main” module, and ones that are in a separate file. For the former ones, you do it like this:

```

1 :- module main_module.
2 :- interface.
3
4 :- module submod.
5 :- interface.
6     % submod interface goes here.
7 :- end_module submod.
8
9 :- implementation.
10
11 :- module submod.
12 :- implementation.
13     % submod implementation goes here.
14 :- end_module submod.
15 % ...

```

Naturally, if you don’t want the definition of `submod` being visible from the outside, you put its interface section in the declaration of the parent module’s implementation section; but to put the implementation section of a nested submodule in the interface section of the parent module is forbidden.

If you want to put the definition of the submodule in a separate file, you must do the following:

- The content of that submodule must be in a file with the file name being the fully qualified name of the submodule (e.g. the source for the submodule `submod` of the parent module `mainmod1` must be put in the file with the name `mainmod1.submod.m`). (Using a different module name is possible; see the next section.)
- Add an `include_module` declaration in the parent module, like this:

```

1 :- module parent_module.
2 :- interface.
3 :- include_module submod1, submod2, ..., submod_n.

```

Either way, having the definition of the submodule doesn’t mean the names within the submodule is immediately available within the parent module; you still need to import the submodule explicitly if you want to use it.

## Using a different module name

Normally module names should match with the name of the file; but if you insist on using a different name, the Mercury compiler would need to know



which name corresponds to which file, or else it won't be able to build. Before discussing the remedy to this problem, let's see what would happen if we don't do anything about it. Assume we have two files: `ex01.m` and `ex02.m`, with the following content respectively:

```

1 % ex01.m
2 :- module ex01.
3
4 :- interface.
5 :- use_module io.
6 :- pred main(io.state::di, io.state::uo) is det.
7
8 :- implementation.
9 :- import_module fac.
10
11 main(!IO) :-
12     fac(5, M),
13     io_write(M, !IO).
```

```

1 :- module fac.
2
3 :- interface.
4 :- pred fac(int, int).
5 :- mode fac(in, out) is det.
6
7 :- implementation.
8 :- import_module int.
9
10 fac(N, M) :-
11     if ( N = 0 ) then ( M = 1 )
12     else if ( N = 1 ) then ( M = 1 )
13     else if ( N < 0 ) then ( M = 1 )
14     else ( fac(N-1, M1), M = N * M1 ).
```

If `ex01.m` imports `ex02.m` by the specified module name (i.e. `fac`), `mmc` would report this error:

```

1 Making Mercury/int3s/ex01.int3
2 ** Error reading file `fac.m' to generate dependencies.
3 ** Module `fac' is imported by module `ex01'.
4 mercury_compile: cannot find source for module `fac' in directories
  `.'
```

You can see `mmc` relies on this correspondence between module name and file name. If we modify `ex01.m` and choose to import `ex02` instead, `mmc` would report this different error:

```

1 ex02.m:001: Error: file `ex02.m' contains the wrong module. Expected
  module
2 ex02.m:001:  `ex02', found module `fac'.
```

A solution to this would be to run the command `mmc -f *.m`. This would produce a file named `Mercury.modules` under the same directory; its content would look something like this:

---

```
1 fac ex02.m
```

---

After this, while you still can't import `ex02` in `ex01.m`, you can now import `fac`. This is because that when the compiler couldn't find the corresponding file by the module name, it would expect this information to be in `Mercury.modules`. It is thus recommended to run `mmc -f *.m` before running `mmc --make` lest you have any such name-mismatch module. This behaviour also technically can be exploited to make the build tool use modules located at other places.

### 3.13 Installing and using libraries

To demonstrate this we will be writing a small library for parsing Gemtext, a markup language. Gemtext is used as the main markup language for the internet protocol Gemini<sup>8</sup>. We will skip the details and only give you the code (which is fully included in Appendix B), since our goal is not to know how to implement it in Mercury but to learn how to make and use libraries.

#### Installing libraries

To compile a target into a library instead of an executable, append `lib` at the front of the main module name; for example, instead of `mmc --make mmgemtext` (which would report a missing `main/2`), run `mmc --make libmmgemtext`. On \*nix systems this would produce a static library and a dynamic library, but currently Melbourne Mercury Compiler does not support compiling dynamic libraries on Windows. You can use the command line argument `--lib-linkage [shared|static]` to specify which kind you want.

To install a library, add `.install` at the end, i.e. run `mmc --make libmmgemtext.install` in this case. Note that if you don't specify an install location (with the command line argument `--install-prefix [location]`), `mmc --make` would install the library alongside with the standard libraries by default, which might not be a location where your user account has the permission to do things with depending on how you have installed the compiler (for example, if you compile and install Melbourne Mercury Compiler from source, your installation directory might be something like `/usr/local/mercury-22.01.8`, which is a directory normal users can't edit due to being inside `/usr/local`. You might be tempted to simply add `sudo` or other similar commands to elevate your privilege in this case, but then

---

<sup>8</sup>For more information on Gemini, please visit <https://geminiprotocol.net>

you would have to make sure the super user account has the directory of the Mercury executable in the PATH environment variable, or else `mmc --make` sometimes would fail silently.

If you manually specify an install prefix, `mmc` would put the compiled files in the directory `lib/mercury` under the prefix path, which you would have to pass to the `--mld` command line argument; see the next section about using installed libraries.

## Using libraries

If you have installed the library at the default location, you can simply `import_module` like importing a module from the standard library. We can use the following sample program as an example:

```

1 :- module ex01.
2
3 :- interface.
4 :- import_module io.
5 :- pred main(io.state::di, io.state::uo) is det.
6
7 :- implementation.
8 :- import_module mmgemtext, mmgemtext.parser, mmgemtext.html.
9 main(!IO) :-
10     DocumentString = "\
11 # heading 1\n\
12 ## heading 2\n\
13 ### heading 3\n\
14 \n\
15 normal text paragraph\n\
16 \n\
17 => https://sebastian.graphics\n\
18 => https://sebastian.graphics/index.html home page\n\
19 \n\
20 * list item 1\n\
21 \n\
22 * second list item 1\n\
23 * second list item 2\n\
24 * second list item 3\n\
25 \n\
26 > quoted line\n\
27 > quoted line 2 \n\
28 \n\
29 ```\n\
30 pre line 1\n\
31 pre line 2\n\
32 pre line 3\n\
33 ```\n",
34     Parsed = mmgemtext.parser.parse_text(DocumentString),
35     Rendered = mmgemtext.html.document_to_html(Parsed),
36     io__write_string(Rendered, !IO).
```

If you run `mmc --make ex01` it would complain about not having the function; to properly compile this, you need to add the command line ar-

gument `--ml mmgemtext`; if you have chosen to install the library at somewhere else, besides `--ml` you will also need to add the `--mld [dir]` command line argument, in which `[dir]` should be the `lib` under the installation prefix.<sup>9</sup>

---

<sup>9</sup>If you have encountered problems during this, try removing the `Mercury` subdirectory generated by previous build attempts and try again; since MMC caches stuff, sometimes it would accidentally rely on outdated info.

# Advanced Concepts & Usage

## 4.1 Instantiatedness

As we have already seen in Chapter 2, 3 of the most common insts in Mercury describes the “bound-ness” of terms: **free** representing a free term, **bound** representing a bound term, and **ground** which represents a fully-bound term. This is not all there is to instantiatedness; in Mercury, you can also define custom insts to describe *states a term could have*. One common use of this semantics is to describe the determinism of a predicate/function parameter since type signatures does not contain determinism info. For example, the function `pred_to_bool` which converts a call to predicate to a boolean value has the following mode declaration; the first parameter is declared to have the type `(pred)` and the mode `(pred) is semidet`, which restricts the allowed range of values for the first parameter to predicate calls that are of determinism `semidet`:

```
1 :- func pred_to_bool((pred)::((pred) is semidet)) = (bool::out) is det
   .
```

...calling it with anything else, like a `multi`, would now cause a compile error:

```
1 :- pred my_pred(int::out) is multi.
2 my_pred(3).
3 my_pred(4).
4 my_pred(5).
5
6 main(!IO) :-
7     X = pred_to_bool((my_pred(_))),
8     % ...
9
10 % this outputs the following error:
```

```

11 % ex01.m:019: In clause for `main(di,uo)':
12 % ex01.m:019:   in argument 1 of call to function `bool.pred_to_bool
    '/1:
13 % ex01.m:019:   mode error: variable `V_8' has instantiatedness
14 % ex01.m:019:   `/* unique */ (pred is multi)',
15 % ex01.m:019:   expected instantiatedness was `(pred is semidet)'.
16 % ex01.m:019: In clause for predicate `main'/2:
17 % ex01.m:019:   warning: variable `X' occurs only once in this scope.

```

You can define such insts for algebraic datatypes as well. While doing that, you need to specify the datatype the inst is for by specifying `for type_name/arity`. For example, an inst for `maybe/1` which requires values to be a fully-bound `yes` can be defined as follow:

```

1 :- inst maybe_yes_ground for maybe/1
2   ---> yes(ground).

```

A non-empty list can be defined as such:

```

1 :- inst non_empty_list for list/1
2   ---> [ ground | ground ].

```

It can then be used to describe situations where the list must be non-empty, for example, in the mode declaration of the predicate `solutions/2` (in module `solutions`). This is a predicate that collects all the solutions of a specified goal into a list; and when that goal is of determinism `multi`, this list would be non-empty. The mode declaration of this situation can thus be as follows:

```

1 :- mode solutions(pred(out) is multi, out(non_empty_list)) is det.
2 :- mode solutions(pred(out) is nondet, out) is det.

```

The first mode declaration restricts the first parameter to those predicates that are of determinism `multi`; since the predicate call is `multi`, the call must have at least one solution, and for this reason we can provide more information to the second parameter and describe it as an non-empty list. We couldn't do the same for the second one, since `nondet` is the most general determinism possible; with a `nondet` predicate call, we can't say anything about the possible state of the list of solutions, since there can be no solutions, one solution or many solutions.

Higher-order insts

## 4.2 Modes

Higher-order modes

## 4.3 Multiple solutions

## 4.4 Definite Clause Grammar

## 4.5 Typeclasses

## 4.6 Existential Types

## 4.7 Purity System





# Parsing With Definite Clause Grammar



## Foreign Language Interface

- 6.1 Calling back and forth
- 6.2 Working with the Java grade
- 6.3 Working with the CSharp grade



# Debugging Mercury Programs



## .gitignore for Mercury projects

Hereby is an example `.gitignore` for Mercury projects. This should cover most of the artifacts generated during the building process.

```
1 # Ignore the 'Mercury' directory at the root level
2 /Mercury/
3
4 # Ignore files with specific extensions in the root directory only
5 /*.err
6 /*.mh
7 /*.c
8 /*.c_date
9 /*.d
10 /*.int
11 /*.int0
12 /*.int1
13 /*.int2
14 /*.int3
```





## Full source code for mmgemtext

`mmgemtext` provides a basic parser and HTML converter for the Gemtext markup language.

Tested with Melbourne Mercury Compiler 22.01.8.

### B.1 Source code listing

#### `mmgemtext.m`

```
1  :- module mmgemtext.
2
3  % mmgemtext - Gemtext library for Mercury
4  %
5  % Zetian Lin 2025, No Rights Reserved
6  %
7  % This code is hereby declared into public domain and Creative
8  % Commons Zero.
9
10 :- interface.
11 :- func version = string.
12
13 :- include_module ast.
14 :- include_module parser.
15
16 :- implementation.
17
18 version = "1.0".
```

**mmgemtext.ast.m**

```

1 :- module mmgemtext.ast.
2
3 % mmgemtext - Gemtext library for Mercury
4 %
5 % Zetian Lin 2025, No Rights Reserved
6 %
7 % This code is hereby declared into public domain and Creative
8 % Commons Zero.
9
10 :- interface.
11
12 :- import_module list.
13
14 :- type gemtext_node
15   --> gmi_text(text_body :: string)
16   ; gmi_link(link_url :: string, link_text :: string)
17   ; gmi_heading(heading_level :: int, heading_text :: string)
18   ; gmi_list(list_items :: list(string))
19   ; gmi_quote(quote_body :: string)
20   ; gmi_pre(pre_lines :: list(string)).
21
22 :- type gemtext_document == list(gemtext_node).

```

**mmgemtext.parser.m**

```

1 :- module mmgemtext.parser.
2
3 % mmgemtext.parser - parser for gemtext.
4 %
5 % Zetian Lin 2025, No Rights Reserved
6 %
7 % This code is hereby declared into public domain and Creative
8 % Commons Zero.
9
10 :- interface.
11 :- import_module mmgemtext.ast.
12
13 :- func parse_text(string) = gemtext_document.
14
15 :- implementation.
16 :- import_module string, list, char, maybe, std_util.
17
18 :- pred read_as_heading(string::in, gemtext_node::out) is semidet.
19 read_as_heading(S, N) :-
20   ( prefix(S, "###") -> (
21       Body = lstrip(remove_prefix_if_present("###", S)),
22       N = gmi_heading(3, Body)
23   )
24   ; prefix(S, "##") -> (
25       Body = lstrip(remove_prefix_if_present("##", S)),
26       N = gmi_heading(2, Body)
27   )
28   ; prefix(S, "#") -> (

```

```

29         Body = lstrip(remove_prefix_if_present("#", S)),
30         N = gmi_heading(1, Body)
31     )
32     ; fail
33 ).
34
35 :- pred read_as_quote(string::in, gemtext_node::out) is semidet.
36 read_as_quote(S, N) :-
37     ( prefix(S, ">") -> (
38         Body = lstrip(remove_prefix_if_present(">", S)),
39         N = gmi_quote(Body)
40     )
41     ; fail
42 ).
43
44
45 :- pred read_as_link(string::in, gemtext_node::out) is semidet.
46 read_as_link(S, N) :-
47     ( prefix(S, ">") -> (
48         PreBody = lstrip(remove_prefix_if_present(">", S)),
49         URL_Length = prefix_length(isnt(char.is_whitespace), PreBody)
50     ),
51     split(PreBody, URL_Length, URL, Text),
52     N = gmi_link(URL, lstrip(Text))
53     ; fail
54 ).
55
56
57 :- pred is_list_item(string::in) is semidet.
58 is_list_item(S) :- prefix(S, "* ").
59
60 :- pred read_as_list(list(string)::in, gemtext_node::out, list(string)
61 ::out) is semidet.
62 read_as_list(L, Res, Rest) :-
63     ( L = [] -> fail
64     ; ( L = [X|_],
65         ( not is_list_item(X) -> fail
66         ; ( Res = gmi_list(map(remove_prefix_if_present("* "),
67                                take_while(is_list_item, L))),
68           Rest = drop_while(is_list_item, L)
69         )
70     )
71 ).
72
73 :- pred check_if_not_pre_toggle(string::in) is semidet.
74 check_if_not_pre_toggle(S) :- not prefix(S, "`").
75
76 :- func drop_pre_toggle(list(string)) = list(string).
77 drop_pre_toggle([]) = [].
78 drop_pre_toggle([X|Xs]) = Res :-
79     (prefix(X, "`") -> Res = Xs; Res = [X|Xs]).
80
81 :- pred read_as_pre(list(string)::in, gemtext_node::out, list(string)
82 ::out) is semidet.
83 read_as_pre(L, Res, Rest) :-
84     ( L = [] -> fail
85     ; ( L = [X|Xs],

```

```

85     ( not prefix(X, "```") -> fail
86     ; ( Res = gmi_pre(take_while(check_if_not_pre_toggle, Xs)),
87       Rest = drop_pre_toggle(drop_while(check_if_not_pre_toggle,
88         Xs))
89     )
90   )
91 ).
92
93 :- pred use_as_text(string::in, gemtext_node::out) is det.
94 use_as_text(S, N) :- N = gmi_text(S).
95
96 :- pred read_node(list(string)::in, maybe(gemtext_node)::out, list(
97   string)::out) is det.
98 read_node(Line, Res, Rest) :-
99   ( ( Line = [], Res = no, Rest = [] );
100     ( Line = [X|Xs],
101       ( read_as_heading(X, N) -> Res = yes(N), Rest = Xs
102         ; read_as_link(X, N) -> Res = yes(N), Rest = Xs
103         ; read_as_quote(X, N) -> Res = yes(N), Rest = Xs
104         ; read_as_list(Line, N, R) -> Res = yes(N), Rest = R
105         ; read_as_pre(Line, N, R) -> Res = yes(N), Rest = R
106         ; ( use_as_text(X, N), Res = yes(N), Rest = Xs )
107       )
108     ).
109
110 :- func read_node_list(list(string)) = list(gemtext_node).
111 read_node_list(Lines) = Res :-
112   ( Lines = [] -> Res = []
113   ; (
114     read_node(Lines, Node, Rest),
115     Res0 = read_node_list(Rest),
116     ( ( Node = no, Res = Res0 );
117       ( Node = yes(Val), Res = [Val|Res0] )
118     )
119   )).
120
121 parse_text(String) = Document :-
122   Document = read_node_list(split_into_lines(String)).

```

### mmgemtext.html.m

```

1 :- module mmgemtext.html.
2
3 % mmgemtext.html - html generator for parsed gemtext.
4 %
5 % Zetian Lin 2025, No Rights Reserved
6 %
7 % This code is hereby declared into public domain and Creative
8 % Commons Zero.
9
10 :- interface.
11 :- import_module mmgemtext.ast.
12
13 :- func document_to_html(gemtext_document) = string.

```

```

14
15 :- implementation.
16 :- import_module list, string.
17
18 :- func node_to_html(gemtext_node) = string.
19 node_to_html(N) = Res :-
20   ( N = gmi_text(Body), Res = format("<p>%s</p>", [s(Body)]) );
21   ( N = gmi_heading(Level, Body),
22     Res = format("<h%d>%s</h%d>", [i(Level), s(Body), i(Level)]) );
23   ( N = gmi_list(ItemList),
24     ItemsResult = map((func(X:string) = format("<li>%s</li>", [s(X)
25   ])), ItemList),
26     Res = format("<ul>%s</ul>", [s(join_list("", ItemsResult))] ) );
27   ( N = gmi_quote(Body),
28     Res = format("<blockquote>%s</blockquote>", [s(Body)]) );
29   ( N = gmi_pre(Lines),
30     Res = format("<pre>%s</pre>", [s(join_list("\n", Lines))] ) );
31   ( N = gmi_link(URL, Text),
32     Res = format("<p><a href=\"%s\">%s</a></p>", [
33       s(URL),
34       s((if ( strip(Text) = "" ) then URL else strip(
35         Text)))
36     ]))
37   ).
38
39 document_to_html(Doc) = Res :-
40   Res = join_list("", map(node_to_html, Doc)).

```

