
A Gentle Introduction To Mercury

The Functional-Logic Programming Language

Zetian Lin



A Gentle Introduction to Mercury

 $The \ Functional\text{-}Logic \ Programming \ Language$

Zetian Lin (Very early draft; October 2025) © 2025 Zetian Lin. Some rights reserved. https://sebastian.graphics

Licensed under Creative Commons Attribution-NonCommercial-NoDerivatives 4.0 International with additional terms. To view a copy of the Creative Commons license, please visit https://creativecommons.org/licenses/by-nc-nd/4.0/.

Additional terms requested by the author:

All content is shared according to the policies above with all parties EXCEPT any Wikimedia Foundation projects. Under NO CIR-CUMSTANCES may the content of this work be reproduced, distributed or transmitted in any form or by any means, including photocopying, recording, or other electronic or mechanical methods for the use of ANY existing and future Wikimedia Foundation project. Permission for any individual or organization to use the content of this work in any form for Wikimedia Foundation projects is hereby explicitly revoked. Any individual or organization that uses the content for any Wikimedia Foundation project will be subject to legal action.

Contents

Contents

1	Intr	oduction 1
	1.1	What is Mercury?
	1.2	The state of things as of now
	1.3	Who's this book for
	1.4	The tool we will be using
	1.5	More importantly why ?
2	A F	irst Taste of Mercury 7
	2.1	Prerequisite: logic programming
	2.2	Writing the program
3	Prog	gramming in Mercury 29
	3.1	Trace Goals
	3.2	Using the standard library
	3.3	Basic Datatypes
	3.4	If-Then-Else
	3.5	Defining functions
	3.6	User-defined types
	3.7	Example 1: Interpreter of a small language
	3.8	Using multiple modules
	3.9	Exception handling
	3.10	Higher-order programming
4	\mathbf{Adv}	anced Concepts & Usage 55
	4.1	Mode & Instantiatedness
	4.2	Multiple solutions & Determinism
	4.3	Definite Clause Grammar
	4.4	Typeclasses
	4.5	Existential Types

CONTENTS

	4.6 Purity System	
5	Parsing With Definite Clause Grammar	57
6	Foreign Language Interface	59
	6.1 Calling back and forth	59
	6.2 Working with the Java grade	59
	6.3 Working with the CSharp grade	
7	Debugging Mercury Programs	61

Introduction

1.1 What is Mercury?

Mercury¹ is a functional logic programming language, which is a paradigm that combines features common in certain functional languages (e.g. higher-order constructs, strong advanced typing) and logic programming semantics (e.g. non-determinism, built-in search). It attempts to combine the clarity and expressiveness of declarative programming with advanced static analysis and error detection features. According to its creators, it's a "general purpose language intended to support the creation of large, reliable and efficient applications." Although a short-lived renewed interest in this paradigm has been generated by Epic Games's programming language Verse² a few years earlier, it has now seemingly died down unfortunately; the paradigm remains interesting regardless, for the obvious reason that it provides extra edge against both purely functional and purely logic languages.

As are most of the languages that are of similar categories, Mercury has been mostly used for academic/research purposes, but a few commercial products have chosen (or once chosen) Mercury as their main language as well, like PrinceXML³, which is an HTML rendering tool used in publishing.

¹https://mercurylang.org

²https://dev.epicgames.com/documentation/en-us/fortnite/

verse-language-reference

³https://www.princexml.com

Other similar languages

A few languages have been said to be similar to Mercury on its different aspects:

- Curry⁴: When discussing functional logic programming, people naturally lean towards a conceptual Haskell-Prolog continuum; if one were to say Mercury is closer to the Prolog side, Curry is clearly closer to the Haskell side of things.
- Clean⁵: Clean is the lesser-known sibling of Haskell famed for its uniqueness typing. It was said that the ability to explicitly specify uniqueness is not unlike that of Mercury.

1.2 The state of things as of now

Mercury has been considered a difficult language to learn, even for people who has previous functional language experiences. While Mercury do have unfamiliar aspects from the viewpoint of programmers who have only written in "conventional" languages like Scala or Haskell, I personally find those aspects aren't as difficult as one may think; I believe, that to a certain point this "difficult" feel comes from other factors, like the lack of a visible software ecosystem:

- Lack of example, tutorials and other introductory text: To be honest, it's not that we don't *have* those, it's simply that most of them tend to be slightly too confusing for common programmers who are used to conventional imperative/functional languages. As of the time of writing (October 2025), we have the following attempts:
 - The tutorial written by Ralph Becket which is listed on Mercury's official website⁶. Unfortunately this tutorial was left unfinished years ago and its author was said to have since stopped doing Mercury-related things.
 - The tutorials on Mercury's GitHub repository wiki⁷, which seemingly has ceased major activities since 2018.

⁴https://curry-lang.org

⁵https://clean-lang.org

⁶https://mercurylang.org/documentation/learning.html

⁷https://github.com/Mercury-Language/mercury/wiki/Tutorial

- "Learn Mercury By Y Minute"⁸, which didn't do a very good job at explaining things to a depth a learner would need their materials to have.
- "Mercury Crash Course" written by J Fondren, on the nowdefunct mercury-in.space website⁹, which honestly was too confusing for newcomers to provide any help.
- A book titled "Declarative Programming in Mercury" is also available by compiling from the TeX source yourself using the source code¹⁰, but this one really isn't an introductory text...
- Lack of package manager: Same as above, it's not like we don't have one in fact we have two, one called Merchant by Stewart Slocum¹¹, and one called mmc-get¹² by J Fondren, which used to have a handful of packages¹³. They are not compatible with each other (of course), which I doubt have ever created any kind of problem since nearly no one used them.
- Lack of libraries: This is the real pain point. If a language did not have enough libraries, people are much less likely to use the language to make things; and if not enough people are using the language, there wouldn't be enough demand and thus there wouldn't be enough libraries. It's a viscious cycle that requires a *lot* of effort to break.¹⁴

But all of these situations won't change if we simply do nothing and hope for that one dumbass to put their love and effort for little to none rewards, which is why I have decided to become that dumbass myself and write this book to bridge the gap somewhat.

⁸https://learnxinyminutes.com/mercury/

⁹You can still read it on Wayback Machine: https://web.archive.org/web/ 20220815044519/https://mercury-in.space/crash.html

 $^{^{10}\}mbox{Available}$ on GitHub: https://github.com/Mercury-Language/books/tree/master/decl_prog

¹¹https://github.com/stewy33/merchant

¹²https://github.com/jrfondren/mmc-get

¹³https://web.archive.org/web/20221130042103/https://mercury-in.space/ packages/packages.list

¹⁴However, a list of available resources (managed by Volker Wysk) can be found at https://volker-wysk.de/mercury/resources.html; although they might not necessary be the ones you want.

1.3 Who's this book for

This book is for programmers who have previous experiences with imperative/functional languages who knows how programming works in principle and are used to filling in the details which could've been thought as incomplete or even missing. While learning Mercury as your first programming language might be a fun experiment to run on yourself, as the state of things being they are right now, it's just a little too much to be truly beginner-friendly, so I can't say I recommend it. Maybe years later when we have as much things as Haskell, this option would be viable again.

1.4 The tool we will be using

As of now (October 2025) there is only one implementation of Mercury: the Melbourne Mercury Compiler. All code in this book has been tested against version 22.01.8. Installing the compiler is probably one of the most painful part: since there is no official prebuilt binaries, you will have to download the source code, compile it and install it yourself. If you're using an operating system with a package manager (like most Linuxes), their package repositories might have packages that ease the complexity of managing a little bit, but the package might simply be a automated build instructions without any pre-built binaries and your machine still needs to build the compiler, which takes a very long time.

Strangely, for the Melbourne Mercury Compiler, I actually recommend you to build from the source code instead of relying on the package repository available: MMC contains different backends and configurations called "grades" and is actually capable of generating code for Java and C Sharp:

-s grade --grade grade

Select the compilation model. This model, which Mercury calls a grade, specifies what properties the resulting executable or library should have. Properties such as 'generates profiling data for mprof when executed' and 'can be debugged with the Mercury debugger'. As such, it controls decisions that must be made the same way in all the modules of a program.

— From the man page of mmc

Certain things also requires a separate "grade" being available (e.g. weirdly, single-precision floating point numbers; see the section about basic

types in Chapter 3). The package in the package repository of your operating system might not be configured with the grades you're interested in enabled.

1.5 More importantly... why?

A standard answer to this question would be "why not?"¹⁵, but I believe that people who would ask would like a different answer. The biggest appeal of Mercury, I think, is probably its ability to provide the power of both functional and logic programming in a rigorous way that can support developments of large-scale projects, much like having a static type system has been proven to be helpful (at least to certain degree) in this regard – before this, most logic programming languages either don't have the static checks Mercury do or were only really suitable for academic purposes. You may argue that you don't need all those and can still do everything with conventional languages like C; I will give you the win since I have no arguments against that, but can you truly convince yourself that that's a good argument?

¹⁵But seriously, if you've got the time, why not?

A First Taste of Mercury

Unfortunately, due to Mercury currently not having a read-eval-print-loop, it's nigh impossible to properly introduce the basic part of Mercury in details without introducing some kind of scaffolding first. For this reason, in this chapter we'll be forcing our way into writing and completing a tiny utility program; after this, we will have a basic idea of how to write a bare-bone command line program in Mercury, which we can then use as a starting point for exploring the language.

2.1 Prerequisite: logic programming

While it might not seem obvious from an outsider's view, logic programming did in fact come from first-order predicate logic; more specificly, it comes from a specific interpretation of a specific class of first-order predicat logic formula. This class of formula, called **Horn Clause** after the logician Alfred Horn, can be one of the following forms:

- "Rule", which takes the form of $Q \leftarrow P_0 \wedge P_1 \wedge \cdots \wedge P_n$. We will read this as "Q holds when all of P_0 and P_1 and the rest all the way to n holds"
- "Fact", which takes the form of $P \leftarrow \top$. We will read this as "P holds (regardless)". Sometimes it's also shortened further into P. in some literature.
- "Goal", which takes the form of $\bot \leftarrow P_0 \land P_1 \land \cdots \land P_n$. We will read this as "Show all of P_0 and P_1 and the rest holds".

Within these formula, all the Q and P_n have the form of $f(arg_0, arg_1, ... arg_n)$ where f is simply a name and all the arg_n are themselves first-order predicate logic terms. If how this has anything to do with writing programs isn't clear to you, it probably will after reading the definition of the factorial function, written in Horn clauses:

```
fac(0,1) \leftarrow \top
fac(1,1) \leftarrow \top
fac(N,M) \leftarrow minus(N,1,N_0) \land fac(N_0,M_0) \land mult(M_0,N,M)
```

Compare this with the following:

```
fac(0)=1 fac(1)=1 fac(N)=M where N_0=N-1, M_0=fac(N_0), M=N	imes M_0
```

...and the following code in Python:

```
1 def fac(n):
2    if n == 0: return 0
3    elif n == 1: return 1
4    else:
5         n0 = n - 1
6         m0 = fac(n0)
7         return n * m0
```

The similarity should be obvious at this point; in a very inaccurate sense, you can think of it as:

- Rules as function definition;
- Predicates as function calls;
- Conjunction as sequencing;
- Facts as base cases;
- Goals as function calls & entry points;

This, of course, is not all there is to it. We've covered the representation, we shall now cover the reduction.

Unification

One thing to keep in mind is that theoretically there is no "computation" (as in calculation) involved; what we have (mostly) is a thing we call *unification*. Unification roughly means to check if two terms "match".

All terms are of one of the following forms:

- Variables, which we will denote with words that starts with an uppercase letter;
- Atoms, which we will denote with words that starts with a lowercase letter;
 - (This name, despite being the commonly used name for languages like LISP or Erlang, might come as obscure for other people; a more descriptive name would probably be "scalar values".)
- Functors², which is of the form $f(a_0, \dots, a_n)$, in which f is an atom and all the a are terms (which can themselves be variables or atoms or functors, etc.)

The unification of two terms thus can be said as follows:

- A variable matches with anything (unless it forms recursive reference; see below);
- Atoms only match with themselves, e.g. the atom *apple* would only match with *apple* and will not match with *pear*. (But obviously it would still match with variables, since variables match with anything)
- Functors match if and only if all the following conditions are met:
 - The "head" (or "the functor's name") matches, e.g. f(A, B) matches with f(a, b) but not g(a, b) because the latter is g instead of f.

¹To some people this (and resolution, which we will discuss later) looks like a rewriting system, and some people do claim that Prolog is just a rewriting system with backtracking. As someone who've worked with rewriting systems before, I'm not going to comment on their opinions, but they do be looking similar.

²The name "functor" has been used to refer to a handful of different concepts; in the context of logic programming, especially languages that's derived from Prolog, you can simply think of it as "some data that's shaped like a function call".

- The arity (i.e. the number of arguments) matches, e.g. f(A, B) matches f(a, b) but not f(c, d, e) despite both being f; the former has 2 arguments while the latter has 3. The convention is to denote the arity after the functor's name, e.g. f/2 and f/3.
- All of the arguments match with each other, e.g. f(a, B) matches with f(a, b) and f(a, c) but not with f(b, a) despite both being f/2; the first arguments, namely a and b, does not match.

The action of unification results in a mapping from variables to terms; (one could say that this is how "value assigning" happens, although that really isn't the accurate wording.) A common name for this mapping would be "subst", short for substitution. The observant might want to ask if one can match a variable with a term that contains the same variable either directly or indirectly (e.g. A with f(A)). Theoretically this is not valid because one can never have a fully grounded solution for such a subst, but whether an occurrence of such case will trigger some kind of runtime panic depends on the language and its implementation. Some languages (like Prolog) permits it (even if this kind of situations almost certainly end in a failure state), some languages (like miniKanren) performs strict occur check and does not allow any such occurrences.

Some of you may ask if pattern matching can also be seen as a form of unification. I've met people online who were very pedantic about the name and fervently insisted that pattern matching is not unification; almost all of those kind of people are doing this because they want to be perceived as more sophisticated than they truly are so they can look down upon other people more safely. I would say one can think of it as somewhat similar; one must understand that pattern matching (and the subsequent binding of variables) is a "one way thing", e.g. unifying f(a, J) with f(K, b) would result in a subst of J = b, K = a, but with pattern matching one side is always a fully grounded term, if we can even consider values as terms in this sense.

Resolution

In this section I would only introduce one way of resolution. It is – I believe – conceptually the simplest kind of resolution. There probably are other kind of resolution technique out there, but they are out of scope for this tutorial.

Remember that a goal is of the form $\bot \leftarrow P_0 \land \cdots \land P_n$ (we'll omit the \bot part for brevity from now on) and act as entry points in our interpretation;

naturally, if there's no more P, the goal is empty, and the computation halts.

The simplest method of resolution is thus as follows. For the goal $\leftarrow P_0 \wedge \cdots \wedge P_n$, we take the left-most subterm, in this case P_0 ; we attempt to find among all the previously defined rules and facts the one that can successfully unify with P_0 (instantiated according to the current subst). If we couldn't find one, then the current attempt is considered a failure (what happends after the failure depends; see later section on cut); If we were successful in finding one, then two things happen:

- Because we performed a unification, we now have a subst;
- And because we picked a rule/fact, we now have a "right-hand side", which could be empty or containing other goals.

This subst resulted from the unification is combined with all the subst we've had so far, and the "right-hand side" of the rule/fact we found would be appended to the left-hand side of the current goal.

Let's look at an example. Assume we have the following program in Horn clauses:

```
parent(tom, john) \leftarrow \top
parent(tom, evans) \leftarrow \top
parent(tom, sarah) \leftarrow \top
sibling(A, B) \leftarrow parent(C, A) \land parent(C, B) \land not(equal(A, B))
```

...and the goal we're asking is $\leftarrow sibling(john, A)$; the following things happen:

- sibling(john, A) matches with sibling(A, B), (but we cannot directly say that john = A, A = B, because the two A should clearly mean different things! with a bit of renaming, we shall say the goal is $sibling(john, A_0)$).
- The unification of the step above results in the subst $john = A, A_0 = B$.
- It contains a right-hand side of positive length; we shall add them to the goal with respect of the subst. The goal is now $\leftarrow parent(C, john) \land parent(C, A_0) \land not(equal(john, A_0))$.

- We now try to find a rule/fact that can unify parent(C, john), and we found parent(tom, john). This gives us C = tom. Since it's a fact, there is no right-hand side to be added. The goal is now $\leftarrow parent(tom, A_0) \wedge not(equal(john, A_0))$.
- We now try to find a rule/fact that can unify $parent(tom, A_0)$, and we again found parent(tom, john). This gives us $A_0 = john$.
- The goal, which is now $\leftarrow not(equal(john, john))$ with respect to the current subst, fails. If we have committed to this choice (by cut or by other means), the entire thing fails; but we didn't, thus we backtrack. The goal is now once again $\leftarrow parent(tom, A_0) \land not(equal(john, A_0))$, and we attempt to find the next thing that can unify with $parent(tom, A_0)$.
- We found parent(tom, evans), which results in the subst $A_0 = evans$.
- The goal is now $\leftarrow not(equal(john, evans))$, which would succeed without a subst (or with an empty subst, depending on how you see it), leaving the goal empty. Since the goal is now empty, we consider whatever we have now is a valid solution. The subst for this solution would be $A = john, C = tom, A_0 = evans$, and the solution itself would be sibling(john, evans).
- If more result is requested, the latest unification and its subst would be discarded and a new instance of $parent(tom, A_0)$ would be looked for, which would result in the subst $A_0 = sarah$ and the solution sibling(john, sarah).
- It's easy to see that we can no longer find any new solutions after this point, so the resolution for the goal ends here.

Cut

Cut is mainly a Prolog thing; when triggered, it forces the resolution to commit to all the choices it has made up till that point, artificially cutting the backtracking tree. This is one of the major reasons why Prolog is being considered as "not so pure". Similar constructs exists in other logic programming languages (e.g. Mercury's cc_multi and cc_nondet), albeit they may work in a fundamentally different manner.

Using our program above as an example; if we write the first fact as $parent(tom, john) \leftarrow !$ instead of $parent(tom, john) \leftarrow \top$ (the exclamation mark is the "cut operator" in Prolog), we wouldn't have any solutions, since

by the time we reached the goal $\leftarrow not(equal(john, john))$, we have already committed to the choices of john and discarded other possibilities.

Disjunction

Other than conjunction which can be interpreted as sequencing, one can also use disjunction in Horn clauses, which can be interpreted as branching. The resolution of a disjunct goal naturally depends on the resolution of its subgoals, and any solution is a solution of the overal disjunct goal if it is also a solution of any of its subgoals. The resolution of a disjunct goal starts from one of its subgoals; if that subgoal fails, it moves onto the next subgoal, until any one of these subgoals succeed (which results in an overall success) or all of them failed (which results in an overall failure). The order in which each branch gets resolved theoretically shouldn't matter, but in reality different order would lead to different backtracking tree, which in languages with cut (e.g. Prolog) and similar constructs could lead to order-dependent results.

Unification, part 2

In short, due to how unification works, you can do certain thing "in reverse". For example, assume that we have a predicate append/3 which, at surface, is something you append two lists together:

```
append(nil, nil, nil) \leftarrow \top
append(nil, A, A) \leftarrow \top
append(cons(H, T), A, cons(H, R)) \leftarrow append(T, A, R)
```

If you query goals where the first two arguments are "solid" terms, you will get the result bound to the variable at the third argument; but at the same time, you can:

- Call the predicate with three "solid" lists, with which you can check if appending the first two lists does end up with the same value as the third one. If that is the case, then you'll eventually hit one of the first two facts and have an overall success; if that isn't the case, you'll have a failure.
- Call the predicate with two variables and one "solid" list, with which you have a goal that asks "which two lists can be appended and end up with the same value as this third list".

For the second one, we'll discuss the case append(A, B, cons(c, nil)) here; you can try to resolve a few examples on your own as an exercise. In the case of append(A, B, cons(c, nil)), the resolution would first match with the second fact (which will give A = nil, B = cons(c, nil)) and then it would match with the third rule (which, if you work it out, will give A = cons(c, nil), B = nil.)

2.2 Writing the program

In this part of the chapter, we will trying to write a program which I called rall, which stands for "adding carraige return to all lines". From this "full name" you can tell it is a very small utility that converts Unix-style newlines (LF) to Windows-style newlines (CRLF). It covers some of the most basic things when it comes to software development, and I reckon it would serve as a nice beginner (in terms of language, not programming in general) project.

File name / module name

Each Mercury source file would be a module, whose name should be the same as the file name (without the extension name part).³ Each module would need to have at least one interface section and one implementation section.

For our program, we would save the file under the name rall.m, thus the module name would be rall.

```
:- module rall.
:- interface.
:- implementation.
```

Entry point

With Mercury, the entry point of the program (e.g. main in C) must be defined as a main/2, i.e. main with 2 parameters. If you attempt to define main as anything else, you would get a compile error:⁴

 $^{^{3}}$ It's possible to have the module name *not* to be the same as the file name, but it's too early to talk about it here. See Chapter 3.

⁴As you may have noticed, this is technically a *linking error* instead of a *compile error*, which means that the compiling *itself* did not generate any error. main with other arities does not get special treatment like main/2 do, and some people do adopt the pattern of having both a main/2 and a different main with a different arity.

```
% mmc ./rall.m
/usr/bin/ld: rall_init.o: in function `mercury_init':
rall_init.c:(.text+0x4a0): undefined reference to `predicate '
main'/2 mode 0>'
collect2: error: ld returned 1 exit status
```

The two parameters of this main/2 must be of type io.state. If you don't define it as io.state, you would get a compile error:

```
./rall.m:005: Error: `main'/2 must have mode `(di, uo)'.
./rall.m:005: Error: both arguments of `main/2' must have type `io .state'.
```

To be able to use io.state, we should first import the io module, which is done with the import_module declaration.

So we now have the following:

```
:- interface.

2  % NOTE: to use io.state we must import the io module.

3  :- import_module io.

4  :- pred main(io.state, io.state).
```

You might wonder what exactly is io.state and why main needs two of them; we will explain this later. At the very least, these are **not** the same as argc and argv which you would see in C. There are dedicated predicates for retrieving command line arguments; we will get to that later.

The (di, uo) part needs explanation. The message says it's a mode. What is a mode? Modes are things that describe the change of instantiatedness before and after the execution of a goal. Instantiatedness – that's a long word. What is instantiatedness? Roughly speaking, it's a concept for describing whether a slot is free or bound. (In Mercury, instantiatedness is actually a tree, because obviously you'd have – or at the very least you could manually construct – terms that are only bounded at certain parts.) Two kinds of basic instantiatedness exists in Mercury: free, which refers to variables that are not bound by any values; bound, which means that the term is not a variable but rather some concrete term at least at that level. Consider the term blah(A, b); it's bound at the top (with blah) and at one of the children node (at b) but free at another children node (at A, provided that A does not have a value already). A mode, ground, also exists, and refers to the terms that are completely bound (e.g. blah(a, b) is ground but blah(A, b) isn't, provided that A is still free.)

With this knowledge, we should first understand what in and out modes actually are. If you have worked with some other languages you might have seen thing similar to this before: C libraries often ask you to pass in a reference for retrieving the actual result because the return value is used for returning error status; you can think of it as "in" (inputting) parameters

and "out" (returning) parameters; some languages (e.g. Ada) even explicitly label them as such. I do not wish to introduce you to wrong analogies that will become detrimental for your future learning, but I have to say they do be somewhat similar.

That said, in and out is properly defined in Mercury, as follows:

```
:- mode in == ground >> ground.
:- mode out == free >> ground.
```

This largely fits with our intuition about in and out parameters: we expect in arguments to be fully grounded and they stay grounded, and we expect out arguments are variables which we will provide a solid term to (thus making it grounded).

There are also "polymorphic" version of in and out, defined as follows:

```
:- mode in(Inst) == Inst >> Inst.
:- mode out(Inst) == free >> Inst.
```

You can indeed define your own instantiatedness, with which then you can use these definitions to get the in and out version of it; but we will not need this for our example this time.

Now we can finally come back to di and uo. di stands for "destructive input", and uo stands for "unique output". If you have used Rust (or, in the case where you are really adventurous, Clean) before, you might have a vague idea of what this is. In Mercury, there are two special instantiatedness named unique and dead, the former conceptually refers to values that can only have one reference at all time, and the latter refers to reference that are once "unique" but is now "dead".

```
% unique output - used to create a "unique" value
c:- mode uo == free >> unique.

% unique input - used to inspect a unique value without causing
% reference to become dead
c:- mode ui == unique >> unique.

% destructive input - used to deallocate or reuse the memory
% occupied by a value that will not be used.
c:- mode di == unique >> dead.
```

Up to now, our code would be something that's similar to this:

```
:- module mercury_rall.

:- interface.

:- import_module io.
:- pred main(io.state, io.state).

:- implementation.
```

If you compile this, the Mercury compiler would complain about not having a mode declaration. For this reason, we will add the following line and compile:

```
:- mode main(di, uo).
```

But this time we would see the compiler complain about different things:

The compiler expects the second argument would be supplied with a free argument and that argument should become a unique value at the end of main! How do we do such a thing? Fortunately the io module has what we need to stuff the body of main, and we will use one of them:

```
main(In, Out) :-
io.write_string("blah", In, Out).
```

But this time the compiler started complaining about other things:

```
./rall.m:007: Error: no determinism declaration for exported
predicate
./rall.m:007: `main'/2.
```

Determinism in this case, informally speaking, refers to "how a certain thing would succeed/fail". It's not something we'd care about in other languages, at least not in an explicit, supported-by-the-language-itself manner; (we normally only talk about when programs terminate at a certain state or not.) In Mercury, we have the following determinism categories:

- Deterministic (det): guaranteed to have one and exactly one solution.
- Semideterministic (semidet): have exactly one solution, but does not guarantee to produce it.
- Multisolution (multi): guaranteed to have a solution among possibly many solutions.
- Nondeterministic (nondet): have possibly many solutions, does not guarantee to produce one.

- Failure (failure): cases where there's zero solutions. They are not actual errors but a part of the logic (e.g. arity mismatch during unification, which will never produce a solution because the arity is different).
- Errorneous (errorneous): also have zero solutions, but they do represent actual errors which in other languages would be represented in the form of runtime exception throw or panic.

Some people might not understand why det and semidet are separate things. Imagine a function that takes the "head" of a linked list; this function is obviously only defined on non-empty list and not defined on empty lists, so for any given list there's either only one solution or no solution, thus semidet instead of det.

The difference between errorneous and failure might be clearer if I explain their behaviours when it comes to negation. Basically in Mercury (and other logic programming languages), you can take a predicate, put a negation on it, and then ask for a case where it does not succeed. Naturally, the negation of determinisms that are guaranteed to produce a result like det and multi would be failure (i.e. negating a definite success would be a failure), the negation of failure would be det (i.e. negating a definite failure would be a definite success), and the negation of semidet and nondet would be semidet (i.e. negating these cases turns it into a case that asks if the match would be successful or not); but negating an errorneous would only produce an errorneous.

(If you're going to ask this question – yes, solving the problem of perfect determinism inference does mean solving the halting problem, and is thus undecidable.)

Other than these six categories there are also this thing called the "committed choice nondeterminism", which adds two more determinism class: cc_multi and cc_nondet. The difference between committed choice nondeterminism and "normal" nondeterminism is that normal nondeterminism supports backtracking while CC nondeterminism, despite potentially having more than one solutions, commits to only one of them and thus does not backtrack. The main entry point can be defined to be det or cc_multi, since both of them guarantee one and only one solution (det, of course, is more strict than cc_multi, and if the Mercury compiler can determine something that should be able to be a det got labelled as a cc_multi, it would spit out a warning saying you could've gone with the stricter option.)

In this case, we should add the string is det at the end of our mode declaration, so that the whole line would be :- mode main(di, uo) is

det.. After this modification, the compiler should have finally stopped complaining and gives you an executable; when you run it, it would display a string blah, which should be obvious. At this point, the code shall look like something like this:

```
:- module rall.
3
       :- interface.
4
       :- import_module io.
5
6
         pred main(io.state, io.state).
7
       :- mode main(di, uo) is det.
9
      % also: you can combine the `pred` and the `mode` line into one
      like this:
10
11
             :- pred main(io.state::di, io.state::uo) is det.
12
       :- implementation.
13
14
      main(In, Out) :-
15
           io.write_string("blah", In, Out).
16
```

Now, if you attempt to write multiple strings like this:

```
main(In, Out):-
io.write_string("blah", In, Out),
io.write_string("blah", In, Out).
```

The compiler would produce this error:

```
./rall.m:013: In clause for `main(di, uo)':
./rall.m:013: in argument 2 of call to predicate `io.
write_string'/3:
./rall.m:013: unique-mode error: the called procedure would clobber its
./rall.m:013: argument, but variable `In' is still live.
For more information, recompile with `-E'.
```

If we look up the declaration of io.write string:

```
:- pred write_string(string::in, io::di, io::uo) is det.
```

You should already know that di specifies a turn of a unique value into a dead value, so In after the first write_string would be considered dead and not unique, which does not fit the requirement of the di of the second write_string. If you try to experiment and do this:

```
main(In, Out) :-
io.write_string("blah", In, Out),
io.write_string("blah", Out, Out2).
```

This would have the following violation: since we declared Out to be uo which is free >> unique, Out must be unique at the end of main, but the second write_string turns Out into a dead because that slot is di which is unique >> dead. A solution to this would be to do this:

```
main(In, Out) :-
io.write_string("blah", In, Out1),
io.write_string("blah", Out1, Out).
```

This would compile, and the generated executable would write "blah" twice, as expected.

To keep coming up with new variable name is tedious if this gets long. To solve this problem, Mercury have something called the "state variable". With state variables, the example above can be written like this:

```
main(!.I0, !:I0):-
io.write_string("blah", !.I0, !:I0),
io.write_string("blah", !.I0, !:I0).
```

!.IO refers to the value bound to IO at the current moment, and !:IO makes the value at that slot bound to IO; so in the first write_string, the current value IO was destroyed, and the value bound to Out1 in our non-state-variable version was bound to IO as its next value; and in the second write_string, the value of Out1, which itself has become the current value of IO and being referred to with !.IO, was destroyed, and the value bound to Out in our non-state-variable version was bound to IO as its next value, and subsequently became the output of main.

You can write !.IO, !:IO as !IO, because the former is also quite tedious as well:

```
main(!IO) :-
io.write_string("blah", !IO),
io.write_string("blah", !IO).
```

It would still feel a bit tedious for people who are used to other programming languages, but I suppose this is what you give up for being as explicit as possible for the sake of improved correctness and things...

Command line arguments

The predicate we need for retrieving the command line arguments is also in the io module:

This would bound the actual value to Argv. From the definition of command_line_arguments we can know that the type of Argv would be a list(string) – a list of strings. It would be necessary to import the list module from now on. This value does not contain the name of the program

(for which io module has other predicates to retrieve). It would also be useful to check if the length of Argv is exactly 2:

```
1 :- import_module list.
2 %
3 main(!IO) :-
       io.command_line_arguments(Argv, !IO),
4
       ( if length(Argv) \= 2
5
6
         then (
7
             % you can specify which stream you are writing to btw
8
             io.write_string(
9
                  io.stderr_stream,
10
                  "usage: rall [inputfilename] [outputfilename] \n",
11
12
         )
         else (
13
14
             % ...
15
         )
16
      ).
```

(Note that if you have previous experience with Prolog, the condition for the if construct might be surprising, since in Prolog, = unify things structurally; for example, the goal 3 + 4 = 7 would fail in Prolog despite 3 plus 4 do equal to 7, since 3 + 4 is a functor (interpreted as `+`(3,4)) and 7 is an atom; this is different in Mercury, where things would be evaluated. Also, length in Mercury is defined both as a predicate and as a function, and in this case the function got evaluated as it's defined in the standard library resulting an actual number.

The then clause is simple – we write the usage string and call it done:

```
:- import_module list.
2 %
3
  main(!IO) :-
4
       io.command_line_arguments(Argv, !IO),
5
       ( if length(Argv) \= 2
6
         then (
7
             % you can specify which stream you are writing to btw
8
             io.write_string(
9
                  io.stderr_stream,
                  "usage: rall [inputfilename] [outputfilename]\n",
10
11
                  !IO)
12
         )
13
         else (
             % ...
14
15
         )
      ).
16
```

Now we extract the command line arguments. list module contains index0 and index1, which are predicates/functions that uses 0-based index and 1-based index respectively; but these aren't the ones we would choose to use here because they are semidet. (Exercise: can you figure out the reason why they're semidet instead of det on your own?) We use their

det counterparts: det_index0 and det_index1, which will throw runtime exceptions when failing:

```
:- import_module list.
2
3 main(!IO) :-
      io.command_line_arguments(Argv, !IO),
5
       ( if length(Argv) \= 2
6
         then (
             io.write_string(
8
                 io.stderr_stream,
                 "usage: rall [inputfilename] [outputfilename] \n",
9
10
         )
11
         else (
12
             InputFilePath = det_index0(Argv, 0),
13
             OutputFilePath = det_index0(Argv, 1)
14
             rall(InputFilePath, OutputFilePath, !IO)
15
16
17
      ).
18
```

By the way, if we import module simply by doing :- import_module name, We don't need to fully specify the module name, as long as there's no ambiguity. Naturally, the good practice for a lot of common cases is to fully specify; but some editors (Emacs, for example) might implement some kind of "smart indent" algorithm which could put the cursor at undesirable positions and be a nuisance. In cases like this, you might find writing code this way to be more comfortable.

It would be also be a good time to write our first predicate that is not the entry point...

Opening & closing files for input/output

In the last code snippet we've write rall(InputFilePath, OutputFilePath, !IO). Despite this looks like rall/3, but it's secretly rall/4, and we should write our declaration as such:

```
:- pred rall(string, string, io.state, io.state).
:- mode rall(in, in, di, uo) is det.
```

io module has open_input/4 and open_output/4, both of which are det:

```
rall(InputFilePath, OutputFilePath, !IO) :-
io.open_input(InputFilePath, RIn, !IO),

% ...
```

The declaration of open input is as follows:

```
:- pred open_input(
    string::in, io.res(io.text_input_stream)::out,
```

```
io::di, io::uo) is det.
```

We know that whatever variable we put at the second slot would be the input stream we want (or in the case of error, whatever error we could catch). Its value is of type io.res, whose definition is as follows:

```
1 :- type res(T)
2 ---> ok(T)
3 ; error(io.error).
```

This is how Mercury defines algebraic datatypes. If you have previous experience with Haskell, OCaml, Rust or similar languages, this should be natural for you.

Pattern matching is quite simple – you just do Var = ok(OkValue) and Var = error(ErrorValue) like this:

Since we combined them with a disjunction, when the unification in the first branch (in this case, Var = ok(OkValue)) fails, the program flow would go to the second branch, and if the unification there fails (and there's a third branch), it would go to the third clause, etc.; and if all branches fail, the whole thing fails (and of course we'd expect the Mercury compiler gives us a determinism verdict of something like nondet, but I digress). We'll handle all the cases a value of type res(T) could have, and if we do that, the compiler should recognize that we'll not fail as long as both of our ok clause and error clause don't have a determinism verdict that implies a possibility of failing.

The definition of io.error is as follows:

```
:- type io.error.  % Use error_message to decode it.
```

This type is opaque; we need error_message to handle it. Its definition is as follows:

```
% Look up the error message corresponding to a particular error
code.

code.

func error_message(io.error) = string.

pred error_message(io.error::in, string::out) is det.
```

Now we know we can obtain a string from an io.error; we can decide either to throw (which would cause the program to exit prematurely) or to write it to standard error (and try to exit afterwards). In this example we use throw because it's simpler:

```
% throw/1 requires the exception module.
continuous in import_module exception.

% ...
continuous throw(error_message(ErrorValue)));
```

Before we move on with it, I must also explain something else. This isn't the case in Prolog, but in Mercury calls are "curried". Using the declaration of error_message above as an example, the term error_message(Var) matches both func error_message/1 and pred error_message/2, and would thus be judged as having the type string from func error_message/1 and the type pred(string) from pred error_message/2 (which is a pred that takes a string as its argument) at the same time, i.e. a type ambiguity. In Prolog it will only match error_message/2, but in Mercury this is how it is...

To resolve this you need to explicitly state the type you want by writing :{type}; in this case, we'll choose the func one and specify it as string, just like this:

```
1  // here, the ":string" part. ---v
2  (RIn = error(ErrorValue), throw(error_message(ErrorValue):string)
);
```

If you don't do this, you'll get a compile error like this:

```
./rall.m:025: In clause for predicate main'/2:
 2
       ./rall.m:025:
                       warning: variable OutputFilePath' occurs only once
       ./rall.m:025:
                       this scope.
3
       ./rall.m:031: In clause for predicate rall'/3:
 4
       ./rall.m:031:
                       error: ambiguous overloading causes type ambiguity
5
      ./rall.m:031:
                       Possible type assignments include:
6
       ./rall.m:031:
7
                       V_20:
8
       ./rall.m:031:
                         pred(
9
       ./rall.m:031:
                           string
                         ۱ (
10
       ./rall.m:031:
       ./rall.m:031:
11
                       or
12
       ./rall.m:031:
                         string'
       ./rall.m:031:
                       You will need to add an explicit type
13
      qualification to
       ./rall.m:031:
                       resolve the type ambiguity. The way to add an
14
      explicit
                       type qualification is to use "with_type". For
       ./rall.m:031:
15
      details see
                       the "Explicit type qualification" sub-section of
       ./rall.m:031:
      the
       ./rall.m:031:
                       "Data-terms" section of the "Syntax" chapter of
17
      the
       ./rall.m:031:
                       Mercury language reference manual.
```

Now we can have the following code. Opening an output stream is a near-identical process:

```
1 % throw/1 requires module `exception`.
2 :- import_module exception.
3 % ...
5 :- pred rall(string, string, io.state, io.state).
6 :- mode rall(in, in, di, uo) is det.
7 rall(InputFilePath, OutputFilePath, !IO) :-
       open_input(InputFilePath, RIn, !IO),
8
9
           ( RIn = error(ErrorValue), throw(error_message(ErrorValue):
      string));
10
           ( RIn = ok(InputStream),
              open_output(OutputFilePath, ROut, !IO),
11
                 ( ROut = error(ErrorValue), throw(error_message(
12
      ErrorValue):string) );
13
                  ( ROut = ok(OutputStream),
14
                    process_stream(InputStream, OutputStream, !IO)
15
16
17
             close_output(OutputStream, !IO),
              close_input(InputStream, !IO)
18
19
20
       ).
21
```

Actually processing the data

We'll do it as follows: each time we read a single character from the input stream, we check if it's a line feed; if it is, we write a carriage return and a line feed to the output stream; or else, we simply write that character to the output stream; we repeat this until we've hit the EOF of the input stream. read_char produces a different kind of result named result, which has a separate constructor eof. You should know how to handle this by now...

Character literals are enclosed with single quotes, by the way:

```
:- pred process_stream(
        io.text_input_stream,
3
        io.text_output_stream,
        io.state, io.state).
  :- mode process_stream(in, in, di, uo) is det.
  process_stream(InStream, OutStream, !IO) :-
      read_char(InStream, InRes, !IO),
8
          ( InRes = eof );
9
          ( InRes = error(Error), throw(error_message(Error):string) );
          ( InRes = ok(Char),
10
11
            ( if Char = ' n'
12
                  write_char(OutStream, '\r', !IO),
13
                  write_char(OutStream, '\n', !IO)
14
              )
15
16
              else (
17
                  write_char(OutStream, Char, !IO)
18
```

```
process_stream(InStream, OutStream, !IO)

1 )
22 ).
23
```

Full program listing

The following code is the listing of the completed rall program:

```
:- module rall.
2 :- interface.
3 :- import_module io.
5 :- pred main(io.state, io.state).
6 :- mode main(di, uo) is det.
8 :- implementation.
10 :- import_module list, exception.
11
12 main(!IO) :-
       io.command_line_arguments(Argv, !IO),
13
14
       ( if length(Argv) \= 2
15
         then (
16
              io.write_string(
                  io.stderr_stream,
17
18
                   "usage: rall [inputfilename] [outputfilename] \n",
19
20
21
          else (
              InputFilePath = det_index0(Argv, 0),
OutputFilePath = det_index0(Argv, 1)
22
23
              rall(InputFilePath, OutputFilePath, !IO)
24
25
         )
26
       ).
27
28 :- pred rall(string, string, io.state, io.state).
29:- mode rall(in, in, di, uo) is det.
30 rall(InputFilePath, OutputFilePath, !IO) :-
       open_input(InputFilePath, RIn, !IO),
(    (RIn = error(ErrorValue), throw(error_message(ErrorValue):
31
32
       string));
33
            ( RIn = ok(InputStream),
34
              open_output(OutputFilePath, ROut, !IO),
35
                ( ROut = error(ErrorValue), throw(error_message(
       ErrorValue):string) );
36
                  ( ROut = ok(OutputStream),
                     process_stream(InputStream, OutputStream, !IO)
37
38
39
              close_output(OutputStream, !IO),
40
41
              close_input(InputStream, !IO)
42
43
       ).
44
45 :- pred process_stream(
```

```
io.text_input_stream,
         io.text_output_stream,
47
48
         {\tt io.state}, {\tt io.state}).
49 :- mode process_stream(in, in, di, uo) is det.
50 process_stream(InStream, OutStream, !IO) :-
       read_char(InStream, InRes, !IO),
       ( ( InRes = eof );
52
           ( InRes = error(Error), throw(error_message(Error):string) );
53
           ( InRes = ok(Char),
54
55
             ( if Char = ' \n'
               then (
56
                    write_char(OutStream, '\r', !IO),
write_char(OutStream, '\n', !IO)
57
58
59
               )
60
               else (
                    write_char(OutStream, Char, !IO)
61
62
63
             ),
            process_stream(InStream, OutStream, !IO)
64
65
       ).
```

Listing 2.1: Full code for rall.m

It might not be idiomatic Mercury code, but at least it compiles and runs.

We can compile it and produce an executable by calling the command \mbox{mmc} rall.m.

Programming in Mercury

3.1 Trace Goals

Before we continue we must talk about how to explore things on one's own. A common strategy of debugging in conventional languages would be to insert printing statements at places where one would expect the program to reach with its state containing certain values; this strategy doesn't work well with Mercury, since to perform IO in Mercury you would need to do it through io.state, which naturally has the implication of spreading out and forcing everything that uses the predicate to adopt two extra io.state parameters.¹

For example, assume we have the following program that prints out the factorial of 5:

```
1 :- module ex01.
2 :- interface.
3 :- import_module io.
4 :- pred main(io.state::di, io.state::uo) is det.
5 :- implementation.
7 :- import_module int.
9 :- pred fac(int, int).
10 :- mode fac(in, out) is det.
11 fac(N, M) :-
      (if (N = 0) then (M = 1)
        else if ( N = 1 ) then ( M = 1 )
13
        else if ( N < 0 ) then ( M = 1 )
14
        else ( fac(N-1, M1), M = N * M1)
15
16
```

¹Which is not unlike the famous "color of function" problem in languages with async (e.g. JavaScript)

```
17 .
18
19 main(!IO) :-
20     io.write_string("Hello, world!\n", !IO),
21     fac(5, M),
22     io.write_int(M, !IO).
```

Let's say we want to add a printing goal in fac. Printing, naturally, requires two io.state just like we've been mentioning. If we do it by adding io.state, we need to perform the following changes:

- Change the pred line into :- pred fac(int, int, io.state, io.state);
- Change the mode line in a similar fashion;
- Change every occurrence of fac accordingly;
- Since for every call site of fac we now need to pass two extra io.state arguments, the environment surrounding the call site should have those extra arguments ready;
 - and for some predicates this could mean adding extra parameters just like what we've done to fac;
 - and now the call sites of those predicates also need to have their signature changed, and this change would also affect call sites that calls them, and the need of change would thus propagate upwards...
- ...and after the debugging is done, we need to change them back.

This kind of coding process not only goes against the spirit of Mercury but is also just extremely tedious. Luckily, for this Mercury prepared a feature called **trace goals**.

Trace goals in Mercury are of the format trace [param1, ...] Goal. In such a goal, all variables bound in the parameter list part is available within the Goal part. There are many different parameters that one can use, but the most common is probably going to be io; by having io(!IO) which bound a pair of io.state to the state variable of your choice, you can use it to display debug messages in Goal freely without having to do the changes listed above. For example, assume we want our program to display the value of each call to fac before they return, we can add a trace goal that introduces a pair of io.state arguments like this:

```
1 :- module ex01.
2 :- interface.
3 :- import_module io.
```

```
4 :- pred main(io.state::di, io.state::uo) is det.
5:- implementation.
  :- import_module int.
9 :- pred fac(int, int).
10 :- mode fac(in, out) is det.
11 fac(N, M) :-
      (if (N = 0) then (M = 1)
12
         else if (N = 1) then (M = 1)
         else if ( N < 0 ) then ( M = 1 )
14
15
        else ( fac(N-1, M1), M = N * M1 )
16
17
      trace [ io(!IO) ] (
18
          io.write_string("fac(", !IO),
          io.write_int(N, !IO),
19
           io.write_string(") = ", !IO),
20
          io.write_int(M, !IO),
21
22
          io.write_string("\n", !I0)
      )
23
24
25
26 main(!IO) :-
      io.write_string("Hello, world!\n", !IO),
27
28
      fac(5, M),
29
      io.write_int(M, !IO).
```

Notice that without changing the signatures of fac we can still obtain io.state and display things.

3.2 Using the standard library

As you read through the documentation of Mercury's standard libraries², you'll see the notice saying that "we recommend that if possible you use only those modules whose stability is described as either 'medium to high' or 'high", you'll see a stability declaration in each of the documentation page, and you'll see that certain modules that should be stable, like int and uint, has a stability of low; should you rely on these modules that are of low stability? Sometimes the answer is yes – these modules have reached high stability in the latest nightly, which means that the interface of these modules, as it's described in the documentation of latest nightly, shouldn't change so much that it breaks basic programs in the future, and thus should be safe to use. In general, my recommendation is to read the

²Available at https://mercurylang.org/information/doc-release/mercury_library/index.html and https://mercurylang.org/information/doc-latest/mercury_library_manual/index.html, for latest stable and latest nightly, respectively.

documentation for the latest nightly when you're not sure if you can still use the same library predicate when the next major release come out.

3.3 Basic Datatypes

Finally, after all that build-up, we can look at the basic datatypes.

Signed and unsigned integers

In Mercury, the types of signed integers are: int, int8, int16, int32 and int64, and the types of unsigned integers are: uint, uint8, uint16, uint32 and uint64. The types with a number suffix are exactly what you think they are, e.g. int8 means an 8-bit signed integer, etc.. The length of int is said to be "implementation-defined" but at least 32 bits. If you want to know how long an int actually is, you can have the following program:

As of now (October 2025), to enable the arithmetic and comparing operations of these types you must import the corresponding module, e.g. to be able to compare uint you must import the uint module. Actually, as a rule of thumb, if you need to do stuff with any of these types (and the basic types described in sections below), it's good measure to import the corresponding module even if you ended up not needing them.

Syntax of integers

A decimal integer is any sequence of decimal digits. Unlike in some languages, having the prefix 0 doesn't make it octal. In Mercury, the corresponding prefixes for non-decimal integers are: 0b for binary, 0o for octal, and 0x for hexadecimal. Mercury also supports a prefix of 0' followed by

any single character; this would result in the character code that character has. (Note that the character immediately after the single quote is considered part of the integer literal regardless of its common usage in other parts of Mercury code; e.g. to get the character code of the character , you write O', with no extra measure of escaping. This works for Unicode characters as well; see the example below.)

A suffix may also be present, with different suffixes and their meanings listed as follows:

Suffix	Type
i or no suffix	Signed (int)
i8	Signed 8-bit (int8)
i16	Signed 16-bit (int16)
i32	Signed 32-bit (int32)
i64	Signed 64-bit (int64)
u	Unsigned (uint)
u8	Unsigned 8-bit (uint8)
u16	Unsigned 16-bit (uint16)
u32	Unsigned 32-bit (uint32)
u64	Unsigned 64-bit (uint64)

An arbitrary number of underscore _ can be inserted between the radix prefix and the first digit (e.g. $0x_123$), between the digits (e.g. $0b_0001_0011$, and between the last digit and the type suffix (e.g. $0b_10010011_u8$. Inserting underscores do not affect the actual value.

An example program showcasing the syntax for integers is listed as follows.³

```
:- module ex01.
3 :- interface.
5 :- use_module io.
6 :- pred main(io.state::di, io.state::uo) is det.
8 :- implementation.
9 :- import_module string.
10 :- import_module int, int8, int16, int32, int64.
11 :- import_module uint, uint8, uint16, uint32, uint64.
12
13 main(!IO) :-
      io.write_string(string.int_to_string(32i) ++ "\n", !IO),
14
      io.write_string(string.int8_to_string(32i8) ++ "\n", !IO),
15
      io.write_string(string.int16_to_string(32i16) ++ "\n", !IO),
16
      io.write_string(string.int32_to_string(32i32) ++ "\n", !IO),
17
18
      io.write_string(string.int64_to_string(32i64) ++ "\n", !IO),
```

 $^{^3\}mathrm{As}$ you can see, sometimes some lexers cannot properly handle valid Mercury code...

```
io.write_string(string.uint_to_string(32u) ++ "\n", !IO),
       io.write_string(string.uint8_to_string(32u8) ++ "\n", !I0),
20
21
       io.write_string(string.uint16_to_string(32u16) ++ "\n", !I0),
       io.write_string(string.uint32_to_string(32u32) ++ "\n", !IO), io.write_string(string.uint64_to_string(32u64) ++ "\n", !IO),
22
23
24
       io.write_string("\n", !IO),
25
       io.write_string(string.uint_to_string(0x_123u) ++ "\n", !IO),
26
       io.write_string(string.uint_to_string(0b_0001_0011u) ++ "\n", !IO)
       io.write_string(string.uint8_to_string(0b0001_0011_u8) ++ "\n", !
27
       io.write_string(string.int_to_string(0'K) ++ "\n", !I0),
       io.write_string(string.int_to_string(0') ++ "\n", !IO),
29
30
       io.write_string(string.int_to_string(0',) ++ "\n", !I0),
       io.write_string(string.int_to_string(0':) ++ "\n", !IO).
```

Floating-point numbers

In Mercury, there is only one type of floating-point number: float. As of now (October 2025; MMC 22.01.8), the default size of this type is double-precision (i.e. 64 bits). Single-precision can be enabled by using the .spf grade suffix, which means to use the command line argument --grade asm_fast.gc.spf or --grade hlc.gc.spf or others while compiling, depending on which backend you want to use. .spf isn't always available; if your call to mmc results in an error like this one:

```
1 mercury_compile: error: the Mercury standard library cannot be found
   in grade
2 `asm_fast.gc.spf'.
```

It means that this particular configuration of backend isn't installed and is thus unavailable. This is likely to be the case, since the default configuration does not include .spf grades.

Character and string

In Mercury, the character type is char and the string type is string.

Syntax of string

Strings in Mercury are enclosed in double-quotes ". A few escape sequences are supported:

Escape sequence	Meaning	Example
\\	Backslash \	\\
\'	Single quote	\'
\"	Double quote	\"
\a	Beep	\a
\b	Backspace	\b
\r	Carriage return	\r
\f	Form feed	\f
\t	Tab	\t
\n	Newline	\n
\v	Vertical tab	\v
/XXX/	Hexadecimal escape	\x37\
\XXX\	Octal escape	\037\
\uXXXX	Unicode	\u3707
\UXXXXXXXX	Unicode	\U00003707

The length of a hexadecimal or an octal escape sequence is not predetermined (viz. there can be 2, 3, 4 or potentially more hexadecimal/octal digits that comes after a $\x/\$ prefix); but if the escaped value turned out to be invalid (e.g. a very big number that goes over the Unicode range), a compile error would occur.

The ending backslash for a hexadecimal/octal escape sequence is necessary. For example, this small program:

...prints two lines of text: $K\$ and $K\$ n, with the second line *not* ending in a newline character. This is because the first string is being read as the combination of $\x4b\$, $\$ and $\$ n, but the second string is being read as the combination of $\x4b\$, $\$ and $\$ n.

Other than \", you can also use "" to represent one single double-quote character within a string.

Multi-line string literals are also possible by having a single backslash at the end of the line; the backslash character and the newline character after it would be ignored. Unlike some languages, you need to insert newline manually; for example, the first call of the following would only produce *one* single line of output:

```
% ...
io__write_string("a\
2
3
      multi\
4
          line\
             string\
5
6
                  literal\
  ", !IO),
      io__write_string("a\n\
      multi\n\
9
       line\n\
10
              string\n\
11
                  literal\n\
12
13 ", !IO),
  % ...
```

Syntax of character

The syntax of character is slightly trickier to describe. In many languages, characters are quoted with single-quote '; but per Prolog tradition, single-quotes are also for atoms (or *names*, as defined in Mercury's language manual), which may contain more than one character. Currently, it seems like that Mercury treats quoted names containing one single character (or escape sequence) as character literals under the right context.

One needs to be careful when using quoted atoms when they intend to write character literals: if a character is an operator, simply quoting it isn't enough, since Mercury considers single-quoted operators are still operators themselves. To prevent such cases from being interpreted as an operation of values, one should surround the quoted name with parentheses. For example, Char = '+' is syntatically invalid, since it's being interpreted as Char = +; one should write Char = ('+') in this case.

Tuples

Mercury supports using N-tuple for any N larger than or equal to zero. In Mercury, tuples are enclosed with braces {} both for types and for values. See the example below:

```
1 :- module ex01.
2
3 :- interface.
4
5 :- use_module io.
6 :- pred main(io.state::di, io.state::uo) is det.
7
8 :- implementation.
9 :- import_module int, string.
```

```
11 :- pred f1({int, string}::in, {string, int}::out) is det.
12 f1({X, Y}, {Y, X}).
13
14 main(!IO) :-
15
      % bind X to a tuple value.
      X = \{3, "abc"\},
16
17
      % get a reversed tuple w/ f1 (defined above).
18
19
      f1(X, Reversed_X),
20
21
       % break the reversed tuple down into different values.
      {MyString, MyInteger} = Reversed_X,
22
      io.write_string(string.int_to_string(MyInteger) ++ ", " ++
      MyString ++ "\n", !IO).
```

Maybe

The maybe datatype

Universal type

3.4 If-Then-Else

There are two syntaxes for if-then-else:

```
1 % syntax 1:
2 if [condition]
3 then [then-clause]
4 else [else-clause]
5
6 % syntax 2:
7 [condition]
8 -> [then-clause]
9 ; [else-clause]
```

The second one is a carry-over from Prolog.

3.5 Defining functions

By now we have all seen how to define *predicates*: you use the :- pred declaration like this:

```
:- pred pred_name(arg1_type, arg2_type, ..., argn_type).
2 :- mode pred_name(arg1_mode, arg2_mode, ..., argn_mode) is pred_det.
```

For example, we have defined a few predicates in Chapter 2 like this:

```
1 % "string" and "io.state" are types.
2 % "in", "di" and "uo" are modes.
```

```
3 % "det" is determinism.
4 :- pred rall(string, string, io.state, io.state).
5 :- mode rall(in, in, di, uo) is det.
```

Sometimes people declare the type signature and the mode signature separately, because in logic programming predicates can work "in reverse" and thus would require multiple mode signature. Take the signatures of the predicate list.append for example:

```
:- pred append(list(T), list(T), list(T)).

:- mode append(di, di, uo) is det.

:- mode append(in, in, out) is det.

:- mode append(in, in, in) is semidet.  % implied

:- mode append(in, out, in) is semidet.

:- mode append(out, out, in) is multi.
```

To explain this line by line:

- list(T) is a type, where T is a type variable, declaring that it works for any list.
- The first mode is simply a version of the second one that works on unique inputs.
- The second mode is for appending two lists together into a third list.
- The third mode is to check if combining the first two arguments result in the same value as the third argument. The check could succeed or fail, thus having the overall determinism of semidet.
- The fourth mode is to remove the "prefix" (described by the first argument) from the third argument. The third argument might not contain the first one as a prefix, thus having the overall determinism of semidet.
- The fifth mode asks the question of "which two lists can be combined and create the third one". There obviously must be at least two solutions (exercise: can you figure out why this is the case?) and there could be even more for some input, thus having the overall determinism of multi.

```
1 :- func f_name(arg1type, arg2type, ...) = result_type.
2 :- mode f_name(arg1mode, arg2mode, ...) = result_mode is f_det.
```

For example, a factorial function might have the following declaration:

```
1 :- func factorial(int) = int.
2 :- mode factorial(in) = out is det.
```

You also need to write f_name(args) = result instead of f_name(args, result) when providing the definition; other than that, the rest is mostly the same. Compare the following two definitions of the factorial function, one defined as a predicate with only one mode, one defined as a function:

```
:- pred factorial1(int, int).
2 :- mode factorial1(in, out) is det.
3 factorial1(X, Y) :
      (X = < 1 -> Y = 1
      ; ( factorial1(X - 1, Y0),
6
          Y = X * YO
      ).
8
10 :- func factorial2(int) = int.
11 :- mode factorial2(in) = out is det.
12 factorial2(X) = Y :-
13
  (X = < 1 -> Y = 1
    ; Y = X * factorial2(X - 1)
14
15
```

Just like predicate declaration, you can combine the type signature and the mode signature in one; and if you do, you need to wrap the return type in parentheses, like this:

```
1 :- func factorial2(int::in) = (int::out) is det.
```

You can technically also declare functions (and predicates) like this, but this is technically not the "proper" way. The semantics of this, is to declare the function as a whole is of some higher-order function/predicate type.

```
:- func factorial2 : (func(int) = int).
```

A function, just like a predicate, can have multiple mode declaration. While most (if not all) functions we have come across would have their arguments having the mode in and their return value the mode out, but in Mercury, the reverse (i.e. arguments having out and return value having in) is technically possible. See the following example, in which the definition of function f1 can be seen as asking the question "which tuple value can be seen as a result of reversing which tuple's components":

```
1 :- module ex01.
2 :- interface.
4 :- use_module io.
6 :- pred main(io.state::di, io.state::uo) is det.
7 :- implementation.
9 :- import_module int, string.
10 :- func f1({int, string}) = {string, int}.
11 :- mode f1(out) = in is det.
```

3.6 User-defined types

Equivalence types

The phrase "equivalence types" in Mercury refers to what's more commonly known as *type aliases*, i.e. they are types that are defined to be fully equivalent in every way to some other types. This is often done for the sake of code readability, for example:

```
1 :- type money == int.
2 :- type assoc_list(Key, Value)
3 == list(pair(Key, Value)).
```

While money and assoc_list(Key, Value) are equivalent to int and list(pair(Key, Value)), one reading the code would have a basic idea that certain ints are intended for values representing an amount of money and certain lists are intended to be used as an association list.

(Honestly, the name "equivalence types" might bring confusion for people who have a Standard ML background, in which the idea of *equality type* refers to types that one can compare values with = in instead of type aliases, similar to types that have an Eq typeclass constraint in Haskell.)

Discriminated unions

Discriminated union, or "tagged union" is a very useful (and thus commonly supported among functional languages) construct. First of all, it can be used to implement what's often called the "enumerated types" in other languages. For example:

This is not unlike the following in C:

```
typedef enum DaysOfWeek {
2
      MONDAY,
3
      TUESDAY
      WEDNESDAY,
4
5
      THURSAY,
6
      FRIDAY.
      SATURDAY,
      SUNDAY,
8
9
 } DaysOfWeek;
```

The general syntax is as follows. (Note that the arrow here consists of three dashes. Mercury uses 2-dash arrow and 1-dash arrow for other things, so one should be mindful not to mix them up.)

The core idea of a discriminated union type is exactly how it sounds: it's a union type with a special value (often called a "tag"; this is also why discriminated unions are also called "tagged union") that one uses to discern the different possible forms of the union. For example, you might often see patterns like this in languages like C:

```
typedef enum ExprType {
2
      LIT_INTEGER,
      BIN_OP,
3
4
      UNARY_OP
      FUNC_CALL,
5
  } ExprType;
6
  typedef struct Expr {
8
9
      ExprType typ;
      // the ~typ~ field is checked before using the ~value~ field.
10
      union {
11
12
           int lit int;
13
           struct { struct Expr *1, *r; char* bop; } bin_op;
           struct { struct Expr *b; char* uop; } un_op;
14
           struct { char* fun_name; struct Expr* args } func_call;
15
      } value;
16
17 }
```

It can be simply defined in Mercury as follows:

We don't need to have a separate: - type exprtype; the tags are "built into" the type expr. When a type is defined this way, these tags are also defined as constructors, i.e. we can obtain a value of expr by simply using the

terms lit_integer/1, bin_op/3, unary_op/2 or func_call/2; for example, the term lit_integer(3) and the term bin_op(lit_integer(3), lit_integer(4), "+") would've automatically determined to have the type of expr.

To retrieve the different values stored within, we can use unification:

```
% ...
1
2
  SomeVar = some_funcall(some_args),
                                       % ...do something w/ IntVal
3
  ( ( SomeVar = lit_integer(IntVal),
    ( SomeVar = bin_op(Left, Right, Op), % ...do something
5
6
    );
    ( SomeVar = unary_op(Body, Op), % ...do something
8
    );
9
    ( SomeVar = func_call(Head, Args), % ... do something
10
11
  )
```

This can be tedious at times, especially in cases where you need to have a new value where only some of the fields have changed, in which you have no choice other than reconstructing the term from the ground up once you've defined your data type this way. But - if we choose to use the record syntax,

Certain commonly-used types are actually defined this way. See the definition of lists and key-value lists:

Booleans are also defined this way:

```
1 % in module `bool`
2 :- type bool
3   ---> no
4   ; yes.
```

Abstract types

Simply put, they are the declarations we've discussed above but without the definition part. These are mainly used in the **interface** section of a module for the cases where you don't wish to expose the internals of a type to other module:

```
1 :- interface.
2 :- type t1.
3 :- type t2(T1, T2).
4 % ...
5
6 :- implementation.
7 :- type t1 --> % ...
8 :- type t2(T1, T2) == % ...
9 % ...
```

Of course, you must properly define them in the implementation section of the same module.

3.7 Example 1: Interpreter of a small language

With what we have learned up till now we can now try to write some small programs.

The language

The language we'll be implementing would be a small, imperative one with very basic features like assignment, branching and loops. Its grammar can be described with the following Backus-Naur Form:

```
 \langle stmt \rangle ::= \text{IDENT ':=' } \langle expr \rangle 
 | \text{ 'IF' } \langle expr \rangle \text{ 'THEN' } \langle stmt \rangle \text{ 'ELSE' } \langle stmt \rangle 
 | \text{ 'BEGIN' } \langle stmt \rangle^* \text{ 'END'} 
 | \text{ 'WHILE' } \langle rel \rangle \text{ 'DO' } \langle stmt \rangle 
 | \text{ 'PRINT' } \langle expr \rangle 
 \langle expr \rangle ::= \text{INTEGER} 
 | \text{ VAR} 
 | \text{ } \langle expr \rangle \text{ ('+'|'-'|'*'|'/') } \langle expr \rangle 
 | \text{ ('-') } \langle expr \rangle 
 \langle rel \rangle ::= \text{TRUE} 
 | \text{ FALSE}
```

```
| \( \left( \expr \right) \) ('>'|'<'|'>='|'<='|'!='|'==') \( \expr \right) \) \( \rel \right) \( \left( \expr \right) \) \( \rel \right) \) \( \rel \right) \( \rel \right) \)
```

From this BNF we can easily obtain the type definition:

```
1 :- import_module list.
2 :- type imp_expr
3
     ---> e_val(int)
     ; e_var(string)
     ; e_binop(string, imp_expr, imp_expr)
     ; e_unaryop(string, imp_expr).
  :- type imp_rel
     ---> r_rel(string, imp_expr, imp_expr)
9
     ; r_binop(string, imp_rel, imp_rel)
10
     ; r_unaryop(string, imp_rel)
11
     ; r_true
     ; r_false
12
13 :- type imp_stmt
     ---> s_assign(string, imp_expr)
14
     ; s_begin(list(imp_stmt))
16
     ; s_if(imp_rel, imp_stmt, imp_stmt)
     ; s_while(imp_rel, imp_stmt)
     ; s_print(imp_expr).
18
```

We will need a place to store the mapping between the variable names and their corresponding values:

```
:- import_module map.
2 :- type imp_env == map(string, int).
```

The evaluation of expression is easy. Since we don't need to update the environment while evaluating expressions for this language, we don't need an extra imp_env::out parameter:

```
1 % needed by `throw`
2 :- import_module exception.
4
   :- pred imp_eval_expr(imp_expr::in, imp_env::in, int::out) is det.
5
   imp_eval_expr(X, Env, Result) :-
        ( X = e_val(XVal), Result = XVal );
6
        (X = e_var(XVar),
          ( map__contains(Env, XVar) -> Result = lookup(Env, XVar)
8
          ; throw("Cannot found var: " ++ XVar)
9
10
11
        (X = e_binop(Op, L, R),
          imp_eval_expr(L, Env, LRes),
imp_eval_expr(R, Env, RRes),
( Op = "+" -> Result = LRes + RRes
; Op = "-" -> Result = LRes - RRes
13
14
15
16
          ; Op = "*" -> Result = LRes * RRes
17
          ; Op = "/" -> Result = LRes div RRes
18
            throw("Unsupported bin op: " ++ Op)
19
20
        );
```

```
22  ( X = e_unaryop(Op, Body),
23     imp_eval_expr(Body, Env, BRes),
24     ( Op = "-" -> Result = O - BRes
25     ; throw("Unsupported unary op: " ++ Op)
26     )
27  ).
```

Evaluating relations, however, requires some not-that-intuitive steps if you choose to first evaluate things into boolean values and dispatch on it. The comparison of numerical values are technically predicate calls, which only succeed or fail instead of producing a value like function calls; if we need a boolean value, we must convert this "succeed or fail" state into a boolean value. This is actually very simple, you just do this:

This is available in the standard library by importing the bool module. Equality check is even more different - when written directly, they're considered as a goal of unification instead a predicate call, which means that we can't even use pred_to_bool. One way to go around this is define a dedicated predicate:

```
1 :- func int_eq(int::in, int::in) = (bool::out) is det.
2 int_eq(X, Y) = (if X = Y then yes else no).
```

With this being available we can define the predicate for evaluating relations like this:

```
:- import_module bool.
  :- pred imp_eval_rel(imp_rel::in, imp_env::in, bool::out) is det.
  imp_eval_rel(X, Env, Result) :-
       (X = r_rel(Op, L, R),
4
         imp_eval_expr(L, Env, LRes),
imp_eval_expr(R, Env, RRes),
5
6
         ( Op = ">" -> Result = pred_to_bool((RRes < LRes))
; Op = "<" -> Result = pred_to_bool((LRes < RRes))
7
8
          ; Op = "==" -> Result = int_eq(LRes, RRes)
9
            Op = ">=" -> Result = pred_to_bool((LRes >= RRes))
10
           Op = "<=" -> Result = pred_to_bool((LRes =< RRes))
11
          ; Op = "!=" -> Result = not(int_eq(LRes, RRes))
12
13
            throw("Unsupported rel op: " ++ Op)
14
15
       (X = r_binop(Op, L, R),
16
17
         imp_eval_rel(L, Env, LRes),
         imp_eval_rel(R, Env, RRes),
18
          ( Op = "&&" -> Result = and(LRes, RRes)
19
          ; Op = "||" -> Result = or(LRes, RRes)
20
         ; throw("Unsupported rel op: " ++ Op)
```

```
22
23
       );
24
       (X = r_unaryop(Op, B),
25
         imp_eval_rel(B, Env, BRes),
         ( Op = "!" \rightarrow Result = not(BRes) 
26
         ; throw("Unsupported rel op: " ++ Op)
27
28
29
30
       ( X = r_{true}, Result = yes );
       ( X = r_false, Result = no ).
```

The predicate for executing statements can thus be defined as follows. Evaluating a statement could update the environment (by assignment statements), so we need an imp_env::out to receive the newest version of it:

```
2 :- pred imp_exec_stmt(imp_stmt::in, imp_env::in, imp_env::out) is det.
3 :- pred imp_exec_stmt_list(list(imp_stmt)::in, imp_env::in, imp_env::
      out) is det.
4 :- pred imp_exec_stmt_while(imp_rel::in, imp_stmt::in, imp_env::in,
      imp_env::out) is det.
6 imp_exec_stmt(X, Env, NewEnv) :-
       ( X = s_assign(Var, Expr),
7
         imp_eval_expr(Expr, Env, Val),
8
         NewEnv = map__set(Env, Var, Val)
9
      );
10
       (X = s_begin(Body),
11
12
         imp_exec_stmt_list(Body, Env, NewEnv)
13
14
       ( X = s_{in}(Cond, Then, Else),
15
         imp_eval_rel(Cond, Env, CondRes),
         ( CondRes = yes -> imp_exec_stmt(Then, Env, NewEnv)
16
17
         ; imp_exec_stmt(Else, Env, NewEnv)
18
19
       ( X = s_while(Cond, Body),
20
21
         imp_exec_stmt_while(Cond, Body, Env, NewEnv)
22
23
       (X = s_print(Expr),
24
         imp_eval_expr(Expr, Env, Val),
25
         trace [ io(!IO) ] (
             io__write_int(Val, !IO),
io__write_string("\n", !IO)
26
27
        ),
28
29
         NewEnv = Env
30
31
32 imp_exec_stmt_list([], X, X).
33 imp_exec_stmt_list([X|Xs], Env, NewEnv) :-
34
       imp_exec_stmt(X, Env, NewEnv1),
       imp_exec_stmt_list(Xs, NewEnv1, NewEnv).
35
37 imp_exec_stmt_while(Cond, Body, Env, NewEnv) :-
38
       imp_eval_rel(Cond, Env, CondRes),
39
       ( CondRes = yes ->
             imp_exec_stmt(Body, Env, NewEnv1),
40
             imp_exec_stmt_while(Cond, Body, NewEnv1, NewEnv)
```

```
42 )
43 ; NewEnv = Env
44 ).
```

You can tell the Mercury compiler didn't pick up the fact that imp_-exec_stmt_while being potentially diverge since its determinism det somehow checks out.

A program in our language is conceptually just a list of statements. We will execute it by calling imp exec stmt list with an empty environment:

```
1 :- pred imp_run(list(imp_stmt)::in) is det.
2 imp_run(X) :-
3     Env = map__init,
4     imp_exec_stmt_list(X, Env, _).
```

Then we can call it in our main:

```
main(!IO) :-
2
      % X := 100
3
       % SUM := 0
       % WHILE X > 0 DO
4
5
             BEGIN
                 SUM := SUM + X
6
                 X := X - 1
             END
8
9
       % PRINT SUM
10
       imp_run(
11
           Γ
12
                s_assign("X", e_val(100)),
13
               s_assign("SUM", e_val(0)),
               s_while(
14
                    r_rel(">", e_var("X"), e_val(0)),
15
16
                        s_assign("SUM", e_binop("+", e_var("SUM"), e_var("
17
      X"))),
                        s_assign("X", e_binop("-", e_var("X"), e_val(1)))
18
19
20
               ),
               s_print(e_var("SUM"))
21
           ]
22
23
      ).
```

Compile and run the program, and you'll see the number 5050 is printed.

Full listing

```
1 :- module ex01.
2 
3 :- interface.
4 
5 :- use_module io.
6 :- pred main(io.state::di, io.state::uo) is det.
7 
8 :- implementation.
```

```
10 :- import_module int, string, list, exception, bool, map.
12 :- type imp_expr
13 ---> e_val(int)
      ; e_var(string)
14
15
     ; e_binop(string, imp_expr, imp_expr)
      ; e_unaryop(string, imp_expr).
16
17 :- type imp_rel
      ---> r_rel(string, imp_expr, imp_expr)
18
     ; r_binop(string, imp_rel, imp_rel)
20
     ; r_unaryop(string, imp_rel)
21
     ; r_true
22
      ; r_false.
23 :- type imp_stmt
24
     ---> s_assign(string, imp_expr)
25
      ; s_begin(list(imp_stmt))
26
      ; s_if(imp_rel, imp_stmt, imp_stmt)
      ; s_while(imp_rel, imp_stmt)
27
28
      ; s_print(imp_expr).
29
30 :- type imp_env == map(string, int).
31
32 :- pred imp_eval_expr(imp_expr::in, imp_env::in, int::out) is det.
33 imp_eval_expr(X, Env, Result) :-
34
       ( X = e_val(XVal), Result = XVal );
35
       (X = e_var(XVar),
          ( map__contains(Env, XVar) -> Result = lookup(Env, XVar)
36
          ; throw("Cannot found var: " ++ XVar)
37
38
39
       (X = e_binop(Op, L, R),
40
         imp_eval_expr(L, Env, LRes),
41
42.
         imp_eval_expr(R, Env, RRes),
         ( Op = "+" -> Result = LRes + RRes
; Op = "-" -> Result = LRes - RRes
43
44
         ; Op = "*" -> Result = LRes * RRes
45
          ; Op = "/" -> Result = LRes div RRes
46
47
          ; throw("Unsupported bin op: " ++ Op)
48
49
       );
       ( X = e_unaryop(Op, Body),
50
         imp_eval_expr(Body, Env, BRes),
( Op = "-" -> Result = 0 - BRes
51
52
          ; throw("Unsupported unary op: " ++ Op)
53
54
55
       ).
56
57 :- func int_eq(int::in, int::in) = (bool::out) is det.
int_eq(X, Y) = (if X = Y then yes else no).
60 :- pred imp_eval_rel(imp_rel::in, imp_env::in, bool::out) is det.
61 imp_eval_rel(X, Env, Result) :-
62 (X = r_rel(Op, L, R),
63
         imp_eval_expr(L, Env, LRes),
64
          imp_eval_expr(R, Env, RRes),
          ( Op = ">" -> Result = pred_to_bool((RRes < LRes)); Op = "<" -> Result = pred_to_bool((LRes < RRes))
65
66
          % ; Op = "==" -> Result = (LRes, RRes)
67
         ; Op = "==" -> Result = int_eq(LRes, RRes)
```

```
; Op = ">=" -> Result = pred_to_bool((LRes >= RRes))
          ; Op = "<=" -> Result = pred_to_bool((LRes =< RRes))
70
          ; Op = "!=" -> Result = not(int_eq(LRes, RRes))
71
72
            throw("Unsupported rel op: " ++ Op)
73
        );
74
        ( X = r_binop(Op, L, R),
  imp_eval_rel(L, Env, LRes),
  imp_eval_rel(R, Env, RRes),
75
76
77
          ( Op = "&&" -> Result = and(LRes, RRes)
78
          ; Op = "||" -> Result = or(LRes, RRes)
79
80
          ; throw("Unsupported rel op: " ++ Op)
81
82
        );
83
        (X = r_unaryop(Op, B),
          imp_eval_rel(B, Env, BRes),
( Op = "!" -> Result = not(BRes)
84
85
          ; throw("Unsupported rel op: " ++ Op)
86
87
       );
88
        ( X = r_true, Result = yes );
89
        ( X = r_false, Result = no ).
90
91
92 :- pred imp_exec_stmt(imp_stmt::in, imp_env::in, imp_env::out) is det.
93 :- pred imp_exec_stmt_list(list(imp_stmt)::in, imp_env::in, imp_env::
       out) is det.
94 :- pred imp_exec_stmt_while(imp_rel::in, imp_stmt::in, imp_env::in,
       imp_env::out) is det.
95
96 imp_exec_stmt(X, Env, NewEnv) :-
        ( X = s_assign(Var, Expr),
97
          imp_eval_expr(Expr, Env, Val),
98
99
          NewEnv = map__set(Env, Var, Val)
100
        ( X = s_begin(Body),
101
102
          imp_exec_stmt_list(Body, Env, NewEnv)
103
        );
        ( X = s_if(Cond, Then, Else),
104
          imp_eval_rel(Cond, Env, CondRes),
( CondRes = yes -> imp_exec_stmt(Then, Env, NewEnv)
105
106
          ; imp_exec_stmt(Else, Env, NewEnv)
107
108
          )
109
        );
        ( X = s_while(Cond, Body),
110
          imp_exec_stmt_while(Cond, Body, Env, NewEnv)
111
112
113
        ( X = s_print(Expr),
114
          imp_eval_expr(Expr, Env, Val),
115
          trace [ io(!IO) ] (
               io__write_int(Val, !IO),
116
               io\_write\_string("\n", !I0)
117
          ),
118
119
          NewEnv = Env
120
121
122 imp_exec_stmt_list([], X, X).
123 imp_exec_stmt_list([X|Xs], Env, NewEnv) :-
        imp_exec_stmt(X, Env, NewEnv1),
124
      imp_exec_stmt_list(Xs, NewEnv1, NewEnv).
```

```
127 imp_exec_stmt_while(Cond, Body, Env, NewEnv) :-
128
        imp_eval_rel(Cond, Env, CondRes),
129
        ( CondRes = yes ->
130
              imp_exec_stmt(Body, Env, NewEnv1),
131
              imp_exec_stmt_while(Cond, Body, NewEnv1, NewEnv)
          )
132
133
         NewEnv = Env
134
136 :- pred imp_run(list(imp_stmt)::in) is det.
137 imp_run(X) :-
138
        Env = map__init,
        imp_exec_stmt_list(X, Env, _).
139
140
141 main(!IO) :-
142
        imp_run(
143
                s_assign("X", e_val(100)),
144
                s_assign("SUM", e_val(0)),
145
146
                 s_while(
                     r_rel(">", e_var("X"), e_val(0)),
147
148
                     s_begin(
149
                              s_assign("SUM", e_binop("+", e_var("SUM"),
150
       e_var("X"))),
                             s_assign("X", e_binop("-", e_var("X"), e_val
151
        (1)))
                         ]
152
153
                ),
154
                s_print(e_var("SUM"))
155
            ]
156
157
        ).
```

3.8 Using multiple modules

We are going to need this so it's better to talk about it early than late.

Basic module usage

It's probably good measure to go through the basic usage of modules. There are two ways to import a module:

- import_module, which we have been using up till now. Names imported this way do not need to be fully qualified; the module name, when referring to the names defined within, can be omitted until there's ambiguity.
- use_module, whose syntax is the same as import_module, but the names imported this way needs to be fully qualified all the time.

Most of the time people would use a period . to separate the module name and the referred name, but Mercury supports using two underscores __ as well; for example, instead of calling io.write_string, you can call io_write_string.

There is also this thing called "module-local mutable variable", which does exactly what it sounds like. However, this feature involves Mercury's purity system, and for that reason we'll talk about it when we get there.

Building with multiple modules

The Melbourne Mercury Compiler comes with two build tools named mmake and mmc --make. The reason why the latter is named mmc --make is because that to use this build tool you do need use the same executable but with a special command line argument. Both covers roughly the same set of features (with an amount of difference in details), both are terribly confusing and under-documented, and both are very confusing. The official documentation recommends mmc --make because "this is the build tool that is likely to receive more development in the future". I recommend using mmc --make instead of directly calling mmc because compiling multi-module program with the Melbourne Mercury Compiler is quite complex and not a task that can be achieved by only calling mmc; it's not unlike building a multi-file C project.⁴ For single-module programs (the ones we have seen till now), you can directly replace mmc [main module] with mmc --bare [main module] (e.g. mmc --make rall.m instead of mmc rall.m). For multi-module programs, you can also use mmc --make [entry] and expect it to handle module dependencies automatically and give you an executable without much errors.

Submodules

There are times when you'd like to group certain things within a module into its own subgroup; in Mercury this can be achieved by defining a submodule. There are two kinds of submodule in Mercury: ones that are defined and embedded within the "main" module, and ones that are in a separate file. For the former ones, you do it like this:

```
1 :- module main_module.
2 :- interface.
3
4 :- module submod.
5 :- interface.
6 % submod interface goes here.
```

⁴I'd say Mercury might have more hassle than C at this point.

Naturally, if you don't want the definition of submod being visible from the outside, you put its interface section in the declaration of the parent module's implementation section; but to put the implementation section of a nested submodule in the interface section of the parent module is forbidden.

If you want to put the definition of the submodule in a separate file, you must do the following:

- The content of that submodule must be in a file with the file name being the fully qualified name of the submodule (e.g. the source for the submodule submod of the parent module mainmod1 must be put in the file with the name mainmod1.submod.m). (Using a different module name is possible; see the next section.)
- Add an include_module declaration in the parent module, like this:

```
1 :- module parent_module.
2 :- interface.
3 :- include_module submod1, submod2, ..., submod_n.
```

Either way, having the definition of the submodule doesn't mean the names within the submodule is immediately available within the parent module; you still need to import the submodule explicitly if you want to use it.

Using a different module name

Normally module names should match with the name of the file; but if you insist on using a different name, the Mercury compiler would need to know which name corresponds to which file. The compiler would expect this information in a file named Mercury.modules residing in the same directory as the module that uses the name-mismatch module. The mmc command provides a flag to generate this file: to check all the modules within a directory and generate the Mercury.modules, run mmc -f. This file can technically be modified so that the build tool would use modules

located at other places; one may manage different libraries a project might use this way, but I'm not sure this is part of the intended usage.

3.9 Exception handling

We already know that we can throw exceptions using throw/1. Now we learn how to handle them.

Handling with predicates

A few predicates have been provided in the exception module:⁵

- try(Goal, Result): Result [].
- try_io(Goal, Result, !IO): Same as try/2, but with extra parameters for IO.
- try_all(Goal, MaybeException, Solution): Collect all solutions of Goal which might throw exceptions into Solution. MaybeException would be bound with a value of type maybe(univ); if an exception did get thrown, that maybe value would contain that exception; and if there has been no exception, it would simply be an empty value.
- finally(Goal, Res, Cleanup, CleanupRes, !IO):

3.10 Higher-order programming

 $^{^{5}\}mathrm{There}$ are other predicates, which have slightly more niche usage so we will not cover them here.

Advanced Concepts & Usage

4.1 Mode & Instantiatedness

Higher-order insts and modes

4.2 Multiple solutions & Determinism

As we have seen in Chapter 2, in Mercury, the following determinism categories exist:

- Deterministic (det): guaranteed to have one and exactly one solution.
- Semideterministic (semidet): have exactly one solution, but does not guarantee to produce it.
- Multisolution (multi): guaranteed to have a solution among possibly many solutions.
- Nondeterministic (nondet): have possibly many solutions, does not guarantee to produce one.
- Failure (failure): cases where there's zero solutions. They are not actual errors but a part of the logic (e.g. arity mismatch during unification, which will never produce a solution because the arity is different).
- Errorneous (errorneous): also have zero solutions, but they **do** represent actual errors which in other languages would be represented in the form of runtime exception throw or panic.

Commited-Choice Determinism

- 4.3 Definite Clause Grammar
- 4.4 Typeclasses
- 4.5 Existential Types
- 4.6 Purity System
- 4.7 Working with other people's libraries

CHAPTER 5

Parsing With Definite Clause Grammar

Foreign Language Interface

- 6.1 Calling back and forth
- 6.2 Working with the Java grade
- 6.3 Working with the CSharp grade

CHAPTER 7

Debugging Mercury Programs